

Г.Х.Геворкян В.Н.Семенов

ЭЛЕКТРО- ТЕХНИЧЕСКИЕ РАСЧЕТЫ

БЕЛСНК

на языке



```
3100 LET N3=N1-N2
3102 FOR K=1 TO N3
3104 FOR K1=1 TO N1
3106 LET S[K,K1]=0
3108 NEXT K1
3110 NEXT K
3112 FOR K=1 TO N1
3114 LET K1=FNA(P[K,1])
3116 S[1,K1+1]=S[1,K1+1]+K
```

```
3182 FOR K=1 TO N2
3184 LET S[1,K]=FNB(S[1,K])
3186 NEXT K
3188 LET K1=1
3190 FOR K2=1 TO N3
3192 FOR K=K1 TO N1
3194 IF P[K,1]>0 THEN
3196 NEXT K
3198 LET K1=K+1
3200 LET K3=P[K,1]
3202 LET K4=FNA(K3)
3204 FOR K5=1
3206 IF
```

ОГЛАВЛЕНИЕ

Предисловие	3
<i>Глава первая.</i> Особенности расчетов на ПЭВМ	7
1.1. БЕЙСИК – язык программирования для ПЭВМ	7
1.2. Режим диалога	9
1.3. Подпрограмма ввода данных	14
<i>Глава вторая.</i> Матричные методы расчета электрических цепей	18
2.1. Введение	18
2.2. Метод контурных токов	20
2.3. Метод узловых потенциалов	43
2.4. Освоение и тестирование программ	60
<i>Глава третья.</i> Расчет цепей методом эквивалентных преобразований	66
3.1. Метод редукции	66
3.2. Подпрограммы эквивалентных преобразований	72
3.3. Автоматизированные методы расчета	79
3.4. Методы тестирования	90
3.5. Частотные характеристики	94
3.6. Длинные линии	102
<i>Глава четвертая.</i> Статистический анализ качества	109
4.1. Метод статистического моделирования	109
4.2. Программы для метода МК	113
4.3. ДСЧ и обработка данных	122
4.4. Примеры применения метода МК	135
<i>Глава пятая.</i> Основные подпрограммы	139
5.1. Решение систем линейных алгебраических уравнений	139
5.2. Действия с комплексными числами	147
5.3. Решение нелинейных уравнений	155
5.4. Подпрограммы для ввода и вывода информации	157
<i>Приложение 1.</i> Пример расчета длинных линий	171
<i>Приложение 2.</i> Программа "Комплексный калькулятор"	172
Список литературы	183

Г.Х.Геворкян В.Н.Семенов

**ЭЛЕКТРО-
ТЕХНИЧЕСКИЕ
РАСЧЕТЫ
на языке
БЕЙСИК**

**70 ПРОГРАММ НА ПЕРСОНАЛЬНОМ
КОМПЬЮТЕРЕ**



**МОСКВА
ЭНЕРГОАТОМИЗДАТ
1989**

ББК 31.21
Г27
УДК 621.31:681.3.06

Рецензент канд. техн. наук В.Н. Котляровский

Геворкян Г.Х., Семенов В.Н.

Г27 Электротехнические расчеты на языке БЕЙСИК: 70 программ на персональном компьютере. — М.: Энергоатомиздат, 1989. — 184 с.: ил.
ISBN 5-283-00521-6

Описаны программы расчетов на языке БЕЙСИК в области электротехники и энергетики, методы обработки данных, статистический анализ качества. Для каждой программы приведены ее текст, алгоритм, обращение к ней и способы использования на отечественных микроЭВМ.

Предназначена для инженерно-технических и научных работников-электротехников.

Г 2202010000-573 125-89
051(01)-89

ББК 31.21

Производственное издание

**Геворкян Геворк Хоренович
Семенов Владимир Николаевич**

ЭЛЕКТРОТЕХНИЧЕСКИЕ РАСЧЕТЫ НА ЯЗЫКЕ БЕЙСИК
70 программ на персональном компьютере

Редактор *В. В. Краюшкин*
Зав. редакцией *М. П. Соколова*
Редактор издательства *Н. Б. Фомичева*
Художественные редакторы *В. А. Гозак-Хозак, Г. И. Панфилова*
Технические редакторы *Г. Н. Лядухина, Т. Н. Тюрина*
Корректор *С. В. Малышева*

ИБ № 2248

Набор выполнен в издательстве. Подписано в печать с оригинала-макета 18.08.89. Т-15826. Формат 60 x 88 1/16. Бумага офсетная № 2. Печать офсетная. Усл. печ. л. 11,27. Усл. кр.-отт. 11,63. Уч.-изд л. 11,67. Тираж 70000 экз. (2-й завод 35001—70000 экз.) Заказ 6823. Цена 60 к.

Энергоатомиздат, 113114, Москва, М-114, Шлюзовая наб., 10.

Отпечатано в ордена Октябрьской Революции и ордена Трудового Красного Знамени МПО "Первая Образцовая типография" Государственного комитета СССР по делам издательств, полиграфии и книжной торговли. 113054, Москва, М-54, ул. Валуевская, 28.

ISBN 5-283-00521-6

© Энергоатомиздат, 1989

ПРЕДИСЛОВИЕ

Когда десять лет назад авторы начали работу по созданию математического обеспечения для мини- и микроЭВМ на языке БЕЙСИК, многие наши коллеги восприняли это как неперспективный шаг, объясняя его любовью к программистской экзотике (особенно если учесть, что разрабатывались не только чисто вычислительные программы для электротехнических задач, но и управляющие программы для систем автоматизации эксперимента). С тех пор положение изменилось кординально, персональные компьютеры (ПЭВМ) становятся основным инструментом расчета инженера-электрика, а все ПЭВМ имеют интерпретатор с языка БЕЙСИК. Число публикаций по языку огромно, только учебников, монографий, справочников и пособий насчитывается более 200; это число продолжает расти. В нашей стране в дополнение к уже изданным [1–5] планируется только в центральных издательствах выпуск десятка различных учебников и пособий. Таким образом, учебной литературы в ближайшее время будет достаточно, однако остается нерешенной проблема распространения пакетов прикладных программ.

Предлагаемая читателю книга содержит описание и полные тексты программ, предназначенных для проведения наиболее распространенных в инженерной практике расчетов стационарных режимов в электрических цепях с линейными элементами. Нам кажется, что ее целесообразно использовать не только в производственных, но и в учебных целях – для обучения студентов электротехнических специальностей.

И действительно, в настоящее время специалистами обсуждается вопрос о том, что использование ПЭВМ приводит к необходимости коренного изменения не только инженерного труда, но и процесса обучения студентов. Возможность диалога – прямого взаимодействия с ПЭВМ повышает наглядность и доступность учебного материала, снижает “математический барьер” между условием и решением задачи, делает более осязаемой и доступной картину физических процессов в электротехнических объектах, и в итоге ускоряется обучение.

Рассмотрим, как осваивать приведенные в книге программы. Они предназначены для расчета электрических цепей ПЭВМ. Читатель должен знать электротехнику в пределах курса ТОЭ, а для освоения метода статистического моделирования (см. гл. 4) достаточно знания обычных вузовских курсов по теории вероятностей и математической статистике.

Читатель должен знать также программирование на языке БЕЙСИК. Для этого достаточно познакомиться с одной из книг, перечисленных в списке литературы, или с пособиями по языку, входящими в состав документации к имеющейся ПЭВМ. Если читатель уже владеет каким-либо алгоритмическим языком высокого уровня, то изучение БЕЙСИКА займет у него всего несколько часов.

Программы, представленные в книге, отлажены и прошли апробацию при решении конкретных задач. Это весьма ответственное утверждение об их надежности (к сожалению, не доказанное строго, а имеющее чисто практическое основание) позволяет нам полагать, что читателю остается лишь ввести выбранную программу с клавиатуры и провести расчет. Однако и этот процесс не так уж прост ввиду большой вероятности субъективных ошибок при вводе и отсутствии достаточного опыта. Поэтому лучше проводить отладку локально, по частям. Для этого после ввода строк, образующих законченный фрагмент, следует придумать небольшой тест, снабдить текст строками с печатью (выводом) результатов и провести проверку введенного фрагмента. Таким образом может быть выявлено большинство ошибок, которые в целой программе выявить, локализовать и исправить намного труднее (известно, что время отладки программ пропорционально квадрату их длины или числа строк).

После локальной проверки следует аналогично отладить отдельные модули, подпрограммы, а затем запустить общий тест. Такими тестами снабжено большинство описанных программ (см. § 2.4 и 3.4). Это позволит оценить работоспособность программы в целом.

Однако на этом работа с программами не завершается. Известно, что на комментарии должно приходиться около трети текста программы и что ". . . отсутствие комментариев — достаточный повод для увольнения программиста" [14]. И тем не менее приводимые листинги имеют минимальные пояснения, а часто лишены их. Это объясняется тем, что мы рассматриваем данную книгу как большой развернутый комментарий к приводимым программам. Пользователю можно рекомендовать, после того как текст будет готов (а еще лучше — в процессе подготовки его по частям), снабжать его подробными комментариями. На это не следует жалеть ни времени, ни памяти ЭВМ. Опыт показывает, что пользователю ПЭВМ зачастую проще и удобнее вывести на дисплей часть текста программы с комментариями и прочесть его, чем обращаться к инструкциям и прочей документации. Развернутые комментарии упрощают работу пользователей, и он сам должен написать их.

Большинство программ снабжено одним вариантом ввода данных (он описан в гл. 1 — ПП 2000), что должно было обеспечить краткость изложения. Однако в зависимости от назначения программ (в частности, при использовании в обучении ввод должен быть более обширным, пространственным, диалог развитым) можно написать нужные подпрограммы ввода данных.

В гл. 5 приведен в качестве примера еще один вариант ПП 2000 для ввода данных, однако он подходит не для всех марок ПЭВМ.

Вывод информации также оформлен минимально кратко. Можно рекомендовать сделать более развернутый вывод результатов, снабдив их подробными комментариями, пояснениями, обеспечить протоколы работы таблицами, графиками, облегчающими восприятие результатов вычислений, наконец, добавив запись выходной информации на машинные носители.

Все программы написаны на минимальном подмножестве языка БЕЙСИК, таким, что они могут выполняться на любой версии языка (см. § 1.1) и на любой ЭВМ. Благодаря этому обеспечена, как нам кажется, максимальная транспортабельность описанного математического обеспечения. Недостатком такого подхода является то, что он порождает упрощенность некоторых программных конструкций. Стремление повысить легкость восприятия, читабельность текста также приводили к использованию наиболее простых операторов и процедур. Поэтому, если версия языка БЕЙСИК, реализованная на имеющейся у пользователя ПЭВМ, допускает более сложные, экономные и компактные формы представления отдельных фрагментов, то можно рекомендовать их использовать.

Опыт внедрения описанных в книге программ показывает, что пользователи обычно приспособливают процедуры ввода-вывода к своим потребностям, уровню квалификации и опыта: сокращается диалог при многовариантных расчетах или, наоборот, он расширяется за счет дополнительных подсказок, инструкций; кроме того, изменяют форму протокола печати результатов, вводят промежуточный вывод на дисплей для контроля за процессом (что позволяет более динамично управлять им), добавляют также графические процедуры (что увеличивает наглядность, доступность и обзорность результатов). При наличии версий БЕЙСИКа, позволяющих записывать несколько операторов в одну строку, можно, используя эту возможность, сократить длину и ускорить работу программы. Так, например, число строк в программе ПП 2000 из гл. 1, приведенной на с. 14–16 за счет этого может быть сокращено на одну треть.

Если же читателю нужно только одно – не вдаваясь в особенности строения программ, провести расчет конкретной электрической цепи, то можно взять одну из приведенных в гл. 2 и 3 программ с именами, оканчивающимися на 3 или 4 (например, МУПЗ или МКТ4), ввести ее в ПЭВМ, протестировать и начать счет.

В рамках одной книги невозможно с одинаковой глубиной и степенью детализации осветить все вопросы применения языка БЕЙСИК при решении задач, стоящих перед инженерами и научными работниками в области электротехники. Эти задачи весьма многообразны: от традиционных расчетов электрических цепей до новейших методов оптими-

защи, от автоматизированных систем управления и САПР до проблем искусственного интеллекта, причем, чем более доступной становится вычислительная техника, тем шире становится область ее применения. Здесь рассматриваются лишь некоторые аспекты применения ПЭВМ, и на выбор тематики, несомненно, оказали влияние и субъективные интересы авторов. Стремление представить в распоряжение читателя набор полностью отлаженных, готовых к использованию программ и превратить читателя в активного пользователя вынудило авторов еще более сократить тематику книги, уменьшить число описываемых методов и алгоритмов расчета. С учетом этих обстоятельств и сформировалось содержание данной книги.

Работа авторов распределена следующим образом: гл. 1–3 написаны совместно Г.Х. Геворкяном и В.Н. Семеновым, гл. 4 – Г.Х. Геворкяном совместно с В.Х. Геворкяном, гл. 5 – В.Н. Семеновым.

Если читатели сочтут возможным сообщить о недостатках, пожеланиях, поделиться опытом своей работы, то авторы с благодарностью учтут это в своей работе. Все замечания просим направлять по адресу: 113114, Москва, М-114, Шлюзовая наб., 10, Энергоатомиздат.

Авторы

ОТ ИЗДАТЕЛЬСТВА

Уважаемый читатель! В Энергоатомиздате подготовлена эталонная дискета с записью всех программ, описанных в книге. Она предназначена для IBM PC совместимых ПЭВМ, диаметр 133 мм, операционная система MSDOS (версия 3,0). Суммарная емкость дискеты 360 кбайт. Тексты программ записаны в кодах ASCII.

По вопросу приобретения копии с этой дискеты обращаться по адресу: 113114, Москва, Шлюзовая наб., 10, Энергоатомиздат, группа реализации продукции.

ОСОБЕННОСТИ РАСЧЕТОВ НА ПЭВМ

1.1. БЕЙСИК — язык программирования для ПЭВМ

Непрерывное развитие технологии производства средств вычислительной техники, обеспечивавшее экспоненциальный рост мощности ЭВМ в направлении к сверх- и супер-ЭВМ, неожиданно привело к диалектическому скачку — появились микропроцессоры, микроЭВМ и ПЭВМ. Их доступность, низкая стоимость аппаратной части, дешевизна машинного времени, легкость обслуживания и интерактивность при эксплуатации привели к существенной переоценке роли ЭВМ. Следствия появления таких машин весьма многообразны в сфере и науки, и техники: они вызывают необходимость изменять технологию расчетов и их децентрализацию, переходить к иному распределению функций во взаимодействии человек — ЭВМ, изменять критерии оценки вычислительных методов и качества программ, разрабатывать новые языки программирования или модифицировать версии старых языков, изменять архитектуру построения ЭВМ, вычислительных сетей и систем. Все это привело к расширению круга задач и появлению нового поколения задач (машинная графика, редактирование текстов, другие нестандартные задачи).

Основным языком высокого уровня на микроЭВМ и ПЭВМ стал БЕЙСИК [1–5]. Созданный первоначально в целях обучения и использования в системах с разделением времени, применявшийся для решения сравнительно простых задач с быстрым получением результатов расчетов [6], БЕЙСИК превратился в настоящее время в некий стандарт языка высокого уровня для персональных ЭВМ. Он обязательно включается в математическое обеспечение всех таких машин, имеется на многих мини-ЭВМ и больших ЭВМ. О его популярности свидетельствует тот факт, что можно насчитать более 300 книг, посвященных различным аспектам программирования на этом языке.

Цель данной книги — способствовать распространению проведения электротехнических расчетов на ПЭВМ. Исходя из желания сделать описываемые программы максимально доступными, было естественно остановить свой выбор на БЕЙСИКе — наиболее распространенном и простом алгоритмическом языке, обладающем повышенной по сравнению с другими языками программирования транспортабельностью (мобильностью) программ [7]. Последнее — следствие простоты, даже примитивности его операторов и структуры. В частности, перенос математи-

ческого обеспечения существенно упрощается вследствие применения в БЕЙСИКе бесформатного ввода-вывода.

Однако более чем двадцатилетнее развитие языка привело к появлению около четырех десятков его версий и диалектов. В одних увеличено число операторов до 120–140 (почти на порядок по сравнению с первоначальной версией), в других введены средства, обеспечивающие программисту возможности, обычно свойственные АСЕМБЛЕРУ, в третьих учитываются многообразные варианты расширения конфигурации машин, дополнительные аппаратные средства. Все это, естественно, уменьшало возможности переноса математического обеспечения с одной ЭВМ на другую.

Авторы ознакомились с большинством известных в стране диалектов БЕЙСИКа, включенных в математическое обеспечение разных типов ЭВМ – серии СМ, АСВТ, ЕС ЭВМ и "Электроника", а также ряда марок отечественных ПЭВМ. Их анализ показал, что можно ограничиться некоторым минимальным подмножеством языка, таким, что оно, с одной стороны, позволяло бы решать достаточно широкий набор задач электротехнических расчетов, а с другой, могло бы быть реализовано на известных версиях для большинства типов ЭВМ при минимальных требованиях к аппаратуре (вычислительным ресурсам).

В качестве операторов ввода-вывода используются только

PRINT, INPUT, READ,

причем, PRINT бесформатный и без указания кода внешних устройств, а INPUT выполняет функцию только ввода, исключены операторы COM, HEX, PEEK, POKE и др. В результате кроме перечисленных операторов ввода-вывода остались лишь DIM, LET, GOTO, IF–THEN, GOSUB, RETURN, FOR–NEXT, END, DEF и функции SIN, COS, LOG, SQR, EXP, TAB, RND, SGN, INT.

С этим минимальным набором программных средств разработаны все программы, описанные в данной книге.

Кроме сокращения списка операторов имеются и другие ограничения, уменьшающие избыточность, обычную как для естественных, так и для искусственных языков; так, например, не используются логические условные переходы, не применяются весьма удобные операторы ON ... THEN, ON ... GO SUB. Во главу угла ставится такой тесно связанный с транспортабельностью программ критерий качества, как читабельность (легкость восприятия). Поэтому для обеспечения понятности и простоты чтения текста не применяется запись нескольких операторов в одной строке, не используются арифметические выражения в операторе печати, строки пишутся по возможности короче, комментарии используются как разделители, а отдельные функциональные блоки описываются в виде подпрограмм и также четко выделяются в тексте. Все это, по нашему мнению, должно упростить чтение, понимание и освоение программ.

По сравнению с языком высокого уровня (ФОРТРАН, ПЛ/1), обычно применяющимся при проведении электротехнических расчетов на больших ЭВМ, БЕЙСИК обладает рядом недостатков: ограниченность длины массивов, отсутствие локальных переменных, слабо развитый аппарат подпрограмм, "маловыразительные" идентификаторы [8], малые возможности для написания структурных программ [9]. Кроме того, программы, закодированные на БЕЙСИКе, обычно транслируются интерпретатором, а не компилятором, это приводит к понижению быстродействия готовых программ в несколько раз, а иногда и на порядок по сравнению с теми же программами, написанными, например, на ФОРТРАНе. Наконец, многие разделяют мнение, что важнейшим препятствием применению БЕЙСИКа в электротехнических расчетах является малый объем оперативной памяти ПЭВМ (обычно 16–32 К). Однако развитие технологии интегральных схем привело к тому, что современные микропроцессоры превосходят процессоры больших ЭВМ, выпускающихся в 60-х годах, уже выпускаются ПЭВМ, сопоставимые по быстродействию и объему оперативной памяти с младшими выпусками ЕС ЭВМ (например, ЕС 1022). Это позволяет считать, что многие отмеченные недостатки языка БЕЙСИК будут сняты поступательным развитием вычислительной техники.

Тем не менее описанные в книге программы разрабатывались так, чтобы их можно было использовать даже на ПЭВМ с минимальным объемом ОЗУ.

1.2. Режим диалога

Работа на микроЭВМ и ПЭВМ в большинстве областей их применения, в том числе и в электротехнике, организуется в диалоговом режиме. Однако особенности и методика организации такого взаимодействия человека с ЭВМ еще недостаточно освещены в отечественной литературе. Поэтому сначала будут рассмотрены общие формы диалога, а затем — особенности ввода-вывода данных в этом режиме. Это позволит в последующих главах не затрагивать вопросы интерактивного взаимодействия и уделять все внимание тем частям программ, которые реализуют непосредственно вычислительные алгоритмы.

Развитие вычислительной техники показало, что наилучшие результаты ее применения достигаются при диалоговом взаимодействии человека с ЭВМ, когда динамично сочетаются эвристические способности человека и жесткая логика машины, когда гибкое мышление и способность человека восполнять недостающую информацию, его умение оперировать неформализованными понятиями дополняются высокой точностью и колоссальной скоростью ЭВМ. Все это приводит к огромному расширению круга решаемых задач и увеличивает число потребителей машинной информации, включая и непрофессиональных пользователей.

Диалог повышает эффективность труда и профессиональных пользователей, особенно если процесс обработки сложен и трудно поддается формальному анализу и описанию (например, многокритериальная оптимизация, обработка изображений и т.п.). В такой ситуации расчет выгоднее вести в диалоговом режиме: оператор, вызывая промежуточные результаты и анализируя на их основе процесс вычислений, имеет возможность вмешаться в него на любой стадии и выбрать в зависимости от сложившегося положения ту или иную численную процедуру, видоизменить расчет, указав иные параметры, и т.п. При этом человек выступает не только как потребитель результатов счета, но и как источник информации для ЭВМ и диспетчер ее работы.

В диалоговом режиме происходит интенсивный обмен информацией – входными, промежуточными и выходными данными – между человеком и ЭВМ. Первые (вырабатываемые человеком для ЭВМ) должны быть полными, точными и легко изменяемыми. Сообщения, выдаваемые ЭВМ в процессе работы (промежуточные результаты), должны содержать указания о ходе расчетов и появившихся ошибок, рекомендации об их исправлении, контрольные результаты. Кроме того, возможны различные запросы, например, о выборе направления счета, в ответ на которые пользователем вводятся необходимые указания. Наконец, выходные данные, вырабатываемые ЭВМ для человека, должны быть краткими, легко обозримыми и воспринимаемыми (например, во многих случаях достаточно выводить числа всего с двумя-тремя значащими цифрами).

Диалоговый режим в настоящее время рассматривается как основная форма общения человека с ЭВМ и широко используется при проведении научно-технических расчетов. Поэтому при разработке программ необходимо иметь в виду не только их алгоритмы и другие вычислительные аспекты, но и набор средств, в полной мере реализующих возможности диалога, в том числе самое большое его преимущество – оперативность взаимодействия. Это повышает производительность труда программистов по данным разным авторам от 50% до нескольких раз по сравнению с пакетным режимом.

Интерактивное взаимодействие осуществляется с целью или сформировать (изменить) алгоритм вычислений, или ввести необходимые для расчета данные. Далее будут приведены примеры таких видов диалога: для первого случая в § 3.3 при описании программы ПРЭСЗ, для второго случая в § 1.3.

Информация, которой обмениваются человек и ЭВМ, может быть закодирована по-разному: а) цифровая машинная информация; б) условная алфавитно-цифровая мнемоника; в) алфавитно-цифровая информация с общенаучной символикой; г) естественный язык. Эта информация может передаваться в виде символов визуальных, звуковых, тактильных и моторных.

Диалог конструируется из отдельных "сеансов связи", которые могут иметь разные формы. В [10] выделено пять форм диалога: естественный язык; команды; меню; стандартные формы; ответы на вопросы. В дальнейшем рассмотрим только четыре из перечисленных форм, исключая естественный язык.

Диалоговые команды применяются при разветвлении программ (когда ответ пользователя интерпретируется и происходит переход на подпрограмму или ветвь основной программы), при вводе небольшого числа исходных данных или для вывода на дисплей напоминаний оператору о необходимых действиях. Например,

```
100 PRINT "ВКЛЮЧИТЕ РЕЛЕ № 4"  
110 PRINT "СНИМИТЕ КАСЕТУ"  
120 INPUT K
```

В отличие от команд (а также вопросов и напоминаний), дающих мало активности пользователю, другая форма диалога — меню — является альтернативной (выбор из нескольких равноценных вариантов) и более активной. Например, в начале работы системы автоматизации эксперимента целесообразно поместить строки

```
130 PRINT "ВЫБЕРИТЕ И ВВЕДИТЕ КОД РАБОТЫ"  
140 PRINT "1 – ОПРОС КАНАЛОВ"  
150 PRINT "2 – ОПРОС КАНАЛОВ И УПРАВЛЕНИЕ"  
160 PRINT "3 – ОПРОС КАНАЛОВ И ВКЛЮЧЕНИЕ РЕЛЕ"  
170 INPUT K
```

которые позволяют оператору выбрать нужный режим ее работы.

При обработке данных можно предложить выбрать ту методику, которая наиболее подходит в данной ситуации:

```
180 PRINT "ВВЕДИТЕ КОД РАБОТЫ"  
190 PRINT "ГАРМОНИЧЕСКИЙ АНАЛИЗ-1"  
200 PRINT "КОРРЕЛЯЦИОННЫЙ АНАЛИЗ-2"  
210 PRINT "СПЕКТРАЛЬНЫЙ АНАЛИЗ-3"  
220 INPUT K
```

Диалоговый режим позволяет решать задачи не по жестко заданной схеме, а более гибко: предусматривается некоторое число вариантов расчета, а порядок действий не фиксируется заранее, он выбирается пользователем. Разработанные в [11] программы машинных экспериментов с моделями радиоэлектронных устройств позволяют в режиме диалога оценить промежуточные результаты, изменить параметры и структуру исследуемой модели, уточнить ход эксперимента во время моделирования.

Наконец, отметим, что меню существенно упрощает использование ЭВМ, позволяя работать непрофессионалам и слабо подготовленным (в области программирования) пользователям, так как им остается лишь выбрать вариант действий и следовать рекомендациям, выдаваемых ЭВМ по ходу выполнения тех или иных процедур.

Еще один вариант диалога — ответы на вопросы — позволяет как вводить исходные данные, так и управлять процессом вычислений.

Для ввода данных применяется и другой вид диалога — заполнение стандартных форм. В этом случае на экран дисплея выдается не только запрос, но и стандартная форма ответа на него — напоминание пользователю, как он должен оформить ответ на данный вопрос.

Диалог рекомендуется оформлять в максимально удобной для пользователя форме без сокращений, условных обозначений, чтобы максимально упростить и ускорить восприятие человеком выводимой на дисплей информации.

При проведении электротехнических расчетов существенное место в диалоге занимает ввод исходных данных. При его проектировании также ставится цель облегчить работу пользователя и свести до минимума допускаемые им ошибки (а они возникают постоянно под действием "машинобоязни" и в условиях дефицита машинного времени, часто мнимого).

Ввод граничных значений частот в программе расчета частотных функций можно оформить так [12]:

```
370 PRINT "ВВЕДИТЕ НАЧАЛЬНУЮ И КОНЕЧНУЮ ЧАСТОТЫ  
(Гц)"  
380 INPUT F1, F2  
390 IF F1 ≠ 0 THEN 430  
400 PRINT "НЕВЕРНО: F1 = 0"  
410 PRINT "ПОВТОРИТЕ ВВОД, ЧТОБЫ F1 > 0"  
420 GOTO 370  
430 REM ПРОДОЛЖЕНИЕ ВВОДА
```

Таким образом, общая схема ввода данных в диалоге должна быть следующей: а) напоминание (строка 370); б) ввод (строка 380); в) контроль (строка 390); г) "эхо" (строки 400 и 410). Такая организация — с развернутой системой подсказок, диагностических сообщений и напоминаний — сокращает не только время взаимодействия человек — ЭВМ, но и уменьшает объем необходимой технической документации.

Спецификой электротехнических расчетов является необходимость ввода больших массивов исходных данных с повторяющейся структурой. Поэтому наиболее целесообразно организовать их ввод в виде циклически повторяющихся запросов и ответов стандартной формы.

Так, параметры схем можно вводить следующим образом:

```
500 PRINT "ВВЕДИТЕ ПАРАМЕТРЫ ЭЛЕМЕНТОВ СХЕМЫ"  
510 FOR I = 1 TO 5  
520 PRINT "R(OM) L(MGH) C(MKF)"  
530 INPUT R(I), L(I), C(I)  
540 NEXT I
```

где каждый раз напоминаются порядок следования физических величин, единицы их измерения и, если необходимо, ограничения.

Массив вводится циклически, затем данные контролируются и запоминаются.

Человек может допускать большое число ошибок при вводе: вероятность искажения отдельно взятого символа достигает 1%. Эти ошибки увеличивают число сбоев, прогонов, понижают достоверность результатов, особенно при моделировании, где получаемые результаты крайне трудно оценить априорно [12, 15]. Важность контроля обуславливает повышенные затраты на его проведение — в некоторых системах он занимает до 30% машинного времени. В простейшем случае применяется эхо-контроль: после ввода массив отображается на дисплее в удобочитаемой форме. Такой подход (с предусмотренной возможностью внесения оперативных исправлений) окупуется — позволяет исправить до 50% ошибок. Более сложным является машинный контроль для выявления ошибочных данных и ситуаций с выдачей рекомендаций (подсказок) по исправлению ошибок. Текст подсказки должен быть простым, без условных обозначений, даже общепринятые обозначения лучше расшифровать и пояснить текстом [13–15].

В некоторых развитых версиях БЕЙСИКА можно реализовать ввод исходных данных в виде электронных таблиц, которые существенно упрощают контроль данных, так как повышается их наглядность.

Режим диалога замедляет проведение расчетов, поэтому после приобретения опыта эксплуатации программы можно сократить объем выводимой информации (пояснений, подсказок). При проведении многократного счета (многовариантные расчеты, когда большинство данных об электрических параметрах и топологии остаются неизменными) также целесообразно отказаться от диалогового ввода больших массивов и применять, например, ввод данных с помощью операторов DATA—READ. Оформив их в виде подпрограмм, тщательно проверив и записав на машинный носитель, можно каждый раз вводить их после загрузки программы. Это исключит субъективные ошибки и резко сократит время проведения расчетов.

Описанные примеры построения диалога относились к случаю, когда средством обмена являются алфавитно-цифровые символы, однако интерактивное взаимодействие может быть построено и с использованием также графической информации в качестве средства коммуника-

ции. В технических расчетах широко применяется машинная графика, на долю которой по [9] приходится до 30% объема математического обеспечения. Можно рекомендовать пользователям включить в диалог те графические средства, которые уже имеются на отечественных ПЭВМ [16, 17].

Программы, обеспечивающие диалог, длиннее обычных, а значит, соответственно выше трудоемкость их разработки. Однако развитие вычислительной техники со всей очевидностью выявило, что успех в ее применении сегодня определяется не собственно возможностями этой техники, а способностью человека взаимодействовать с ней [18].

1.3. Подпрограмма ввода данных

При проведении электрических расчетов на ЭВМ диалог наиболее часто используется на этапе введения исходных данных. Последние для задач, описываемых в данной книге (задачи расчета линейных цепей с R -, L - и C -элементами), состоят из номеров узлов схемы, значений параметров пассивных элементов в ветвях и источников ЭДС и токов. Наиболее распространенным подходом является организация ввода в диалог, когда последовательно вводятся строка за строкой и ими заполняется соответствующий массив.

В качестве одного из вариантов можно использовать для ввода данных ПП 2000. Она предусматривает выполнение трех видов работ (в дальнейшем называемых режимами работы): 1) ввод ориентированного графа (что необходимо для тестирования описываемых в последующих разделах вычислительных программ); 2) ввод данных для расчета на постоянном токе; 3) ввод данных для расчета на переменном токе.



```

1994 REM *****
1995 REM * ПП ВВОДА ЗНАЧЕНИЙ ЭЛЕМЕНТОВ ВЕТВЕЙ *
1996 REM * И ТОПОЛОГИИ СХЕМЫ *
1997 REM *****
2000 PRINT "УКАЖИТЕ ВИД РАБОТЫ:"
2002 PRINT "1. ВВОД ТОПОЛОГИИ СХЕМЫ,"
2004 PRINT "2. ВВОД ТОПОЛОГИИ И ПАРАМЕТРОВ ВЕТВЕЙ"
2006 PRINT " ( ВЕТВИ СОДЕРЖАТ ТОЛЬКО R ),"
2008 PRINT "3. ВВОД ТОПОЛОГИИ И ПАРАМЕТРОВ ВЕТВЕЙ"
2010 PRINT " ( ВЕТВИ СОДЕРЖАТ ЭЛЕМЕНТЫ R,L,C )."
2012 INPUT N9
2014 PRINT "ВВЕДИТЕ ТАБЛИЦУ ПАРАМЕТРОВ СХЕМЫ,"
2016 PRINT "ПРЕДВАРИТЕЛЬНО ПРОНУМЕРОВАВ ЕЕ УЗЛЫ"
2018 PRINT "НЕОТРИЦАТЕЛЬНЫМИ ЦЕЛЫМИ ЧИСЛАМИ"

```

```

2020 PRINT "( НУМЕРАЦИЯ ДОЛЖНА БЫТЬ ПОСЛЕДОВАТЕЛЬНОЙ --"
2022 PRINT "ОТ 0 И ВЫШЕ БЕЗ ПРОПУСКОВ ЧИСЕЛ )."
2024 PRINT "ПРИ ВВОДЕ ОТВЕЧАЙТЕ НА ЗАПРОС"
2026 PRINT "ПРОГРАММЫ, СОДЕРЖАЩИЙ СЛЕДУЮЩИЕ"
2028 PRINT "ОБОЗНАЧЕНИЯ: N1 И N2 - НОМЕРА УЗЛОВ"
2030 PRINT "ВЕТВИ СХЕМЫ, R,L,C - ЗНАЧЕНИЯ"
2032 PRINT "АКТИВНОГО СОПРОТИВЛЕНИЯ, ИНДУКТИВНОСТИ"
2034 PRINT "И ЕМКОСТИ ВЕТВИ СХЕМЫ."
2036 PRINT "ВВОД N1=-1 ПОЗВОЛЯЕТ ИСПРАВИТЬ"
2038 PRINT "ОШИБКУ, ДОПУЩЕННУЮ НА ПРЕДЫДУЩЕМ "
2040 PRINT "ШАГЕ, А N1=-2 ОЗНАЧАЕТ ОКОНЧАНИЕ"
2042 PRINT "ВВОДА."
2044 LET NO=1000
2046 LET N1=1
2048 LET N2=0
2050 PRINT "ВЕТВЬ N";N1
2052 IF N9=3 THEN 2078
2054 IF N9=2 THEN 2066
2056 PRINT "N1,N2=";
2058 INPUT K1,K2
2060 IF K1<0 THEN 2100
2062 LET P[N1,1]=K1+K2/NO
2064 GOTO 2092
2066 PRINT "N1,N2,R=";
2068 INPUT K1,K2,K3
2070 IF K1<0 THEN 2100
2072 LET P[N1,1]=K1+K2/NO
2074 LET P[N1,2]=K3
2076 GOTO 2092
2078 PRINT "N1,N2,R,L,C="
2080 INPUT K1,K2,K3,K4,K5
2082 IF K1<0 THEN 2100
2084 LET P[N1,1]=K1+K2/NO
2086 LET P[N1,2]=K3
2088 LET P[N1,3]=K4
2090 LET P[N1,4]=K5
2092 LET N1=N1+1
2094 LET N2=N2+(K1-N2)*SGN(1+SGN(K1-N2))
2096 LET N2=N2+(K2-N2)*SGN(1+SGN(K2-N2))
2098 GOTO 2050
2100 IF K1=-2 THEN 2108

```

```

2102 IF N1=1 THEN 2050
2104 LET N1=N1-1
2106 GOTO 2050
2108 LET N1=N1-1
2110 PRINT "СХЕМА СОДЕРЖИТ ";N1;" ВЕТВЕЙ."
2112 PRINT "ПЕЧАТЬ ТАБЛИЦЫ ПАРАМЕТРОВ НУЖНА"
2114 PRINT "(1-ДА, 0-НЕТ) ?"
2116 INPUT K
2118 IF K=0 THEN 2154
2120 IF N9=3 THEN 2132
2122 IF N9=2 THEN 2128
2124 PRINT "N1.N2"
2126 GOTO 2134
2128 PRINT "N1.N2","R"
2130 GOTO 2134
2132 PRINT "N1.N2","R","L","C"
2134 PRINT "-----"
2136 FOR K=1 TO N1
2138 IF N9=3 THEN 2150
2140 IF N9=2 THEN 2146
2142 PRINT P[K,1]
2144 GOTO 2152
2146 PRINT P[K,1],P[K,2]
2148 GOTO 2152
2150 PRINT P[K,1],P[K,2],P[K,3],P[K,4]
2152 NEXT K
2154 RETURN

```

Программа состоит из трех частей. Первая (строки 2000–2042) отображает на дисплей краткую инструкцию пользователю. Вторая часть (строки 2044–2108) осуществляет диалоговый ввод номеров узлов и параметров ветвей цепи. Третья часть позволяет распечатать в целях контроля введенную информацию в виде таблицы. В соответствии с инструкцией до начала работы с ПП необходимо пронумеровать узлы электрической цепи и выбрать направление тока в ветвях. Положительное направление принимается от узла, номер которого указывается при вводе первым, к узлу, номер которого вводится вторым. Для обеспечения работоспособности программ и успешного тестирования правила инструкции должны строго выполняться.

Физические величины вводятся только в следующих единицах — Ом, Гн, Ф. Данные набираются последовательно в соответствии с подсказкой, выдаваемой на каждом шаге на дисплей.

При режиме 3, например, следует на запрос

ВЕТВЬ 6

N1, N2, R, L, C?

вести для ветви № 6, соединяющей узлы 17 и 5, имеющей направление тока в этой ветви от узла 17 к узлу 5 и включающей в себя активное сопротивление $R = 0,13$ Ом, индуктивность $L = 0,01$ Гн и емкость $C = 17$ мкФ, следующую последовательность:

17,5,0.13,0.01,1.7E-5

Примеры ответов на запросы ПП можно найти также в протоколах работы тех программ, которые описываются в гл. 2.

Введенные данные размещаются в двумерном массиве P, который следует описать (оператором DIM) в головной программе, вызывающей ПП 2000. Номера узлов (N1, N2) в упакованном виде размещаются в первом столбце массива. Упаковка состоит в том, что два целых неотрицательных числа (номера узлов) представляются в виде одного десятичного числа, целая часть которого равна 1, а дробная равна 2×10^p , где $p = 1, 2, 3, \dots$, а 10^p называется параметром упаковки (его идентификатор — N0). В листинге ПП 2000 принято $p = 3$, $N0 = 1000$.

Следующие столбцы массива P занимают величины R, L, C. Поэтому при работе в режиме 1 массив P описывается строкой

1 DIM P(N1,1)

для режима 2 строкой

1 DIM P(N1,2)

и для режима 3 строкой

1 DIM P(N1,4)

В указанные строки вместо N1 следует подставить конкретное значение числа ветвей анализируемой схемы. В последующих разделах можно найти ряд примеров описания массива P для некоторых программ.

Помимо массива P выходная информация, вырабатываемая ПП 2000, содержит упаковочный параметр N0, числа ветвей N1 и номер наибольшего узла N2.

Во время диалога возможно исправление случайно допущенных ошибок. Для этого следует на очередной запрос ввести -1 вместо N1, а остальные величины — указать произвольно, например, равными нулю. В результате этого ЭВМ выдаст запрос на ввод предыдущей строки, что позволит исправить ошибку.

Работа ПП 2000 завершается по инициативе пользователя вводом -2 вместо N1 в ответе на очередной запрос.

ПП 2000 является стандартной для всех описанных в данной книге программ расчета электрических цепей. Вместе с тем для пользователя, уже приобретшего достаточный опыт работы с этими программами, возможно простейшее ее усовершенствование: вставив строку

2013 GO TO 2044

можно ускорить ввод, исключив отображение инструкции на экран. В зависимости от особенностей применения возможны и иные способы усовершенствования этой ПП.

В заключение отметим, что хотя ПП 2000 используется в большинстве программ, приведенных в гл. 2 и 3 данной книги, ее следует рассматривать лишь как один из простых вариантов реализации многочисленных возможных проблемно-ориентированных диалоговых систем.

Более сложный вариант программы ввода данных, основанный на идее заполнения таблиц, приведен в § 5.4.

Глава вторая

МАТРИЧНЫЕ МЕТОДЫ РАСЧЕТА ЭЛЕКТРИЧЕСКИХ ЦЕПЕЙ

2.1. Введение

Наиболее распространенной задачей в электротехнике, требующей применения вычислений на ЭВМ, по-видимому, является расчет линейных электрических цепей.

Проектирование, изготовление и эксплуатация электротехнических устройств, аппаратов, машин и электрических систем требуют проведения расчета и анализа стационарных режимов их работы, что зачастую сводится к расчету цепей с линейными элементами. Однако и исследование переходных процессов в электрических устройствах и системах требует знания характеристик установившихся режимов, которые в этом случае выступают как исходные данные для расчета динамических параметров и анализа протекающих во времени процессов. С этой целью применяется эквивалентирование системы – представление ее многополюсником – линейной электрической схемой с комплексными сопротивлениями и источниками ЭДС [19]. Полученная схема замещения рассчитывается по известным методам, и находятся проводимости и сопротивления, токи и мощности, протекающие через узлы схемы, и т.п. Анализ поведения цепи в частотной области также может быть сведен в ряде случаев к расчету линейных схем. А линеаризация и кусочно-линейная аппроксимация характеристик нелинейных элементов (прием, широко применяющийся при различных расчетах) еще более расширяют область использования подобных вычислений.

Проблеме расчета электрических цепей на ЭВМ посвящена обширная литература, список которой насчитывает не один десяток публикаций.

В этих работах рассмотрены разнообразные вычислительные алгоритмы и описаны программы для проведения широкого круга расчетов линейных и нелинейных цепей, установившихся и переходных процессов в них (укажем лишь несколько книг [20–24]). Опубликованы программы на АЛГОЛе и ФОРТРАНе, во многих изданиях публиковались многочисленные программы на БЕЙСИКе для расчетов с использованием мини- и микроЭВМ, на микрокалькуляторах.

Имеется и книга [25], целиком посвященная решению электротехнических задач с помощью программ на языке БЕЙСИК для программируемых калькуляторов. Программы на БЕЙСИКе для ПЭВМ, относящиеся к некоторым частным случаям расчета цепей, приведены в [67], а в [68] есть также программы для анализа цепей с нелинейными элементами и расчета переходных процессов. Однако в отличие от распространенных программ на языках ФОРТРАН, ПЛ/1 и других, образующих весьма сложные системы математического обеспечения – крупные пакеты прикладных программ, охватывающие весьма широкий спектр электротехнических задач, публикации о программах на БЕЙСИКе носят разрозненный, частный характер, а программы недостаточно универсальны и мобильны.

В данной главе описаны алгоритмы и программы на языке БЕЙСИК для расчета линейных электрических цепей с использованием методов контурных токов (КТ) и узловых потенциалов (УП). Оба метода (особенности применения, алгоритмы, вопросы точности и пр.) достаточно подробно освещены в учебной и научной литературе [26–28], так же как и операции с матрицами, и в дальнейшем о них приводятся лишь те сведения, которые необходимы для понимания программ.

В целях упрощения и облегчения освоения программ сначала описываются программы для расчета на постоянном токе, а затем они модифицируются и дополняются для проведения расчетов на переменном токе.

Оба метода основаны на применении матричной алгебры и даются в двух вариантах – с использованием матричных операторов БЕЙСИКа и без них. Такой подход представляется целесообразным, так как упрощается изложение алгоритма расчета: в простейшем случае – для расчета на постоянном токе – программа, записанная с матричными операторами, представляет собой как бы конспективное переложение формул, прямо отражающих методику вычислений. Однако такие операторы имеются не во всех версиях БЕЙСИКа, поэтому приводятся также программные тексты без матричных операторов.

Алгоритмы программ для реализации методов КТ и УП имеют много общего, что позволяет выделить ряд подпрограмм, используемых в обоих случаях, а структуру программ построить единообразно, применив единые обозначения массивов и одинаковую структуру данных.

Подпрограммы обозначаются ПП (НС), где НС – номер первой строки этой ПП. Обращение к ним (вызов ПП) производится стандартным для языка БЕЙСИК способом: GOSUB (НС).

Расчет любой электрической схемы начинается с рассмотрения ее топологии, подготовки исходных данных (часто эта процедура выполняется вручную и требует дополнительных трудозатрат), а затем следует ввод в ПЭВМ исходной информации. Описанная в гл. 1 ПП 2000 для ввода данных о топологии и параметров ветвей применима во всех программах, описанных в гл. 2 и 3. Для ее использования следует в нужном месте вставить обращение GOSUB 2000. Такой подход достаточно удобен, однако не исключает возможности использования иной ПП, разработанной с учетом специфики решаемых задач и квалификации пользователя. В частности, при многовариантном расчете целесообразно данные считывать оператором READ, записав их в текст программы с помощью оператора DATA и восстанавливая его счетчик оператором RESTORE после каждого считывания. При использовании в учебных целях рекомендуется расширить диалог, увеличив количество указаний, рекомендаций и сообщений, введя промежуточную печать и добавив развитую систему контроля как исходной, так и промежуточной информации.

Общность программ, приведенных в § 2.2 и 2.3 (одинаковый математический аппарат, унифицированная структура программ и единообразное оформление данных), позволяет применять также и единую методику проверки их работоспособности. Для тестирования применяются расчеты топологически эквивалентных схем (см. рис. 2.1 и 2.2) с одинаковым набором исходных данных. Это позволяет сопоставлять результаты расчета, полученные методами КТ и УП на постоянном (соответственно – на переменном) токе. Более подробно о тестировании см. в § 2.4.

2.2. Метод контурных токов

▼ Метод КТ основан на втором законе Кирхгофа (законе напряжений) для независимых контуров. В предположении, что цепь не содержит источников тока или последние заменены эквивалентными источниками ЭДС [26–29], он записывается в виде матричного соотношения

$$(\mathbf{SRS}^T)\mathbf{X} = \mathbf{SU} \quad (2.1)$$

где \mathbf{S} – матрица контуров (матрица инциденций), описывающая связь с контурами; \mathbf{R} – диагональная матрица сопротивлений ветвей; \mathbf{S}^T – транспонированная матрица \mathbf{S} ; \mathbf{X} – вектор контурных токов; \mathbf{U} – вектор источников ЭДС ветвей.

Из (2.1) можно найти неизвестные токи в контурах, а по ним – токи в ветвях. При условии, что элементы матрицы \mathbf{R} не зависят от токов, выражение (2.1) представляет собой систему линейных уравнений. Введем обозначение для матрицы коэффициентов системы

$$\mathbf{A} = \mathbf{SRS}^T \quad (2.2)$$

и столбца свободных членов (вектора ЭДС контуров)

$$C = SU. \quad (2.3)$$

Тогда с учетом этих обозначений (2.1) запишется в виде

$$AX = C. \quad (2.4)$$

Его решение имеет вид

$$X = A^{-1}C. \quad (2.5)$$

Для получения токов в ветвях (вектор Z) достаточно умножить X слева на транспонированную матрицу инцидентий (матрицу контуров):

$$Z = S^T X. \quad (2.6)$$

▼ Соотношения (2.2) – (2.5) определяют общую схему расчета по методу КТ и могут быть непосредственно записаны в виде последовательности матричных операторов на языке БЕЙСИК:

```
1150 MAT T = TRN(S)
1160 MAT G = S * R
1170 MAT A = G * T
1180 MAT C = S * U
1190 MAT E = INV(A)
1200 MAT X = E * C
1210 MAT Z = T * X
```

где G и E – вспомогательные матрицы для хранения промежуточных результатов.

Строки 1150–1180 подготавливают матрицу системы и столбец свободных членов, строкой 1190 матрица обращается, затем находятся токи в контурах (строка 1200) и в ветвях (строка 1220).

Чтобы получить всю программу целиком, следует к приведенному фрагменту добавить описание массивов, блок ввода исходных данных, операторы подготовки матрицы инцидентий и вывод результатов расчета.

Прежде, чем описывать недостающие части программы, обратимся к примеру – схеме, изображенной на рис. 2.1. Это позволит излагать материал более конкретно. Данная схема имеет шесть ветвей с включенными в них сопротивлениями, один источник ЭДС и три независимых контура.

Точный подсчет числа ветвей и контуров необходим, так как при использовании матричных операторов требуется точно указать размерности всех массивов. В общепринятых версиях БЕЙСИКа не допускается неявное описание размерности массивов следующим образом:




Рис. 2.1. Тестовая схема для расчета по методу контурных токов

```

1000 DIM R(K, K), U(K), Z(K)
1010 DIM S(L, K), T(K, L), G(L, K)
1020 DIM A(L, L), E(L, L), C(L), X(L)

```

где K — число ветвей схемы; L — число независимых контуров, подсчитываемое по формуле $L = K - M + 1$ (M — число узлов схемы). Поэтому в данном случае для расчета конкретной схемы, изображенной на рис. 2.1, следует в строки 1000–1020 подставить вместо K и L соответственно 6 и 3:

```

1000 DIM R(6,6), U(6), Z(6)
1010 DIM S(3,6), T(6,3), G(3,6)
1020 DIM A(3,3), E(3,3), C(3), X(3)

```

Важное значение имеет методика составления матрицы инцидентий, рассмотрение которой проведем на примере схемы, изображенной на рис. 2.1.

Вначале следует выбрать положительные направления токов в ветвях, выделить независимые контуры и установить направления их обхода (задать положительное направление контурных токов). На схеме принятые направления изображены стрелками. Матрица инцидентий заполняется по следующему правилу: каждый ее элемент S_{ij} , соответствующий i -й строке (i -му контуру) и j -му столбцу (j -й ветви), равен:

$$S_{ij} = \begin{cases} 0, & \text{если } j\text{-я ветвь не входит в } i\text{-й контур;} \\ 1, & \text{если направление тока в ветви совпадает с направлением} \\ & \text{тока в контуре;} \\ -1, & \text{если эти токи противоположны.} \end{cases}$$

Матрица инцидентий рассматриваемого примера соответствует табл. 2.1.

Таблица 2.1

№ кон- ту- ра	№ ветви					
	1	2	3	4	5	6
I	1	-1	-1	0	0	0
II	0	1	0	1	0	-1
III	0	0	1	-1	1	0

Вторая строка ее, например, отражает тот факт, что во второй контур не входят ветви 1, 3, 5, ток в ветвях 2 и 4 совпадает с принятым направлением контурного тока, а в шестой ветви имеет противоположное направление.

Составленную таким образом матрицу инцидентий можно при расчете ввести в процессе диалогового взаимодействия, однако и можно записать ее в оператор DATA и заполнять массив S во время вычислений. Последнее имеет свои преимущества. Опишем этот вариант:

```
1110 DATA 1,-1,-1,0,0,0
1120 DATA 0,1,0,1,0,-1
1130 DATA 0,0,1,-1,1,0
1140 MAT READ S
```

Следующим шагом является подготовка диагональной матрицы – матрицы сопротивлений ветвей. Она в нашем случае имеет размерность 6 x 6, диагональные элементы ее равны сопротивлениям ветвей. Последние удобнее вводить оператором READ в виде вектора и саму матрицу сформировать во время проведения расчетов:

```
1040 MAT READ Z
1050 MAT R = ZER
1060 FOR I = 1 TO 6
1070 LET R(I,I) = Z(I)
1080 NEXT I
```

Необходимо также задать компоненты вектора ЭДС ветвей, для чего достаточно двух строк:

```
1100 MAT U = ZER
1110 LET U(6) = 10
```

так как источник ЭДС имеется только в шестой ветви.

Объединив все части в одну программу и добавив простой вывод результатов, получим приведенный ниже полный текст программы МКТО.



```
900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * КОНТУРНЫХ ТОКОВ. ПОСТОЯННЫЙ ТОК. *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ БЕЙСИКА. *
908 REM * НАЗВАНИЕ: "МКТО". *
910 REM *****
1000 DIM R[6,6],U[6],Z[6]
1010 DIM S[3,6],T[6,3],G[3,6]
1020 DIM A[3,3],E[3,3],C[3],X[3]
1030 DATA 6,2,1,5,4,3
1040 MAT READ Z
1050 MAT R=ZER
1060 FOR I=1 TO 6
1070 LET R[I,I]=Z[I]
1080 NEXT I
1090 MAT U=ZER
1100 LET U[6]=10
1110 DATA 1,-1,-1,0,0,0
1120 DATA 0,1,0,1,0,-1
1130 DATA 0,0,1,-1,1,0
1140 MAT READ S
1150 MAT T=TRN(S)
1160 MAT G=S*R
1170 MAT A=G*T
1180 MAT C=S*U
1190 MAT E=INV(A)
1200 MAT X=E*C
1210 MAT Z=T*X
1220 PRINT "ТОКИ В ВЕТВЯХ"
1230 MAT PRINT Z
1240 END
```

Остается произвести расчет. Запустим программу и по окончании вычислений получим:

```
ТОКИ В ВЕТВЯХ
-.413223
-1.05785
-.363636
-.694215
-.776859
1.47107
```

▼ Недостатком составленной программы является необходимость подготовки вручную матрицы инцидентий, что для сложных схем представляется достаточно трудоемкой процедурой. Трудности возрастают с ростом размерности схем (а на практике обычными являются задачи с числом узлов более 100), что приводит к увеличению числа ошибок. Определенные сложности возникают и при расчете непланарных схем, которым соответствуют непланарные графы (т.е. графы, которые не размещаются на плоскости без самопересечений).

В связи с этим разрабатываются алгоритмы и программы для автоматизации составления матриц инцидентий. Решение этой проблемы, описанное ниже, накладывает некоторые ограничения на свойства графов, задающих топологию анализируемых цепей: такой граф должен быть связным (т.е. в нем всегда должен существовать путь между двумя произвольными его вершинами, и у него должны отсутствовать висячие вершины (степень вершины не должна быть меньше 2)). Можно, видимо, полагать, что встречающиеся в практических расчетах электрические цепи удовлетворяют этим требованиям и принятие этих ограничений не уменьшает общность изложенного далее алгоритма.

Процесс автоматизированного построения матрицы инцидентий состоит из двух этапов — построение дерева и нахождение независимых (главных) контуров в схеме.

Дерево — это система ветвей, связывающих все узлы (вершины графа) без образования контуров [29]. При его построении сначала выбирается произвольный узел (корень дерева) и заносится в список вершин. Затем находится очередной узел, находящийся от первого на расстоянии не более одной ветви, и также заносится в список. Если таких узлов несколько, то их выбор произволен. Следующий шаг повторяет предыдущую процедуру, если еще имеются узлы, не занесенные в список и являющиеся смежными с последним по счету узлом (из занесенных в список). Остановка этого процесса составления списка не означает завершения построения дерева, соответствующего данному графу, поскольку в списке могут отсутствовать некоторые узлы. Если такие узлы имеются, то описанная процедура повторяется. Признаком завершения процедуры является включение всех узлов в дерево.

Второй этап — нахождение независимых контуров — производится по следующему алгоритму: выбирается произвольная не вошедшая в дерево хорда и выбирается путь, связывающий узлы выбранной хорды и проходящий целиком по дереву. Для этого берется одна из вершин хорды и перебираются узлы дерева в сторону его корня. После этого берется вторая вершина хорды, и также фиксируется путь от нее к корню дерева. Из объединения полученных двух путей и хорды, а затем последующего исключения повторяющихся ребер и формируется контур схемы. Процесс этот реализуется для каждой следующей хорды и завершается после перебора всех хорд. В результате получается совокупность независимых контуров данной схемы.

Выбор положительного направления токов в ветвях определяется в соответствии со следующим правилом. Поскольку при вводе данных с помощью ПП 2000 (см. гл. 1) ветвь задается номерами двух узлов N1 и N2, то за положительное направление тока принимается направление от N1 к N2. А направление контурных токов однозначно определяется описанной выше процедурой построения самого контура.

После выделения контуров и направлений токов заполнение матрицы инцидентий производится по обычной методике.

Разработанная по описанному алгоритму программа не требует дополнительной информации и дополнительной оперативной памяти, кроме той, что выделена под массив, где хранятся номера смежных пар узлов. Она оптимизирована по быстродействию. Текст ее оформлен в виде подпрограммы ПП 3000.



```

290 2 REM *****
290 4 REM * ПП ФОРМИРОВАНИЯ МАТРИЦЫ ИНЦИДЕНЦИЙ *
290 6 REM * ДЛЯ РАСЧЕТА ПО МЕТОДУ КОНТУРНЫХ ТОКОВ. *
290 8 REM *****
3000 DEF FNA(A)=INT(ABS(A))
3002 DEF FNB(B)=INT((ABS(B)-FNA(B))*NO+.5)
3004 DEF FNC(C)=FNA(C)*SGN(C)
3006 DEF FND(D)=SGN(SGN(D)+.5)
3100 LET N3=N1-N2
3102 FOR K=1 TO N3
3104 FOR K1=1 TO N1
3106 LET S[K,K1]=0
3108 NEXT K1
3110 NEXT K
3112 FOR K=1 TO N1
3114 LET K1=FNA(P[K,1])
3116 LET S[1,K1+1]=S[1,K1+1]+1
3118 LET K1=FNB(P[K,1])
3120 LET S[1,K1+1]=S[1,K1+1]+1
3122 NEXT K
3124 LET K1=0
3126 LET K2=FNA(P[1,1])
3128 GOTO 3178
3130 FOR K=1 TO N1
3132 LET K3=P[K,1]
3134 IF K3<0 THEN 3142
3136 IF (FNA(K3)-K2)*(FNB(K3)-K2)*0 THEN 3142
3138 LET K4=FNA(K3)+FNB(K3)-K2

```

```

3140 IF FNC(S[1,K4+1])>0 THEN 3168
3142 NEXT K
3144 LET S[1,K2]=FNB(S[1,K2])/NO
3146 FOR K=1 TO K1
3148 LET K3=FNB(S[1,K])
3150 LET K2=FNA(P[K3,1])
3152 IF FNC(S[1,K2+1])<0 THEN 3164
3154 LET K2=FNB(P[K3,1])
3156 IF FNC(S[1,K2+1])<0 THEN 3164
3158 NEXT K
3160 PRINT "СХЕМА ММЕЕТ НЕСВЯЗНЫЕ ГРАФ."
3162 STOP
3164 LET S[2,K1]=(K+1)/NO
3166 GOTO 3178
3168 LET P[K,1]=-P[K,1]
3170 LET K1=K1+1
3172 LET K3=S[1,K1]
3174 LET S[1,K1]=(FNA(K3)+K/NO)*FND(K3)
3176 LET K2=K4
3178 LET S[1,K2+1]=1-ABS(S[1,K2+1])
3180 IF K1<N2 THEN 3130
3182 FOR K=1 TO N2+1
3184 LET S[1,K]=FNB(S[1,K])/NO
3186 NEXT K
3188 LET K1=1
3190 FOR K2=1 TO N3
3192 FOR K=K1 TO N1
3194 IF P[K,1]>0 THEN 3198
3196 NEXT K
3198 LET K1=K+1
3200 LET K3=P[K,1]
3202 LET K4=FNA(K3)
3204 FOR K5=1 TO 2
3206 IF K4=FNA(P[1,1]) THEN 3246
3208 FOR K6=1 TO N2
3210 LET K7=FNB(S[1,K6])
3212 LET K8=P[K7,1]
3214 IF (K4-FNA(K8))*(K4-FNB(K8))=0 THEN 3220
3216 NEXT K6
3218 GOTO 3246
3220 LET KO=FND(-ABS(K4-FNA(K8)))*SGN(K5-1.5)

```

```

3222 LET K4=FNA(K8)+FNB(K8)-K4
3224 LET K9=S[K2,K7]
3226 LET KO=FNC(K9)+KO
3228 LET S[K2,K7]=KO+FNB(K9)*FND(KO)/NO
3230 LET K6=K6-1
3232 IF K6=0 THEN 3246
3234 IF N3=1 THEN 3240
3236 LET K6=K6+(FNB(S[2,K6])-K6-1)*SGN(FNB(S[2,K6]))
3238 IF K6=0 THEN 3246
3240 LET K7=FNB(S[1,K6])
3242 LET K8=P[K7,1]
3244 GOTO 3220
3246 LET K4=FNB(K3)
3248 NEXT K5
3250 LET S[K2,K]=1+FNB(S[K2,K])/NO
3252 NEXT K2
3254 FOR K=1 TO N1
3256 LET P[K,1]=ABS(P[K,1])
3258 LET S[1,K]=FNC(S[1,K])
3260 IF N3=1 THEN 3264
3262 LET S[2,K]=FNC(S[2,K])
3264 NEXT K
3266 RETURN

```

Входной информацией для ПП 3000 является массив параметров ветвей P , подготовливаемый в ПП 2000, а также число ветвей (в программе оно обозначено $N1$), номер наибольшего узла схемы ($N2$) и упаковочный параметр ($N0$). О структуре массива P и назначении прочих упомянутых переменных см. описание ПП 2000 (гл. 1).

По окончании работы ПП 3000 в двумерном массиве S размерностью $N3 \times N1$ располагается матрица инцидентий. (Отметим, что $N3$ число контуров — определяется в процессе счета.)

Если граф рассматриваемой схемы является несвязным, то выдается соответствующее сообщение и вычисления прекращаются.

▼ Рассматривая программу МКТО, можно заметить еще один путь ее усовершенствования: заменить ввод данных с помощью операторов READ-DATA на диалоговый режим ввода, что обеспечивает большую ее гибкость и универсальность. Такой ввод можно организовать, используя ПП 2000. Однако для этого нужно внести соответствующие изменения в текст МКТО.

Вместо строк 1030–1140 надо написать

```
1030 GOSUB 2000
1040 GOSUB 2500
1050 GOSUB 3000
1110 MAT R = ZER
1120 FOR K = 1 TO N1
1130 LET R(K,K) = P(K,2)
1140 NEXT K
```

Строка 1140 содержит обращение к ПП 2500, в которой вводятся ЭДС ветвей. А строки 1110–1140 полностью аналогичны по содержанию строкам 1050–1080 программы МКТО, так как во втором столбце массива Р записываются активные сопротивления ветвей (см. описание ПП 2000). После внесения этих изменений получаем программу МКТ1, полный текст которой приведен ниже. Для проведения расчетов нужно загрузить в ЭВМ с машинных носителей (или ввести с клавиатуры) подпрограммы автоматического составления матриц инцидентов (ПП 3000) и ввода данных (ПП 2000).

```
900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * КОНТУРНЫХ ТОКОВ. ПОСТОЯННЫЙ ТОК . *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ ВЭИСКА . *
908 REM * НАЗВАНИЕ: "МКТ1". *
910 REM *****
1000 DIM P[6,2],R[6,6],U[6],Z[6]
1010 DIM S[3,6],T[6,3],G[3,6]
1020 DIM A[3,3],E[3,3],C[3],X[3]
1021 REM-----
1022 REM ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ
1023 REM И ТОПОЛОГИИ СХЕМЫ
1024 REM-----
1030 GOSUB 2000
1031 REM-----
1032 REM ВВОД ИСТОЧНИКОВ ЭДС ВЕТВЕЙ
1033 REM-----
1040 GOSUB 2500
1041 REM-----
1042 REM ФОРМИРОВАНИЕ МАТРИЦ ИНЦИДЕНЦИЙ
1043 REM-----
1100 GOSUB 3000
1101 REM-----
```



```

1102 REM ФОРМИРОВАНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1103 REM ВЕТВЕЙ
1104 REM-----
1110 MAT R=ZER
1120 FOR K=1 TO N1
1130 LET R[K,K]=P[K,2]
1140 NEXT K
1141 REM-----
1142 REM ВЫЧИСЛЕНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1143 REM КОНТУРОВ
1144 REM-----
1150 MAT T=TRN(S)
1160 MAT G=S*R
1170 MAT A=G*T
1171 REM-----
1172 REM ВЫЧИСЛЕНИЕ ВЕКТОРА
1173 REM Э Д С КОНТУРОВ
1174 REM-----
1180 MAT C=S*U
1181 REM-----
1182 REM ОБРАЩЕНИЕ МАТРИЦЫ
1183 REM СОПРОТИВЛЕНИЙ КОНТУРОВ
1184 REM-----
1190 MAT E=INV(A)
1191 REM-----
1192 REM ВЫЧИСЛЕНИЕ КОНТУРНЫХ ТОКОВ
1193 REM-----
1200 MAT X=E*C
1201 REM-----
1202 REM ВЫЧИСЛЕНИЕ ТОКОВ В ВЕТВЯХ
1203 REM-----
1210 MAT Z=T*X
1211 REM-----
1212 REM ВЫВОД РЕЗУЛЬТАТОВ
1213 REM-----
1500 GOSUB 1700
1510 STOP
1670 REM *****
1680 REM * ПП ПЕЧАТИ ЗНАЧЕНИЙ ТОКОВ В ВЕТВЯХ *
1690 REM *****

```

```

1700 PRINT "ТОКИ В ВЕТВЯХ : "
1710 MAT PRINT Z
1720 RETURN
2460 REM *****
2470 REM * ВВОД ИСТОЧНИКОВ ПОСТОЯННОГО *
2480 REM * НАПРЯЖЕНИЯ ВЕТВЕЙ *
2490 REM *****
2500 MAT U=ZER
2510 LET U[6]=10
2520 RETURN
9999 END

```

READY

В качестве примера возьмем схему, изображенную на рис. 2.1. Отметим, что в ветвях, содержащих источники ЭДС (в данном случае — это ветвь 6), внутреннее сопротивление источника ЭДС должно быть объединено с активным сопротивлением резистора, имеющегося в данной ветви. В дальнейшем будем полагать, что эта процедура всегда уже выполнена. Пронумеруем ветви и узлы и составим табл. 2.2.

Таблица 2.2

№ ветви	№ узлов	R, Ом	E, В	№ ветви	№ узлов	R, Ом	E, В
1	0 2	6	0	4	1 3	5	0
2	0 1	2	0	5	2 3	4	0
3	1 2	1	0	6	0 3	3	10

Запускаем программу на выполнение, выбрав режим № 2, так как расчет ведется на постоянном токе. Далее, отвечая на последовательные запросы программы, вводим в соответствии с табл. 2.2 следующие данные:

```

0,2,6
0,1,2
1,2,1
1,3,5
2,3,4
0,3,3
-2,0,0

```

Последняя строка содержит признак окончания ввода данных (см. описание ПП 2000 в предыдущей главе). После этого ЭВМ выполняет расчет и выводит значения токов в ветвях, приведенные на с. 24.

▼ Хотя матричные операторы придают программным текстам краткость и наглядность, однако они обладают двумя недостатками. Во-первых, при их использовании следует строго следить за соответствием размерностей, например, перемножаемых матриц. Во-вторых, требуется описывать и выделять память под дополнительные массивы для хранения промежуточных результатов – тоже матриц. Поэтому программы с матричными операторами обладают недостаточной гибкостью, универсальностью и эффективностью. Наконец, такие операторы имеются не во всех версиях языка БЕЙСИК. В связи с этим возникает необходимость дать другой вариант программы расчета электрической схемы – без использования матричных операторов.

В новой программе (МКТ2) сохраним обозначения и последовательность отдельных этапов расчета, использованные в программе МКТ1, сократив благодаря этому объем дополнительных замечаний и пояснений. В качестве примера снова рассмотрим расчет схемы на рис. 2.1 на постоянном токе.

Просматривая текст программы МКТ1, будем заменять действия с матричными операторами соответствующими фрагментами, которые выполняют эквивалентные операции. В ряде случаев, как это будет отмечено далее, открываются некоторые возможности усовершенствовать текст программы в целях повышения ее эффективности.

Ввод данных и формирование матрицы инцидентий остаются без изменений.

Определение элементов матрицы сопротивлений контуров A , которая находится по формуле (2.2), можно осуществить более экономным способом по сравнению с программой МКТ1, если учесть, что R – это диагональная матрица:

$$A_{kl} = \sum_{i=1}^{N1} S_{ki} S_{li} R_i, \quad (2.7)$$


где S_{ki} и S_{li} – элементы матрицы инцидентий; R_i – сопротивление i -й ветви.

Эти вычисления оформим в виде сегмента с номерами строк 1100–1170 (см. с. 33), а нахождение вектора контурных ЭДС без матричных операторов выполним строками 1180–1220. Для решения уравнения (2.4) нужно обратить матрицу A с помощью ПП 5200 (1222–1280), после завершения этой процедуры обращенная матрица будет на месте исходной (более подробно см. в гл. 5). Теперь осталось получить вектор контурных токов (1292–1350), найти токи в ветвях (1352–1410) и вывести результаты (1670–1740). После объединения этих фрагментов с ПП 2000, ПП 3000 и ПП 5200 получаем программу МКТ2 для расчета электрических цепей методом контурных токов.

```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * КОНТУРНЫХ ТОКОВ. ПОСТОЯННЫЙ ТОК . *
906 REM * ПРИМЕНЕНИЕ МЕТОДА ОБМЕНА ДЛЯ *
908 REM * ОБРАЩЕНИЯ МАТРИЦЫ СОПРОТИВЛЕНИЙ *
910 REM * КОНТУРОВ. НАЗВАНИЕ: "МКТ2". *
912 REM *****
1000 DIM P[6,2],U[6],Z[6],S[3,6],A[3,4],C[3],X[3]
1001 REM-----
1002 REM ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ
1003 REM И ТОПОЛОГИИ СХЕМЫ
1004 REM-----
1010 GOSUB 2000
1011 REM-----
1012 REM ВВОД ИСТОЧНИКОВ Э.Д.С. ВЕТВЕЙ
1013 REM-----
1020 GOSUB 2500
1021 REM-----
1022 REM ФОРМИРОВАНИЕ МАТРИЦЫ ИНЦИДЕНЦИЙ
1023 REM-----
1100 GOSUB 3000
1101 REM-----
1102 REM ВЫЧИСЛЕНИЕ КОЭФФИЦИЕНТОВ МАТРИЦЫ
1103 REM СОПРОТИВЛЕНИЙ И ВЕКТОРА Э.Д.С.
1104 REM КОНТУРОВ
1105 REM-----
1110 FOR I=1 TO N3
1120 FOR J=1 TO N3
1130 LET A[I,J]=0
1140 FOR K=1 TO N1
1150 LET A[I,J]=A[I,J]+S[I,K]*S[J,K]*P[K,2]
1160 NEXT K
1170 NEXT J
1180 LET C[I]=0
1190 FOR K=1 TO N1
1200 LET C[I]=C[I]+S[I,K]*U[K]
1210 NEXT K
1220 NEXT I
1221 REM-----
1222 REM ОБРАЩЕНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1223 REM КОНТУРОВ

```



```

1224 REM-----
1230 LET A1=N3
1240 LET A3=1.00000E-10
1250 GOSUB 5200
1260 IF A4>0 THEN 1300
1270 PRINT "МАТРИЦА ВЫРОЖДЕНА,"
1280 PRINT "ВЫЧИСЛЕНИЯ ПРЕКРАЩАЮТСЯ."
1290 STOP
1291 REM-----
1292 REM ВЫЧИСЛЕНИЕ КОНТУРНЫХ ТОКОВ
1293 REM-----
1300 FOR I=1 TO N3
1310 LET X[I]=0
1320 FOR J=1 TO N3
1330 LET X[I]=X[I]+A[I,J]*C[J]
1340 NEXT J
1350 NEXT I
1351 REM-----
1352 REM ВЫЧИСЛЕНИЕ ТОКОВ В ВЕТВЯХ
1353 REM-----
1360 FOR K=1 TO N1
1370 LET Z[K]=0
1380 FOR I=1 TO N3
1390 LET Z[K]=Z[K]+S[I,K]*X[I]
1400 NEXT I
1410 NEXT K
1411 REM-----
1412 REM ВЫВОД РЕЗУЛЬТАТОВ
1413 REM-----
1500 GOSUB 1700
1600 STOP
1670 REM *****
1680 REM * ПП ПЕЧАТИ ЗНАЧЕНИЙ ТОКОВ В ВЕТВЯХ *
1690 REM *****
1700 PRINT "ТОКИ В ВЕТВЯХ."
1710 PRINT "N ВЕТВИ", "I"
1720 FOR K=1 TO N1
1730 PRINT K, Z[K]
1740 NEXT K
1750 RETURN
2460 REM *****

```

```

2470 REM * ВВОД ИСТОЧНИКОВ ПОСТОЯННОГО *
2480 REM * НАПРЯЖЕНИЯ ВЕТВЕЙ *
2490 REM *****
2500 FOR K=1 TO N1
2510 LET U[K]=0
2520 NEXT K
2530 LET U[6]=10
2540 RETURN
9999 END

```

READY

Однако необходимо еще описать использованные массивы P, A, X, S, U, Z. Поскольку в программе отсутствуют матричные операторы, пользователь вправе отказаться от жестких ограничений на размерность массивов. Их можно, например, описать следующим образом:

```
1000 DIM P(K, 2), U(K), Z(K), S(L, K), A(L, L + 1), C(L), X(L)
```

причем K больше числа ветвей или равно ему, L равно числу независимых контуров.

Как уже отмечалось ранее, в этой строке вместо K и L надо подставить конкретные значения. После загрузки программы запустим ее на выполнение и проведем расчет для схемы, изображенной на рис. 2.1. Результаты расчета совпадают с приведенными на с. 24.

▼ Перейдем к расчету цепей на переменном токе с использованием метода контурных токов и снова приведем два варианта программы — с использованием матричных операторов (программа MKT3) и без них (MKT4). Методы и алгоритмы расчета аналогичны описанным ранее для постоянного тока. Отличие состоит в том, что все вычисления ведутся над комплексными числами, поэтому программы содержат дополнительные блоки, а длина их текста увеличивается. Усложняется также ввод исходных данных, так как каждая ветвь в общем случае может содержать активное сопротивление, индуктивность и емкость. Введенные параметры нужно пересчитать с учетом круговой частоты на комплексные сопротивления, для чего нужна специальная программа. Изменяется и ПП печати результатов.

Для подсчета комплексных сопротивлений необходимо выполнять действия с комплексными числами, для которых в большинстве известных версий языка БЕЙСИК не предусмотрены специальные функции, поэтому необходимо составлять ПП, реализующие сложение, умножение и деление таких величин. Кроме того, нужны ПП определения модуля комплексного числа и угла между действительной и мнимой его частями (аргумент комплексного числа). Эти ПП приведены в гл. 5.

Все матрицы при расчете на переменном токе (за исключением матрицы инциденций) должны быть комплексными – содержать действительные и мнимые части (т.е. действительные и мнимые матрицы). Соответственно в (2.1) и (2.6) матрицу R и векторы U, X, Z следует дополнить мнимыми членами и записать в виде $R + jQ, U + jV, X + jY, Z + jW$.

Соотношения (2.2) и (2.3) приобретают вид

$$A + jB = S(R + jQ)S^T; \quad (2.8)$$

$$C + jD = S(U + jV). \quad (2.9)$$

Комплексные токи в контурах

$$X + jY = (A + jB)^{-1}(C + jD), \quad (2.10)$$

а токи в ветвях находим по

$$Z + jW = S^T(X + jY). \quad (2.11)$$

Так как в большинстве версий языка БЕЙСИК нет матричного оператора обращения комплексной матрицы, эту процедуру надо запрограммировать. Чтобы получить комплексную матрицу $E + jF$, обратную матрице $A + jB$, следует воспользоваться одной из двух пар формул:

1) при $\det(A) \neq 0$

$$E = (A + BA^{-1}B)^{-1}; \quad F = -A^{-1}BE;$$

2) при $\det(B) \neq 0$

$$F = -(AB^{-1}A + B)^{-1}; \quad E = -B^{-1}AF,$$

где $\det(A)$ и $\det(B)$ – определители матриц A и B .

С учетом сказанного и приведенных выше обозначений метод контурных токов для расчета электрической цепи, изображенной на рис. 2.1, на переменном токе может быть представлен программой МКТЗ.



```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * КОНТУРНЫХ ТОКОВ. ПЕРЕМЕННЫЙ ТОК . *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ БЕЙСИКА . *
908 REM * НАЗВАНИЕ: "МКТЗ" . *
910 REM *****
1000 DIM P[6,4],R[6,6],Q[6,6],U[6],V[6]
1010 DIM Z[6],W[6],S[3,6],T[6,3],G[3,6]
1020 DIM A[3,3],B[3,3],E[3,3],F[3,3]
1030 DIM X[3],Y[3],H[3],C[3],D[3]

```

```

1031 REM-----
1032 REM ВВОД ТАБЛИЦ ЭЛЕМЕНТОВ ВЕТВЕЙ
1033 REM И ТОПОЛОГИИ СХЕМЫ
1034 REM-----
1040 GOSUB 2000
1041 REM-----
1042 REM ЗАДАНИЕ ЧАСТОТЫ
1043 REM-----
1050 PRINT "F(Гц)=";
1060 INPUT F
1070 LET W=2*3.14159*F
1071 REM-----
1072 REM ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО
1073 REM НАПРЯЖЕНИЯ ВЕТВЕЙ ( ЧАСТОТЫ F )
1074 REM-----
1080 GOSUB 2500
1081 REM-----
1082 REM ФОРМИРОВАНИЕ МАТРИЦЫ ИНЦИДЕНЦИЙ
1083 REM-----
1100 GOSUB 3000
1101 REM-----
1102 REM ФОРМИРОВАНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1103 REM ВЕТВЕЙ
1104 REM-----
1110 MAT R=ZER
1120 MAT Q=ZER
1130 FOR K=1 TO N1
1140 GOSUB 1800
1150 LET R[K,K]=R
1160 LET Q[K,K]=Q
1170 NEXT K
1171 REM-----
1172 REM ВЫЧИСЛЕНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1173 REM И ВЕКТОРА Э Д С МОНТУРОВ
1174 REM-----
1180 MAT T=TRN(S)
1190 MAT G=S*R
1200 MAT A=G*T
1210 MAT C=S*U
1220 MAT G=S*Q
1230 MAT B=G*T

```

```

1240 MAT D=S*V
1241 REM-----
1242 REM ОБРАЩЕНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1243 REM КОНТУРОВ ( ДЕТ(A)≠0 )
1244 REM-----
1250 MAT E=INV(A)
1260 MAT F=E*B
1270 MAT E=B*F
1280 MAT B=A+E
1290 MAT E=INV(B)
1300 MAT B=F*E
1310 MAT F=(-1)*B
1311 REM-----
1312 REM ВЫЧИСЛЕНИЕ КОНТУРНЫХ ТОКОВ
1313 REM-----
1320 MAT X=E*C
1330 MAT H=F*D
1340 MAT X=X-H
1350 MAT Y=E*D
1360 MAT H=F*C
1370 MAT Y=Y+H
1371 REM-----
1372 REM ВЫЧИСЛЕНИЕ ТОКОВ В ВЕТВЯХ
1373 REM-----
1380 MAT Z=T*X
1390 MAT W=T*Y
1391 REM-----
1392 REM ВЫВОД РЕЗУЛЬТАТОВ
1393 REM-----
1500 GOSUB 1700
1510 STOP
1670 REM *****
1680 REM * ПЕЧАТЬ ЗНАЧЕНИЙ ТОКОВ В ВЕТВЯХ *
1690 REM *****
1700 PRINT "ТОКИ В ВЕТВЯХ."
1710 PRINT "N ВЕТВИ", "RE(I)", "IM(I)"
1720 FOR K=1 TO N1
1730 PRINT K, Z[K], W[K]
1740 NEXT K
1750 RETURN
1790 REM *****

```

```

1791 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОГО СОПРОТИВЛЕНИЯ *
1792 REM * К-ТОЙ ВЕТВИ ( R+j*Q ) *
1793 REM *****
1800 LET R=P[K,2]
1810 LET Q=W*P[K,3]
1820 IF P[K,4]=0 THEN 1840
1830 LET Q=Q-1/W/P[K,4]
1840 RETURN
2460 REM *****
2470 REM * ВВОД ИСТОЧНИКОВ ПЕРЕМЕННОГО *
2480 REM * НАПРЯЖЕНИЯ ВЕТВЕЙ *
2490 REM *****
2500 MAT U=ZER
2510 MAT V=ZER
2520 LET U[6]=10
2530 RETURN
9999 END

```

Комплексное сопротивление k -й ветви при заданной круговой частоте вычисляется в отдельной ПП 1800, вызываемой при заполнении диагональных элементов матрицы $R + jQ$ (строки 1110–1170).

Описание размерностей массивов требует особого внимания. В строках 1000–1030 указываются нужные величины:

```

1000 DIM P(K, 4), R(K, K), Q(K, K), U(K, K), V(K, K)
1010 DIM Z(K), W(K), S(L, K), T(K, L), G(L, K)
1020 DIM A(L, L), B(L, L), E(L, L), F(L, L)
1030 DIM X(L), Y(L), H(L), C(L), D(L)

```

где K — число ветвей, а L — число независимых контуров, которые следует заранее подсчитать.

При переходе к расчету иной схемы следует определить размерности массивов и заново ввести строки с описанием последних, а также строки 2500–2530, где организуется ввод вектора ЭДС.

В строках 1050–1070 вводится частота источников синусоидального напряжения ветвей (в герцах) и определяется круговая частота. Алгоритм программы составлен в предположении, что все эти источники имеют одинаковую частоту. (Для расчета электрической цепи, содержащей в отдельных своих ветвях источники с разными частотами, следует использовать принцип суперпозиции, проведя расчеты в отдельности для каждой частоты и затем просуммировав результаты.) Далее вызывается ПП 2500, которая формирует вектор источников ЭДС, ввод последних

производится в виде отдельных действительной и мнимой частей (векторы U и V).

Рассчитаем схему, топологически эквивалентную показанной на рис. 2.1, но имеющую в каждой ветви R , L , C . Сначала в качестве тестовой задачи произведем с помощью программы МКТ3 расчет на постоянном токе, задав в качестве исходных данных нулевые индуктивности и емкости, а R — в соответствии с табл. 2.2. Частоту введем также равной 0, напряжение источника ЭДС в ветви 6 принимаем равным 10 В. Запустив программу МКТ3 на выполнение, выберем режим работы 3 (см. описание ПП 2000 в гл. 1) и введем данные. Можно убедиться, что полученные данные совпадают с приведенными на с. 24. Другие способы тестирования описаны в § 2.4.

▼ Перейдем к описанию программы МКТ4 — программы расчета электрических цепей на переменном токе по методу КТ без применения матричных операторов языка БЕЙСИК. Как и в случае перехода от МКТ1 к МКТ2, она получена путем последовательной замены всех матричных операторов МКТ3 соответствующими программными фрагментами. Однако в целях повышения эффективности программы решения системы линейных уравнений в отличие от МКТ3 определяется непосредственно методом Гаусса.

Ввод данных и топологии, формирование матрицы инцидентий в автоматическом режиме остаются без изменений (применяются те же ПП 2000 и ПП 3000). Не изменяются также фрагменты ввода частоты и параметров источников ЭДС ветвей.

Для решения системы уравнений применяется ПП 5800 (которая описана в гл. 5), в соответствии с особенностями ее построения входные и выходные данные (коэффициенты матрицы, свободные члены и решение) располагаются в массивах A (действительные компоненты) и B (мнимые).

Применение метода Гаусса обосновано тем, что он позволяет отказаться от обращения матрицы системы уравнений и последующего умножения этой матрицы на столбец свободных членов. Это приводит к сокращению времени счета и повышению точности получаемых результатов.



```
900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * КОНТУРНЫХ ТОКОВ. ПЕРЕМЕННЫЙ ТОК. *
906 REM * ВЫЧИСЛЕНИЕ КОНТУРНЫХ ТОКОВ *
908 REM * МЕТОДОМ ГАУССА. НАЗВАНИЕ: 'МКТ4'. *
910 REM *****
1000 DIM P[6,4],U[6],V[6],Z[6],W[6],A[3,4],B[3,4],S[3,6]
1001 REM-----
1002 REM ВВОД ТАБЛИЦ ЭЛЕМЕНТОВ ВЕТВЕЙ
1003 REM И ТОПОЛОГИИ СХЕМЫ
```

```

1004 REM-----
1010 GOSUB 2000
1011 REM-----
1012 REM ЗАДАНИЕ ЧАСТОТЫ
1014 REM-----
1020 PRINT "F(Гц)=";
1030 INPUT F
1040 LET W=2*3.14159*F
1041 REM-----
1042 REM ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО
1043 REM НАПРЯЖЕНИЯ ВЕТВЕЙ
1044 REM-----
1050 GOSUB 2500
1051 REM-----
1052 REM ФОРМИРОВАНИЕ МАТРИЦЫ ИНЦИДЕНЦИЙ
1053 REM-----
1100 GOSUB 3000
1101 REM-----
1102 REM ФОРМИРОВАНИЕ МАТРИЦЫ СОПРОТИВЛЕНИЙ
1103 REM И ВЕКТОРА Э Д С КОНТУРОВ
1104 REM-----
1110 LET N4=N3+1
1120 FOR K=1 TO N1
1130 GOSUB 1800
1140 FOR I=1 TO N3
1150 FOR J=1 TO N3
1160 IF K>1 THEN 1190
1170 LET A[I,J]=0
1180 LET B[I,J]=0
1190 LET S=S[I,K]*S[J,K]
1200 LET A[I, ]=A[I,J]+S*R
1210 LET B[I,J]=B[I,J]+S*Q
1220 NEXT J
1230 IF K>1 THEN 1260
1240 LET A[I,N4]=0
1250 LET B[I,N4]=0
1260 LET A[I,N4]=A[I,N4]+S[I,K]*U[K]
1270 LET B[I,N4]=B[I,N4]+S[I,K]*V[K]
1280 NEXT I
1290 NEXT K
1291 REM-----

```

```

1292 REM ВЫЧИСЛЕНИЕ КОНТУРНЫХ ТОКОВ
1293 REM-----
1300 LET A1=N3
1310 LET A2=1
1320 LET A3=1.00000E-10
1330 GOSUB 5800
1340 IF A4>0 THEN 1380
1350 PRINT "МАТРИЦА СИСТЕМЫ ВЫРОЖДЕНА,"
1360 PRINT "ВЫЧИСЛЕНИЯ ПРЕКРАЩАЮТСЯ."
1370 STOP
1371 REM-----
1372 REM ВЫЧИСЛЕНИЕ ТОКОВ В ВЕТВЯХ
1373 REM-----
1380 FOR K=1 TO N1
1390 LET Z[K]=0
1400 LET W[K]=0
1410 FOR I=1 TO N3
1420 LET Z[K]=Z[K]+S[I,K]*A[I,N4]
1430 LET W[K]=W[K]+S[I,K]*B[I,N4]
1440 NEXT I
1450 NEXT K
1451 REM-----
1452 REM ВЫВОД РЕЗУЛЬТАТОВ
1453 REM-----
1500 GOSUB 1700
1510 STOP
1670 REM *****
1680 REM * ПП ПЕЧАТИ ЗНАЧЕНИЙ ТОКОВ В ВЕТВЯХ *
1690 REM *****
1700 PRINT "ТОКИ В ВЕТВЯХ."
1710 PRINT "N ВЕТВИ", "RE(I)", "IM(I)"
1720 FOR K=1 TO N1
1730 PRINT K, Z[K], W[K]
1740 NEXT K
1750 RETURN
1790 REM *****
1791 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОГО СОПРОТИВЛЕНИЯ *
1792 REM * K-й ВЕТВИ ( R+J*Q ) *
1793 REM *****
1800 LET R=P[K,2]
1810 LET Q=W*P[K,3]

```

```

1820 IF P[K,4]=0 THEN 1840
1830 LET Q=Q-1/W/P[K,4]
1840 RETURN
2460 REM *****
2470 REM * ВВОД ИСТОЧНИКОВ ПЕРЕМЕННОГО *
2480 REM * НАПРЯЖЕНИЯ ВЕТВЕЙ *
2490 REM *****
2500 FOR K=1 TO N1
2510 LET U[K]=0
2520 LET V[K]=0
2530 NEXT K
2540 LET U[6]=10
2550 RETURN
9999 END

```

Остановимся на описании массивов. В строке

```
1000 DIM P(K, 4), U(K), V(K), Z(K), W(K), A(L, L + 1), B(L, L + 1), S(L, K)
```

по прежнему K – число ветвей схемы, L – число независимых контуров схемы.


В МКТ4 (как и в МКТ2) в отличие от программ с матричными операторами допускается резервирование массивов несколько большей размерности, чем это необходимо для проведения конкретного расчета. Это позволяет расширить область применения этих программ, так как теперь нет необходимости для каждой конкретной анализируемой цепи переделывать описание размерностей массивов.

Протестировать программу МКТ4 рекомендуется аналогично программе МКТ3.

2.3. Метод узловых потенциалов

▼ Метод узловых потенциалов широко применяется при расчетах электрических цепей. Так, например, в [30], где приведен сопоставительный анализ нескольких отечественных пакетов прикладных программ, отмечается, что в них используется только метод УП. Одной из причин, способствовавших распространенности этого метода, является то, что матрица инцидентий в этом случае имеет очень простой вид. Она задает связь узлов и ветвей цепи и тем самым непосредственно описывает ее топологию. Если в ЭВМ введен в определенном порядке список ветвей и смежных узлов, то по нему легко подготовить матрицу системы линейных уравнений, не составляя матрицу инцидентий, и достичь при этом определенной экономии оперативной памяти. Так же легко формируется и столбец свободных членов.

Рис. 2.2. Тестовая схема для расчета по методу узловых потенциалов



Как известно, метод УП основан на первом законе Кирхгофа. В предположении, что имеющиеся в цепи источники ЭДС заменены источниками тока, применение метода сводится к составлению и последующему решению уравнения

$$(\mathbf{SRS}^T)\mathbf{X} = \mathbf{SU}, \quad (2.15)$$

где \mathbf{S} — матрица инцидентий; \mathbf{R} — диагональная матрица проводимостей ветвей; \mathbf{S}^T — транспонированная матрица \mathbf{S} ; \mathbf{X} — вектор напряжений в узлах; \mathbf{U} — вектор источников токов в ветвях.

Решение (2.15) имеет вид

$$\mathbf{X} = (\mathbf{SRS}^T)^{-1}\mathbf{SU} \quad (2.16)$$

и может быть закодировано в виде последовательности матричных операторов БЕЙСИКа (подобно строкам 1150–1210 в программе МКТО).

Кроме этой последовательности в программу расчета должны быть включены блоки ввода и вывода и описание массивов.

▼ Опишем порядок составления программы, которую назовем МУПО, для расчета схемы, изображенной на рис. 2.2. Она топологически эквивалентна схеме, приведенной на рис. 2.1. Различие между ними состоит в том, что в последней источник ЭДС заменен на эквивалентный источник тока, внутреннее сопротивление которого равно внутреннему сопротивлению источника ЭДС. Методика такого преобразования описана в ряде руководств по ТОЭ, см., например, [29]. Такой выбор примера позволяет проверять и сопоставлять результаты расчетов, получаемые разными программами (МКТ и МУП). Как и в МКТО, описание массивов дается строками

```
1000 DIM P(K, K), U(K)
1010 DIM S(L, K), T(K, L), G(L, K)
1020 DIM A(L, L), E(L, L), X(L)
```

где K — число ветвей; $L = M - 1$ при числе узлов схемы, равном M . Для рассматриваемой схемы следует ввести с клавиатуры эти строки, подставив в них $K = 6$ и $L = 3$.

Особенность в подготовке к вычислениям по методу УП состоит в выделении базисного узла (называемого также нулевым узлом), в котором потенциал условно принимается равным 0. Получаемые по данному методу значения потенциалов определены относительно потенциала базисного узла. Истинные значения всех потенциалов можно получить простым пересчетом в том случае, когда известен потенциал одного из узлов.

При нумерации базисному узлу целесообразно присвоить нулевой номер, так как это существенно упростит в дальнейшем подготовку системы линейных уравнений.

После выбора направлений токов приступим к построению матрицы инцидентий, которая дает связь между узлами схемы (базисный узел в нее не включается) и условно выбранными направлениями токов в ветвях и составляется по следующему правилу:

$S_{ij} = 0$, если ветвь не примыкает к узлу;

$S_{ij} = 1$, если ветвь примыкает к узлу, причем ток направлен в сторону узла;

$S_{ij} = -1$, если ветвь примыкает к узлу, причем ток направлен от узла.

Пользуясь этим правилом, составим матрицу инцидентий, приведенную в табл. 2.3.

Таблица 2.3

№ узла	№ ветви					
	1	2	3	4	5	6
1	0	1	-1	-1	0	0
2	1	0	1	0	-1	0
3	0	0	0	1	1	1

На этом подготовительная часть работы закончена, можно набирать текст программы.

```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * УЗЛОВЫХ ПОТЕНЦИАЛОВ. ПОСТОЯННЫЙ ТОК . *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ ВЕЙСИКА . *
908 REM * НАЗВАНИЕ: "МУПО" . *
910 REM *****
1000 DIM R[6,6],U[6],S[3,6]
1010 DIM T[6,3],G[3,6],A[3,3]
1020 DIM B[3,3],C[3],X[3]

```



```

1030 DATA 6,2,1,5,4,3
1040 MAT READ U
1050 MAT R=ZER
1060 FOR I=1 TO 6
1070 LET R[I,I]=1/U[I]
1080 NEXT I
1090 MAT U=ZER
1100 LET U[6]=10/3
1110 DATA 0,1,-1,-1,0,0
1120 DATA 1,0, 1,0,-1,0
1130 DATA 0,0,0,1,1,1
1140 MAT READ S
1150 MAT T=TRN(S)
1160 MAT G=S*R
1170 MAT A=G*T
1180 MAT C=S*U
1190 MAT E=INV(A)
1200 MAT X=E*C
1210 PRINT "ПОТЕНЦИАЛЫ УЗЛОВ."
1220 PRINT "(ПОТЕНЦИАЛ БАЗИСНОГО УЗЛА"
1230 PRINT "ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ)."

```

Учитывая, что все данные вводятся оператором READ, остается только запустить МУПО на выполнение. Получаются следующие результаты:

ПОТЕНЦИАЛЫ УЗЛОВ
(ПОТЕНЦИАЛ БАЗИСНОГО УЗЛА
ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ)

2.1157

2.47934

5.58678

Зная величины потенциалов, можно получить токи в ветвях и сравнить их с приведенными на с. 24.

▼ Хотя процесс подготовки матрицы инцидентий для МУП не представляет особых трудностей, тем не менее для автоматизации этой процедуры была разработана программа ПП 3000.

```

2900 REM *****
2902 REM * ПП ФОРМИРОВАНИЯ МАТРИЦЫ ИНЦИДЕНЦИЙ *
2904 REM * ДЛЯ РАСЧЕТА ПО МЕТОДУ УЗЛОВЫХ *
2906 REM * ПОТЕНЦИАЛОВ. *
2908 REM *****
3000 DEF FNA(A)=INT(A)
3002 DEF FNB(B)=INT((B-FNA(B))*NO+.5)
3100 LET N3=N2
3102 FOR K=1 TO N3
3104 FOR K1=1 TO N1
3106 LET S[K,K1]=0
3108 NEXT K1
3110 NEXT K
3112 FOR K=1 TO N1
3114 LET K1=FNA(P[K,1])
3116 IF K1=0 THEN 3120
3118 LET S[K1,K]=-1
3120 LET K1=FNB(P[K,1])
3122 IF K1=0 THEN 3126
3124 LET S[K1,K]=1
3126 NEXT K
3128 RETURN

```



Следует обратить особое внимание на тот факт, что матрицы инцидентий для методов КТ и УП различны, соответственно различны и алгоритмы ПП 3000 для МКТ и для МУП.

Входными данными для ПП 3000 служит массив параметров ветвей P, который вместе с переменными N0, N1, N2 определяется в ПП 2000.

По окончании работы ПП 3000 в массиве S размерностью N3 x N1 (N3 вычисляется в ПП 3000) располагается матрица инцидентий.

▼ Будем последовательно усложнять решаемые задачи и описывать соответствующие программы. Сначала сконструируем программу МУП1. Ее назначение — расчет электрических цепей на постоянном токе методом узловых потенциалов с применением матричных операторов БЕЙСИКА. Для нее фрагмент МУПО ввода исходных данных заменим обращением к подпрограммам ПП 2000, ПП 2500 и ПП 3000. Назначение первой и третьей известны, а ПП 2500 обеспечивает ввод источников тока ветвей, и каждый пользователь должен составлять ее сам для своей конкретной схемы. Для тестовой схемы, приведенной на рис. 2.2, она выглядит так:

```

2500 MAT U = ZER
2510 LET U(6) = 10/P(6,2)
2520 RETURN

```

Строка 2510 позволяет здесь задать источник тока, эквивалентный источнику ЭДС схемы, изображенной на рис. 2.1, с напряжением 10 В и внутренним сопротивлением 3 Ом, что обеспечивает сопоставимость результатов расчета.

Приведем полный текст программы МУП1.



```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * УЗЛОВЫХ ПОТЕНЦИАЛОВ. ПОСТОЯННЫЙ ТОК . *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ ВЭЙСКИА . *
908 REM * НАЗВАНИЕ: "МУП1" . *
910 REM *****
1000 DIM P[6,2],R[6,6],U[6]
1010 DIM S[3,6],T[6,3],G[3,6]
1020 DIM A[3,3],B[3,3],C[3,3],X[3]
1021 REM-----
1022 REM ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ
1023 REM И ТОПОЛОГИИ СХЕМЫ
1024 REM-----
1030 GOSUB 2000
1031 REM-----
1032 REM ВВОД ИСТОЧНИКОВ ТОКА ВЕТВЕЙ
1033 REM-----
1040 GOSUB 2500
1041 REM-----
1042 REM ФОРМИРОВАНИЕ МАТРИЦЫ ИНЦИДЕНЦИЙ
1043 REM-----
1100 GOSUB 3000
1101 REM-----
1102 REM ФОРМИРОВАНИЕ МАТРИЦЫ ПРОВОДИМОСТЕЙ
1103 REM ВЕТВЕЙ
1104 REM-----
1110 MAT R=ZER
1120 FOR K=1 TO N1
1130 LET R[K,K]=1/P[K,2]
1140 NEXT K
1141 REM-----
1142 REM ВЫЧИСЛЕНИЕ МАТРИЦЫ ПРОВОДИМОСТЕЙ

```

```

1143 REM УЗЛОВ
1144 REM-----
1150 MAT T=TRN(S)
1160 MAT G=S*R
1170 MAT A=G*T
1171 REM-----
1172 REM ВЫЧИСЛЕНИЕ ВЕКТОРА
1173 REM УЗЛОВЫХ ТОКОВ
1174 REM-----
1180 MAT C=S*U
1181 REM-----
1182 REM ОБРАЩЕНИЕ МАТРИЦЫ
1183 REM УЗЛОВЫХ ПРОВОДИМОСТЕЙ
1184 REM-----
1190 MAT E=INV(A)
1191 REM-----
1192 REM ВЫЧИСЛЕНИЕ УЗЛОВЫХ
1193 REM ПОТЕНЦИАЛОВ
1194 REM-----
1200 MAT X=E*C
1201 REM-----
1202 REM ВЫВОД РЕЗУЛЬТАТОВ
1203 REM-----
1500 GOSUB 1700
1510 RETURN
1670 REM *****
1680 REM * ПП ПЕЧАТИ НАПРЯЖЕНИЙ В УЗЛАХ *
1690 REM *****
1700 PRINT "ПОТЕНЦИАЛЫ УЗЛОВ."
1710 PRINT "ПОТЕНЦИАЛ ВАЗИСНОГО (НУЛЕВОГО)"
1720 PRINT "УЗЛА ПРИНИМАЕТСЯ РАВНЫМ НУЛД."
1730 MAT PRINT X
1740 RETURN
2470 REM *****
2480 REM * ВВОД ИСТОЧНИКОВ ТОКА ВЕТВЕЙ *
2490 REM *****
2500 MAT U=ZER
2510 LET U[6]=10/P[6,2]
2520 RETURN
9999 END

```

Массивы в ней в общем случае описываются строками

```
1000 DIM P(K, 2), R(K, K), U(K)
1010 DIM S(L, K), T(K, L), G(L, K)
1020 DIM A(L, I), E(L, L), C(L), X(L)
```

где K – число ветвей схемы; $L = M - 1$ при числе узлов, равном M . Для тестовой схемы следует подставить в строки 1000–1020 значения $K = 6$ и $L = 3$.

Проведя вычисления, убедимся в ее работоспособности, а затем сопоставим полученные данные с результатами расчета по программе МКТ1 для тестовой схемы (рис. 2.2).

Укажем, что в данной и последующих программах следует обращать особое внимание на знак (полярность) источников токов. Он принимается положительным, если его действие совпадает с условно выбранным направлением тока в данной ветви.

▼ Переработаем программу МУП1, приспособив ее для версий БЕЙСИКА, не содержащих матричных операторов. Это можно было бы сделать по аналогии с описанной МКТ2, однако в силу простоты матрицы инцидентий, как отмечалось выше, возможно систему линейных уравнений (для которой нужны матрицы удельных проводимостей и вектор узловых токов) сформировать непосредственно, минуя этап составления матрицы инцидентий. Благодаря этому достигается существенная экономия объема используемой памяти и сокращается время счета. В этом сказываются преимущества метода УП по сравнению с методом КТ.

Дальнейшей экономии машинных ресурсов и увеличения быстродействия по сравнению с МКТ2 можно достичь, если решать систему линейных уравнений методом Гаусса.

Законченный текст программы МУП2 со всеми внесенными усовершенствованиями приведен ниже.



```
900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * УЗЛОВЫХ ПОТЕНЦИАЛОВ. ПОСТОЯННЫЙ ТОК .*
906 REM * ВЫЧИСЛЕНИЕ УЗЛОВЫХ ПОТЕНЦИАЛОВ *
908 REM * МЕТОДОМ ГАУССА. НАЗВАНИЕ: "МУП2" . *
910 REM *****
1000 DIM P[6,2],U[6],A[3,4]
1010 DEF FNA(A)=INT(P[A,1])
1020 DEF FNB(B)=INT((P[B,1]-FNA(B))*NO+.5)
1021 REM-----
1022 REM ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ
1023 REM И ТОПОЛОГИИ СХЕМЫ
1024 REM-----
```

```

10 30 GOSUB 2000
10 31 REM-----
10 32 REM ВВОД ИСТОЧНИКОВ ТОКА ВЕТВЕЙ
10 33 REM-----
10 40 GOSUB 2500
10 41 REM-----
10 42 REM ФОРМИРОВАНИЕ МАТРИЦЫ УЗЛОВЫХ
10 43 REM ПРОВОДИМОСТЕЙ И ВЕКТОРА УЗЛОВЫХ
10 44 REM ТОКОВ
10 45 REM-----
1100 LET N3=N2+1
1110 FOR I=1 TO N2
1120 FOR J=1 TO N3
1130 LET A[I,J]=0
1140 NEXT J
1150 NEXT I
1160 FOR K=1 TO N1
1170 LET G=1/P[K,2]
1180 LET I=FNA(K)
1190 IF I=0 THEN 1220
1200 LET A[I,I]=A[I,I]+G
1210 LET A[I,N3]=A[I,N3]-U[K]
1220 LET J=FNB(K)
1230 IF J=0 THEN 1290
1240 LET A[J,J]=A[J,J]+G
1250 LET A[J,N3]=A[J,N3]+U[K]
1260 IF I*J=0 THEN 1290
1270 LET A[I,J]=A[I,J]-G
1280 LET A[J,I]=A[I,J]
1290 NEXT K
1291 REM-----
1292 REM ВЫЧИСЛЕНИЕ УЗЛОВЫХ ПОТЕНЦИАЛОВ
1293 REM-----
1300 LET A1=N2
1310 LET A2=1
1320 LET A3=1.00000E-10
1330 GOSUB 5600
1340 IF A4>0 THEN 1500
1350 PRINT "МАТРИЦА СИСТЕМЫ ВЫРОЖДЕНА,"
1360 PRINT "ВЫЧИСЛЕНИЯ ПРЕКРАЩАЮТСЯ."
1370 STOP

```

```

1371 REM-----
1372 REM ВЫВОД РЕЗУЛЬТАТОВ
1373 REM-----
1500 GOSUB 1700
1510 STOP
1670 REM *****
1680 REM * ПЕЧАТЬ НАПРЯЖЕНИЙ В УЗЛАХ *
1690 REM *****
1700 PRINT "ПОТЕНЦИАЛЫ УЗЛОВ."
1710 PRINT "ПОТЕНЦИАЛ ВАЖНОГО (НУЛЕВОГО)"
1720 PRINT "УЗЛА ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ."
1730 PRINT "N УЗЛА", "U"
1740 FOR K=1 TO N2
1750 PRINT K, A[K, N3]
1760 NEXT K
1770 RETURN
2470 REM *****
2480 REM * ВВОД ИСТОЧНИКОВ ТОКА ВЕТВЕЙ *
2490 REM *****
2500 FOR K=1 TO N1
2510 LET U[K]=0
2520 NEXT K
2530 LET U[6]= 10/ P[6, 2]
2540 RETURN
9999 END

```

Из текста программы наглядно видна достигнутая экономия объема памяти, необходимой для хранения массивов. Они описываются одной строкой:

```
1000 DIM P(K, 2), U(K), A(L, L + 1)
```

где K — число ветвей схемы, и при числе узлов, равном M , $L = M - 1$.

Протестировать программу можно описанным выше способом.

▼ Перейдем к расчету цепей на переменном токе. Программа МУПЗ, предназначенная для расчета таких цепей с использованием матричных операторов, может быть построена аналогично МКТЗ. Алгоритмы расчета в этих двух программах практически полностью совпадают. Отличие состоит в том, что при формировании матрицы проводимостей ветвей в элементах большой диагонали следует размещать не комплексные сопротивления ветвей, подсчитанные при заданной круговой частоте, а их проводимости. В связи с этим помимо ПП 1800, предназначенной для вычисления комплексных сопротивлений ветвей, введена также

ПП 1850, которая, вызывая 1800, находит их проводимости. В данной программе используется матрица инцидентий, поэтому для ее составления привлекается ПП 3000 (в варианте, разработанном для метода узловых потенциалов).

С полным текстом программы можно познакомиться по ее листингу, приведенному ниже.



```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * УЗЛОВЫХ ПОТЕНЦИАЛОВ. ПЕРЕМЕННЫЙ ТОК . *
906 REM * МАТРИЧНЫЕ ОПЕРАТОРЫ ВЭЙСИКА . *
908 REM * НАЗВАНИЕ: "МУПЗ" . *
910 REM *****
1000 DIM P[6,4],R[6,6],Q[6,6],U[6],V[6]
1010 DIM S[3,6],T[6,3],G[3,6],A[3,3],B[3,3]
1020 DIM E[3,3],F[3,3],C[3],D[3],H[3],X[3],Y[3]
1021 REM-----
1022 REM ВВОД ТАБЛИЦЫ ЭЛЕМЕНТОВ ВЕТВЕЙ
1023 REM И ТОПОЛОГИИ СХЕМЫ
1024 REM-----
1030 GOSUB 2000
1031 REM-----
1032 REM ЗАДАНИЕ ЧАСТОТЫ
1033 REM-----
1040 PRINT "F(Гц)=-";
1050 INPUT F
1060 LET W=2*3.14159*F
1061 REM-----
1062 REM ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО
1063 REM ТОКА ВЕТВЕЙ ( ЧАСТОТЫ F )
1064 REM-----
1070 GOSUB 2500
1071 REM-----
1072 REM ФОРМИРОВАНИЕ МАТРИЦЫ ИНЦИДЕНЦИЙ
1073 REM-----
1100 GOSUB 3000
1101 REM-----
1102 REM ФОРМИРОВАНИЕ МАТРИЦЫ
1103 REM ПРОВОДИМОСТЕЙ ВЕТВЕЙ
1104 REM-----
1110 MAT R=ZER
1120 MAT Q=ZER

```

```

1130 FOR K=1 TO N1
1140 GOSUB 1850
1150 LET R[K,K]=R
1160 LET Q[K,K]=Q
1170 NEXT K
1171 REM-----
1172 REM ВЫЧИСЛЕНИЕ МАТРИЦЫ
1173 REM УСЛОВНЫХ ПРОВОДИМОСТЕЙ
1174 REM И ВЕКТОРА УСЛОВНЫХ ТОКОВ
1175 REM-----
1180 MAT T=TRN(S)
1190 MAT G=S*R
1200 MAT A=G*T
1210 MAT C=S*U
1220 MAT G=S*Q
1230 MAT B=G*T
1240 MAT D=S*V
1241 REM-----
1242 REM ОБРАЩЕНИЕ МАТРИЦЫ УСЛОВНЫХ
1243 REM ПРОВОДИМОСТЕЙ ( DET(A)*O )
1244 REM-----
1250 MAT E=INV(A)
1260 MAT F=E*B
1270 MAT B=B*F
1280 MAT B=A+E
1290 MAT E=INV(B)
1300 MAT B=F*E
1310 MAT F=(-1)*B
1311 REM-----
1312 REM ВЫЧИСЛЕНИЕ ПОТЕНЦИАЛОВ УЗЛОВ
1313 REM-----
1320 MAT X=E*C
1330 MAT H=F*D
1340 MAT X=X-H
1350 MAT Y=E*D
1360 MAT H=F*C
1370 MAT Y=Y+H
1371 REM-----
1372 REM ВЫВОД РЕЗУЛЬТАТОВ
1373 REM-----
1500 GOSUB 1700

```

```

1510 STOP
1670 REM *****
1680 REM * ПЕЧАТЬ ПОТЕНЦИАЛОВ УЗЛОВ *
1690 REM *****
1700 PRINT "ПОТЕНЦИАЛЫ УЗЛОВ."
1710 PRINT "ПОТЕНЦИАЛ ВАЗИСНОГО (НУЛЕВОГО)"
1720 PRINT "УЗЛА ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ."
1730 PRINT "N УЗЛА", "RE(U)", "IM(U)"
1740 FOR K=1 TO N2
1750 PRINT K, X[K], Y[K]
1760 NEXT K
1770 RETURN
1790 REM *****
1791 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОГО СОПРОТИВЛЕНИЯ *
1792 REM * K-Й ВЕТВИ ( R+J*Q ) *
1793 REM *****
1800 LET R=P[K,2]
1810 LET Q=W*P[K,2]
1820 IF P[K,4]=0 THEN 1840
1830 LET Q=Q-1/W/P[K,4]
1840 RETURN
1841 REM *****
1842 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОЙ ПРОВОДИМОСТИ *
1843 REM * K-Й ВЕТВИ ( R+J*Q ) *
1844 REM *****
1850 GOSUB 1800
1860 LET Z=R*R+Q*Q
1870 LET R=R/Z
1880 LET Q=-Q/Z
1890 RETURN
2460 REM *****
2470 REM * ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО *
2480 REM * ТОКА ВЕТВЕЙ ( ЧАСТОТЫ F ) *
2490 REM *****
2500 MAT U=ZER
2510 MAT V=ZER
2520 LET K=6
2530 GOSUB 1850
2540 LET U[6]=10*R
2550 LET V[6]=10*Q
2560 RETURN
9999 END

```

В ней в строках 1000–1020 дано описание массивов, используемых при решении тестовой задачи. Для общего случая, когда требуется иная размерность массивов, они должны быть описаны следующим способом:

1000 DIM P(K, 4), R(K, K), Q(K, K), U(K), V(K)
 1010 DIM S(L, K), T(K, L), G(L, K), A(L, L), B(L, L)
 1020 DIM E(L, L), F(L, L), C(L), D(L), H(L), X(L), Y(L)

где K и L – те же, что и для МУП1.

Для тестирования программы используем схему, изображенную на рис. 2.2, параметры ветвей которой приведены в табл. 2.4, а частоту примем равной 0,159155 Гц. Источник переменного тока ветвей вводится в ПП 2500 (см. строки 2500–2580).

Таблица 2.4

N ветвей	Нузлов	R, Ом	L, Гн	C, Ф	I, А
1	0 2	6	6	0	0
2	0 1	2	2	0	0
3	1 2	1	1	0	0
4	1 3	5	6	0	0
5	2 3	4	4	0	0
6	0 3	3	3	0	10

Запустив программу на выполнение и указав в соответствии с запросами ПП 2000 все исходные данные, получим

ПОТЕНЦИАЛЫ УЗЛОВ.

ПОТЕНЦИАЛ БАЗИСНОГО (НУЛЕВОГО)

УЗЛА ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ.

N УЗЛА	RE(U)	IM(U)
1	2.1157	-2.38419E-07
2	2.47934	-2.38419E-07
3	5.558677	-4.76837E-07

Протестировать программу МУП3 можно аналогично тому, как это сделано для МКТЗ в § 2.2 (см. также § 2.4).

Опишем программу МУП4. Ее назначение – расчет цепи на переменном токе без матричных операторов. Алгоритм подготовки и решения систем линейных уравнений в ней совпадает с использованным в МУП2, в частности здесь не применяется матрица инцидентий, что позволяет повысить эффективность программы. Полный ее текст приведен ниже.

```

900 REM *****
902 REM * РАСЧЕТ ЭЛЕКТРИЧЕСКОЙ ЦЕПИ МЕТОДОМ *
904 REM * УЗЛОВЫХ ПОТЕНЦИАЛОВ. ПЕРЕМЕННЫЙ ТОК. *
906 REM * ВЫЧИСЛЕНИЕ ПОТЕНЦИАЛОВ МЕТОДОМ *
908 REM * ГАУССА. НАЗВАНИЕ: "МУП4" . *
910 REM *****
1000 DIM P[6,4],A[3,4],B[3,4],U[6],V[6]
1010 DEF FNA(A)=INT(P[A,1])
1020 DEF FNB(B)=INT((P[B,1]-FNA(B))*NO+.5)
1021 REM-----
1022 REM ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ
1023 REM И ТОПОЛОГИИ СХЕМЫ
1024 REM-----
1030 GOSUB 2000
1031 REM-----
1032 REM ЗАДАНИЕ ЧАСТОТЫ
1033 REM-----
1040 PRINT "F(Гц)=-";
1050 INPUT F
1060 LET W=2*3.14159*F
1061 REM-----
1062 REM ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО
1063 REM ТОКА ВЕТВЕЙ ( ЧАСТОТЫ F )
1064 REM-----
1070 GOSUB 2500
1071 REM-----
1072 REM ФОРМИРОВАНИЕ МАТРИЦЫ УЗЛОВЫХ
1073 REM ПРОВОДИМОСТЕЙ И ВЕКТОРА
1074 REM УЗЛОВЫХ ТОКОВ
1075 REM-----
1100 LET N3=N2+1
1110 FOR I=1 TO N2
1120 FOR J=1 TO N3
1130 LET A[I,J]=0
1140 LET B[I,J]=0
1150 NEXT J
1160 NEXT I
1170 FOR K=1 TO N1
1180 GOSUB 1850
1190 LET I=FNA(K)
1200 IF I=0 THEN 1250
1210 LET A[I,I]=A[I,I]+R

```



```

1220 LET B[I,I]=B[I,I]+Q
1230 LET A[I,N3]=A[I,N3]-U[K]
1240 LET B[I,N3]=B[I,N3]-V[K]
1250 LET J=FNB(K)
1260 IF J=0 THEN 1360
1270 LET A[J,J]=A[J,J]+R
1280 LET B[J,J]=B[J,J]+Q
1290 LET A[J,N3]=A[J,N3]+U[K]
1300 LET B[J,N3]=B[J,N3]+V[K]
1310 IF I*J=0 THEN 1360
1320 LET A[I,J]=A[I,J]-R
1330 LET B[I,J]=B[I,J]-Q
1340 LET A[J,I]=A[I,J]
1350 LET B[J,I]=B[I,J]
1360 NEXT K
1361 REM-----
1362 REM ВЫЧИСЛЕНИЕ УЗЛОВЫХ ПОТЕНЦИАЛОВ
1363 REM-----
1370 LET A1=N2
1380 LET A2=1
1390 LET A3=1.00000E-10
1400 GOSUB 5800
1410 IF A4>0 THEN 1500
1420 PRINT "МАТРИЦА СИСТЕМЫ ВЫРОЖДЕНА,"
1430 PRINT "ВЫЧИСЛЕНИЯ ПРЕКРАЩАЮТСЯ."
1440 STOP
1441 REM-----
1442 REM ВЫВОД РЕЗУЛЬТАТОВ
1443 REM-----
1500 GOSUB 1700
1510 STOP
1670 REM *****
1680 REM * ПЕЧАТЬ ПОТЕНЦИАЛОВ УЗЛОВ *
1690 REM *****
1700 PRINT "ПОТЕНЦИАЛЫ УЗЛОВ."
1710 PRINT "ПОТЕНЦИАЛ ВАРИАНТА (НУЛЕВОГО)"
1720 PRINT "УЗЛА ПРИНИМАЕТСЯ РАВНЫМ НУЛЮ."
1730 PRINT "N УЗЛА", "RE(U)", "IM(U)"
1740 FOR K=1 TO N2
1750 PRINT K, A[K,N3], B[K,N3]
1760 NEXT K

```

```

1770 RETURN
1790 REM *****
1791 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОГО СОПРОТИВЛЕНИЯ *
1792 REM * К-Й ВЕТВИ ( R+J*Q ) *
1793 REM *****
1800 LET R=P[K,2]
1810 LET Q=W*P[K,3]
1820 IF P[K,4]=0 THEN 1840
1830 LET Q=Q-1/W/P[K,4]
1840 RETURN
1841 REM *****
1842 REM * ВЫЧИСЛЕНИЕ КОМПЛЕКСНОЙ ПРОВОДИМОСТИ *
1843 REM * К-Й ВЕТВИ ( R+J*Q ) *
1844 REM *****
1850 GOSUB 1800
1860 LET Z=R*R+Q*Q
1870 LET R=R/Z
1880 LET Q=-Q/Z
1890 RETURN
2460 REM *****
2470 REM * ВВОД ИСТОЧНИКОВ СИНУСОИДАЛЬНОГО *
2480 REM * ТОКА ВЕТВЕЙ *
2490 REM *****
2500 FOR K=1 TO N1
2510 LET U[K]=0
2520 LET V[K]=0
2530 NEXT K
2540 LET K=6
2550 GOSUB 1850
2560 LET U[6]=10*R
2570 LET V[6]=10*Q
2580 RETURN
9999 END

```

Можно заметить, что в программе значительно сокращен объем памяти, выделяемой под массивы, которые описываются следующим образом:

```
1000 DIM P(K, 4), A(L, L + 1), B(L, L + 1), U(K), V(K)
```

где K и L имеют тот же смысл, что и для программы МУПЗ.

2.4. Освоение и тестирование программ

▼ Программы из § 2.2 и 2.3 используют ПП, сводный список которых приведен в табл. 2.5 и 2.6. Часть из них может использоваться во всех расчетах без каких-либо изменений, так как текст их оптимизирован в пределах, задаваемых возможностями выбранного подмножества языка БЕЙСИК. Однако некоторые из них должны быть представлены в иной форме, более удобной и привычной для пользователя. В меньшей степени это относится к ПП 2000, которая допускает некоторую модификацию, главным образом, за счет сокращения или расширения текста диалога,


Таблица 2.5

№ ПП	Назначение ПП	МКТ1	МКТ2	МКТ3	МКТ4
1700	Выходная печать	+	+	+	+
1800	Вычисление комплексного сопротивления ветвей			+	+
2000	Ввод топологии и параметров схемы	+	+	+	+
2500	Ввод источников ЭДС ветвей	+	+	+	+
3000	Формирование матрицы инцидентий	+	+	+	+
5200	Обращение матрицы методом обмена		+		
5800	Решение системы линейных уравнений методом Гаусса				+

Таблица 2.6

№ ПП	Назначение ПП	МУП1	МУП2	МУП3	МУП4
1700	Выходная печать	+	+	+	+
1800	Вычисление комплексного сопротивления ветвей			+	+
1850	Вычисление комплексной проводимости ветвей			+	+
2000	Ввод топологии и параметров ветвей	+	+	+	+
2500	Ввод источников тока ветвей	+	+	+	+
3000	Формирование матрицы инцидентий	+		+	
5600	Решение системы линейных уравнений с действительными коэффициентами методом Гаусса		+		
5800	Решение системы линейных уравнений с комплексными коэффициентами методом Гаусса				+

Рис. 2.3. Граф связей между программами МКТ и МУП. Штриховые линии обозначают сходство решаемых задач



подсказок, более компактного представления данных (при вводе и отображении на дисплей), особенно в случае, если диалог осуществляется через печатающее устройство или результаты счета выводятся на печать.

В целом программы в § 2.2 и 2.3 обеспечивают выполнение расчетов на ПЭВМ электрических цепей при значительном сокращении подготовительных работ и ручных операций. В частности, в них автоматически решается задача составления матрицы инцидентий. Однако в ряде случаев целесообразно применять их без автоматического составления матрицы инцидентий (тогда она составляется 1 раз и вводится операторами DATA-READ). В инженерной практике одна и та же схема обчисляется многократно, так как преобладающими являются многовариантные расчеты, связанные с оптимизацией параметров, минимизацией потерь, анализом гармоник и частотных характеристик. И все это при неизменной топологии и (или) неизменных параметрах ветвей. Очевидно, что вводить для каждого варианта большие массивы данных нецелесообразно. Проще записать их на машинные носители и считывать с них по мере необходимости. Для того, чтобы программы, не использующие ПП 3000, сохранили работоспособность, из последней следует перенести в основной текст строки 3000-3002, которые содержат операторы DEF определения функций, так как эти функции используются и в других частях программ. Ограничения на сложность анализируемых электрических цепей в основном определяются машинными ресурсами.

▼ На рис. 2.3 изображен граф связей между приведенными в данной гл. восемью программами. Штриховые линии обозначают сходство решаемых задач (общность назначения программ), линии со стрелками — "генетическую" связь между программами.

При решении практических задач пользователь сам должен сделать выбор, какую из них надо применить в работе.

Если имеющаяся версия БЕЙСИКа содержит матричные операторы, то можно использовать их, так как текст программ будет короче и нагляднее. Но такие программы будут занимать больше места в памяти.

Можно отметить, что наиболее эффективное использование ресурсов, по-видимому, достигается при работе с программами, имеющими обозначения с номерами 2 и 4 (например, МКТ2, МКТ4, МУП4).

Достаточно сложным вопросом является сравнительная оценка программ, так как она затрудняется целым рядом факторов. К их числу относятся топологическая сложность рассчитываемых цепей, технические характеристики ПЭВМ и особенности их операционной системы. Бурное развитие технологии элементной базы вычислительной техники привело к тому, что в настоящее время находятся в эксплуатации ЭВМ, различающиеся по техническим характеристикам на порядок и больше, а их дальнейшее развитие трудно прогнозировать. Это затрудняет оценку эффективности математического обеспечения. Кроме того, применение интерпретаторов или трансляторов дает различие в скорости вычислений на 1–2 порядка, поэтому не представляется возможным выйти за рамки относительных и весьма неточных оценок, не имеющих большой практической ценности.

▼ При практическом освоении программ можно рекомендовать следующую последовательность действий:

- 1) ввести с клавиатуры текст выбранной программы и записать на машинные носители;
- 2) ввести исходные данные для тестовой задачи;
- 3) провести расчет и сравнить с известными результатами;
- 4) если имеются различия в результатах, то проверить текст, исправить ошибки и снова провести расчет.

При решении собственных задач следует помнить, что требуется изменить ПП 2500, в которой задается вектор источников ЭДС (или источников тока). Кроме того, при использовании МКТ1 и МКТ2, МУП1 и МУП2 необходимо изменить описание массивов, приведя их в соответствие с размерностью задач, как это было описано ранее.

▼ Важнейшим этапом внедрения программ является их тестирование, на что может затрачиваться более 50% всего времени, отведенного на освоение математического обеспечения. На начальной стадии желательно проверить программу на примерах расчета при постоянном токе для определенной регулярной конфигурации (например, лестничные или решетчатые схемы), для которых результаты легко подсчитать заранее, или для схем, изображенных на рис. 2.1 и 2.3. Такое тестирование (и вообще расчет цепи переменного тока при нулевой частоте) позволяет оценить правильность кодирования текста программы и ввода ее в ЭВМ с машинных носителей, проверить укрупненно ее структуру и выявить большую часть ошибок.

В дальнейшем используется более полный тест, охватывающий уже все структурные единицы программы. Наиболее целесообразно провести расчет при искусственно подобранных исходных данных, обеспечивающих получение таких результатов, которые можно оценить иными способами, например аналитически.

В частности, можно рекомендовать выбрать частоту и параметры ветвей (R_i , L_i , C_i) так, чтобы комплексное сопротивление в каждой i -й

ветви было равно $R_i(1 + j)$. Наиболее просто это достигается при условиях

$$L_i = R_i; C_i = 0; f = 1/2\pi = 0,159155. \quad (2.17)$$

Рассмотрим более подробно метод контурных токов. При выполнении условий (2.17) матрица сопротивления ветвей равна $R(1 + j)$, где R – действительная диагональная матрица для случая расчета той же схемы на постоянном токе. Положим теперь, что вектор ЭДС ветвей U не содержит мнимых компонент. При этих условиях соотношение (2.1) примет вид

$$(SR(1 + j)S^T)(X + jY) = SU, \quad (2.18)$$

где по-прежнему S – матрица инцидентий; $X + jY$ – неизвестный (искомый) вектор контурных токов.

Отсюда можно получить

$$(SRS^T)(X + jY) = 0,5(1 - j)SU, \quad (2.19)$$

что эквивалентно двум уравнениям с действительными матрицами

$$(SRS^T)X = 0,5SU; \quad (2.20)$$

$$(SRS^T)Y = -0,5SU.$$

Решая их относительно X и Y , находим

$$X = 0,5Z; \quad (2.21)$$

$$Y = -0,5Z,$$

где $Z = (SRS^T)^{-1}SU$ – решение уравнения (2.1) для постоянного тока.

Очевидно, что соотношения между токами в ветвях на переменном токе будут иметь те же коэффициенты пропорциональности.

При условиях (2.17) проведем расчет схемы рис. 2.1 с помощью программы МКТЗ. Протокол работы программы приведен ниже.

RUN

УКАЖИТЕ ВИД РАБОТЫ,

1. ВВОД ТОПОЛОГИИ СХЕМЫ.
2. ВВОД ТОПОЛОГИИ И ПАРАМЕТРОВ ВЕТВЕЙ (ВЕТВИ СОДЕРЖАТ ТОЛЬКО R),
3. ВВОД ТОПОЛОГИИ И ПАРАМЕТРОВ ВЕТВЕЙ (ВЕТВИ СОДЕРЖАТ ЭЛЕМЕНТЫ R, L, C).

?3

ВЕТВЬ № 1

N1, N2, R, L, C= ?0, 2, 6, 6, 0



ВЕТВЬ № 2

N1, N2, R, L, C= ? 0, 1, 2, 2, 0

ВЕТВЬ № 3

N1, N2, R, L, C= ? 1, 2, 1, 1, 0

ВЕТВЬ № 4

N1, N2, R, L, C= ? 1, 3, 5, 5, 0

ВЕТВЬ № 5

N1, N2, R, L, C= ? 2, 3, 4, 4, 0

ВЕТВЬ № 6

N1, N2, R, L, C= ? 0, 3, 3, 3, 0

ВЕТВЬ № 7

N1, N2, R, L, C= ? -2, 0, 0, 0, 0

СХЕМА СОДЕРЖИТ 6 ВЕТВЕЙ.

ПЕЧАТЬ ТАБЛИЦЫ ПАРАМЕТРОВ НУЖНА

(1-ДА, 0-НЕТ) ?

?0

F(ГЦ) = ?.159155

ТОКИ В ВЕТВЯХ.

N ВЕТВИ

	RE(I)	IM(I)
1	-.206612	.206612
2	-.528926	.528926
3	-.181818	.181818
4	-.347108	.347108
5	-.38843	.38843
6	.735537	-.735537

Сравнив эти данные с приведенными на с. 24 для МКТО, убеждаемся, что в этом случае выполняется соотношение (2.21) и программа МКТЗ работает правильно.

Аналогичные рассуждения можно провести и для метода УП и получить соотношения, подобные (2.21).

Таким образом, при искусственно выбранных значениях параметров ветвей и частоты, соответствующих условиям (2.17), действительная и мнимая части вектора контурных токов будут пропорциональны вектору контурных токов при расчете на постоянном токе с коэффициентами пропорциональности 0,5 и -0,5.

▼ После того как пользователем получены достаточно убедительные подтверждения правильности работы программы, можно рекомендовать

использовать "встроенное тестирование" — добавить в ее структуру фрагмент, содержащий все исходные данные одной из тестовых задач и результаты ее решения. При инициации программы предварительно обрабатывается тестовая задача и полученный результат сравнивается с хранящимся решением. Итоги сравнения выводятся на дисплей, а затем программа переходит к решению текущей задачи.

Все программы, описанные в данной главе, имеют одинаковую структуру и используют одинаково организованные подпрограммы (за исключением ПП 3000, которая составлена по разным алгоритмам для МКТ и МУП), то же относится к структуре и обозначениям массивов. Однако пользователь после освоения этих программ может вносить в них изменения в целях дальнейшего их совершенствования. Эту работу можно разделить на два направления: изменение процедур ввода-вывода и совершенствование алгоритма вычислений.

Модификацию процедур ввода и вывода информации необходимо проводить, если требуется адаптировать программы к имеющейся конфигурации ПЭВМ, приспособить к специфике решаемых задач и уровню квалификации пользователя. Вместо дисплея можно использовать иные средства: при вводе — магнитные ленты и диски, при выводе результатов счета — печать на принтере и запись на магнитные и прочие машинные носители. Диалог с пользователем может быть расширен в целях вывода дополнительных указаний, рекомендаций, сообщений о тех или иных ситуациях. Его следует изменить также при использовании иных средств обмена информацией.

Так, например, при проведении стереотипных вычислений на одинаковых по цели задачах он может быть сокращен. Однако наиболее кардинальные изменения внесут разнообразные средства машинной графики в диалоге. Эти средства, на наш взгляд, — важнейшее направление для улучшения комфортности работы с математическим обеспечением.

Наиболее очевидным направлением совершенствования алгоритмов расчетов является, по-видимому, применение методов обработки слабозаполненных матриц [31] и других приемов, позволяющих сократить время счета в 3–4 раза, а объем необходимой памяти — в 1,5–2 раза [32].

Наконец, отметим, что если реализованная на данной ПЭВМ версия интерпретатора БЕЙСИК позволяет записывать в одной строке несколько операторов, то это также сокращает длительность вычислений.

РАСЧЕТ ЦЕПЕЙ МЕТОДОМ ЭКВИВАЛЕНТНЫХ ПРЕОБРАЗОВАНИЙ

3.1. Метод редукции

▼ В предыдущей главе были приведены алгоритмы и программы расчета электрических цепей с применением методов контурных токов и узловых потенциалов. Они универсальны, и в этом их основное достоинство. Однако эти методы основаны на использовании операций с матрицами, большинство из которых являются слабозаполненными. Это обстоятельство приводит к снижению вычислительной эффективности программ — увеличению времени счета и потребности в оперативной памяти.

Поэтому наряду с методами КТ и УП в практике электротехнических расчетов на ЭВМ находят применение и другие методы и вычислительные алгоритмы, в том числе — метод редукции или сворачивания электрических цепей, примеры реализации которого на ПЭВМ излагаются в данной главе.

Редукция широко применяется при анализе электронных схем и многополюсных электрических систем, характеризующихся большим количеством элементов и сложной структурой их соединения. Рассматриваемые схемы в целях проведения анализа представляются в виде схем замещения или "расчетных схем" с линейными элементами и не содержащими источников тока и ЭДС, а затем при помощи цепочки эквивалентных преобразований сводятся в общем случае к многополюснику. Его вершины (полюсы) выделяются заранее в соответствии с постановкой задачи и в преобразованиях не участвуют (не сокращаются). Наиболее простой и в силу этого наиболее распространенный способ преобразования электрических цепей состоит в замене последовательно и параллельно соединенных ветвей одной эквивалентной, что, однако, возможно далеко не всегда. Более сложным представляется преобразование участка схемы, содержащего сопротивление, соединенные треугольником, в трехлучевую звезду, а также преобразование участка со звездообразным соединением в многоугольник. В последнем случае применение ЭВМ наиболее целесообразно, поскольку объем вычислений становится значительным и зависит от числа сторон результирующего многоугольника. Отметим, что известны и другие виды преобразований [33, 34], однако в данной главе они не рассматриваются.

Многие пользователи считают метод редукции более простым и надежным, чем иные методы расчета, обеспечивающим минимум ошибок и хорошие возможности контроля при реализации на ЭВМ. Последнее основано на том, что метод состоит в программировании цепочки эквивалентных преобразований, отражающих каждый шаг "ручного" сворачивания цепи до представления ее многополюсником (в частном случае — двухполюсником).




Рис. 3.1. Последовательность эквивалентных преобразований тестовой схемы

Такой подход эквивалентен в математическом смысле решению системы линейных уравнений методом последовательного исключения неизвестных. Алгоритм редукции по существу является одним из вариантов решения систем линейных уравнений по методу Гаусса (прямой ход). Проблема выбора текущего номера удаляемого узла здесь аналогична выбору главного элемента при решении системы [35]. Программа, разработанная по такому методу, проще тестируется (можно проконтролировать каждый шаг расчета), а массивы, необходимые для вычислений, требуют значительно меньшего объема памяти. И хотя программа, реализующая свертывание с использованием эквивалентных преобразований, длиннее описанных ранее в гл. 2 и не обладает универсальностью, более того, она составляется каждый раз практически заново для цепей с новой топологией, однако ее разработка и особенно использование иногда могут быть весьма эффективными. При проведении многовариантных расчетов с повышенной кратностью повторения целесообразно вместо методов КТ и УП применять метод редукции.

▼ Метод редукции позволяет, как отмечалось выше, преобразовывать схемы, не содержащие источников тока и ЭДС, т.е. состоящие только из пассивных элементов. В качестве примера рассмотрим схему, изображенную на рис. 3.1, а и полученную из схемы, приведенной на рис. 2.1, удалением ветви, содержащей источник ЭДС. Решим задачу: рассчитать эквивалентное сопротивление этой схемы относительно узлов 0 и 3 на постоянном токе.

На рис. 3.1, б – д показаны последовательные этапы преобразования тестовой схемы в двухполюсник (отметим, что последовательность операций сворачивания может быть и иной). Соответствующие этим этапам

последовательные эквивалентные преобразования выражаются соотношениями

$$\begin{aligned}
 R_6 &= R_1 + R_3 + R_1 R_3 / R_5; \\
 R_7 &= R_3 + R_5 + R_3 R_5 / R_1; \\
 R_8 &= R_5 + R_1 + R_5 R_1 / R_3; \\
 R_9 &= R_2 R_6 / (R_2 + R_6); \\
 R_{10} &= R_4 R_7 / (R_4 + R_7); \\
 R_{11} &= R_9 + R_{10}; \\
 R_{12} &= R_8 R_{11} / (R_8 + R_{11}).
 \end{aligned}
 \tag{3.1}$$

где R_{12} — результирующее (эквивалентное) сопротивление двухполюсника.

По этим соотношениям, которые представляют собой алгоритм эквивалентирования, составим программу на БЕЙСИКе. Назовем ее ПРЭСО (программа расчета эквивалентной схемы). Текст ее приведен ниже.



```

900 REM *****
902 REM * ПРЕОБРАЗОВАНИЕ ЭЛЕКТРИЧЕСКОЙ СХЕМЫ *
904 REM * МЕТОДОМ РЕДУКЦИИ. ПОСТОЯННЫЙ ТОК. *
906 REM * НАЗВАНИЕ: "ПРЭСО" . *
908 REM *****
1000 DATA 6,2,1,5,4
1010 READ R1,R2,R3,R4,R5
1020 LET R6=R1+R3+R1*R3/R5
1030 LET R7=R3+R5+R3*R5/R1
1040 LET R8=R5+R1+R5*R1/R3
1050 LET R9=R2*R6/(R2+R6)+R4*R7/(R4+R7)
1060 LET R=R8*R9/(R8+R9)
1070 PRINT "СОПРОТИВЛЕНИЕ ЭКВИВАЛЕНТНОГО ДВУХПОЛЮСНИКА"
1080 PRINT "R=";R
9999 END
    
```

Проводя расчет при $R_1 = 6$ Ом, $R_2 = 2$ Ом, $R_3 = 1$ Ом, $R_4 = 5$ Ом, $R_5 = 4$ Ом, получаем искомое сопротивление двухполюсника $R_9 = 3,79775$ Ом.

▼ Положим, что в той же схеме в каждой ее ветви имеются активные и реактивные элементы R, L, C . Разработаем программу ПРЭС1 для расчета этой схемы на переменном токе, т.е. найдем эквивалентное сопротивление двухполюсника при частоте F .

Известен прием, облегчающий проведение практических расчетов: прежде, чем проводить расчет на переменном токе (вручную или с помощью ЭВМ), сначала составляют последовательность эквивалентных преобразований для расчета на постоянном токе в виде соотношений, похожих на (3.1), а затем переписывают их с учетом комплексных сопротивлений. Для нас таким конспектом, планом или наброском алгоритма будет служить программа ПРЭС0. В ней каждую арифметическую операцию с действительными числами следует заменить обращением к подпрограммам, выполняющим операции с комплексными числами.

Для данной схемы нужны четыре ПП: умножения и деления комплексных чисел, замены параллельных ветвей, а также вспомогательная ПП (их номера 2500, 2540, 2590 и 2690). Обращение к ним целесообразно унифицировать, что уменьшает вероятность появления ошибок. Вызов первых трех ПП производится одинаково. Исходные данные для них (комплексные числа) заносятся в первую и вторую строки вспомогательного массива Z, а результат по окончании работы ПП размещается в третьей строке того же двумерного массива. В нем в первом столбце должна находиться действительная часть, а во втором — мнимая часть комплексного числа. Вспомогательная ПП 2690 позволяет экономно заменить строки 1020—1040 в программе ПРЭС0 на эквивалентные при расчете на переменном токе.

Использование этих ПП позволяет существенно упорядочить написание довольно длинного текста программы расчета тестовой схемы. В программе предусмотрены: а) ввод исходных данных, в том числе значения частот при анализе на переменном токе; б) подготовка таблицы сопротивлений; в) проведение последовательности преобразований в соответствии с порядком свертывания; г) вывод результатов.

Как и в программах МКТ и МУП, приведенных в гл. 2, в программе ПРЭС1 подготовка топологии и ввод параметров ветвей осуществляется с помощью ПП 2000 (однако не универсальной, текст которой приведен в гл. 1, а написанной специально для ПРЭС1). Формирование таблицы и последовательные этапы эквивалентных преобразований осуществляются ПП 1500. Эти ПП вызываются из головной программы (строки с номерами 1000—1100), где вводится также и значение частоты. Такая блочная структура облегчает видоизменение ГП, необходимое при переходе к расчету другой схемы. Текст программы ПРЭС1 приведен ниже, а о методике ее тестирования см. в § 3.4.

```

990 REM *****
992 REM * ПРЕОБРАЗОВАНИЕ ЭЛЕКТРИЧЕСКОЙ СХЕМЫ *
994 REM * МЕТОДОМ РЕДУКЦИИ. ПЕРЕМЕННЫЙ ТОК. *
996 REM * НАЗВАНИЕ: "ПРЭС1" . *
998 REM *****

```



```

1000 DIM R[10],M[10],L[5],C[5].Z[5,2]
1010 GOSUB 2000
1020 PRINT "F=";
1030 INPUT F
1040 GOSUB 1500
1050 PRINT "RE(Z)=";Y1,"IM(Z)=";Y2
1060 PRINT "РАСЧЕТ ПРИ ДРУГОЙ ЧАСТОТЕ ПОВТОРИТЬ"
1070 PRINT "(1-ДА,О-НЕТ) ?"
1080 INPUT Y
1090 IF Y=1 THEN 1020
1100 STOP
1490 REM *****
1492 REM * ВЪЧЛЕНИЕ RE(Z) И IM(Z) *
1494 REM * ДВУПЛОСЧИКА ПРИ ЧАСТОТЕ F *
1496 REM *****
1500 LET W=2*3.14159*F
1510 FOR I=1 TO 5
1520 LET M[T]=W*L[I]
1530 IF W*C[T]=0 THEN 1550
1540 LET M[T]=M[T]-1/W/C[I]
1550 NEXT I
1560 FOR I=1 TO 3
1570 LET J=I*2-1
1580 FOR K=1 TO 3
1590 LET L=J+K*2-2
1600 LET M=L-INT(T/6)*6
1610 LET Z[K,1]=R[M]
1620 LET Z[K,2]=M[M]
1630 NEXT K
1640 GOSUB 2700
1650 LET R[5+I]=Z[4,1]
1660 LET M[5+I]=Z[4,2]
1670 NEXT I
1680 LET Z[1,1]=R[2]
1690 LET Z[1,2]=M[2]
1700 LET Z[2,1]=R[6]
1710 LET Z[2,2]=M[6]
1720 GOSUB 2600
1730 LET R[9]=Z[3,1]
1740 LET M[9]=Z[3,2]
1750 LET Z[1,1]=R[4]

```

```

1760 LET Z[1,2]=M[4]
1770 LET Z[2,1]=R[7]
1780 LET Z[2,2]=M[7]
1790 GOSUB 2600
1800 LET Z[1,1]=R[8]
1810 LET Z[1,2]=M[8]
1820 LET Z[2,1]=R[9]+Z[3,1]
1830 LET Z[2,2]=M[9]+Z[3,2]
1840 GOSUB 2600
1850 LET Y1=Z[3,1]
1860 LET Y2=Z[3,2]
1870 RETURN
1990 REM *****
1992 REM * ВВОД ПАРАМЕТРОВ СХЕМЫ *
1994 REM *****
2000 DATA 6,6,0
2010 DATA 2,2,0
2020 DATA 1,1,0
2030 DATA 5,5,0
2040 DATA 4,4,0
2050 FOR I=1 TO 5
2060 READ R[I],L[I],C[I]
2070 NEXT I
2080 RESTORE
2090 RETURN
2500 REM *****
2502 REM * УМНОЖЕНИЕ *
2504 REM *****
2510 LET Z[3,1]=Z[1,1]*Z[2,1]-Z[1,2]*Z[2,2]
2520 LET Z[3,2]=Z[1,1]*Z[2,2]+Z[2,1]*Z[1,2]
2530 RETURN
2540 REM *****
2542 REM * ДЕЛЕНИЕ *
2544 REM *****
2550 LET Z=Z[2,1]*Z[2,1]+Z[2,2]*Z[2,2]
2560 LET Z[3,1]=(Z[1,1]*Z[2,1]+Z[1,2]*Z[2,2])/Z
2570 LET Z[3,2]=(Z[2,1]*Z[1,2]-Z[1,1]*Z[2,2])/Z
2580 RETURN
2590 REM *****
2592 REM * ЗАМЕНА ДВУХ ПАРАЛЛЕЛЬНЫХ *
2594 REM * СОПРОТИВЛЕНИЙ ЭКВИВАЛЕНТНЫМ *

```

```

2596 REM *****
2600 LET Z[4,1]=Z[1,1]+Z[2,1]
2610 LET Z[4,2]=Z[1,2]+Z[2,2]
2620 GOSUB 2510
2630 LET Z[1,1]=Z[3,1]
2640 LET Z[1,2]=Z[3,2]
2650 LET Z[2,1]=Z[4,1]
2660 LET Z[2,2]=Z[4,2]
2670 GOSUB 2550
2680 RETURN
2690 REM *****
2692 REM * ВСПОМОГАТЕЛЬНАЯ ПП *
2694 REM *****
2700 LET Z[5,1]=Z[1,1]+Z[2,1]
2710 LET Z[5,2]=Z[1,2]+Z[2,2]
2720 LET Z[4,1]=Z[3,1]
2730 LET Z[4,2]=Z[3,2]
2740 GOSUB 2620
2750 LET Z[4,1]=Z[3,1]+Z[5,1]
2760 LET Z[4,2]=Z[3,2]+Z[5,2]
2770 RETURN
9999 END

```

3.2. Подпрограммы эквивалентных преобразований

▼ Если проанализировать процесс свертывания электрических цепей, например, схемы, изображенной на рис. 2.1, то можно заметить, что он состоит из преобразования звезда — треугольник, трех преобразований параллельно соединенных ветвей и одного преобразования последовательно соединенных ветвей. Очевидно, что, разработав нужные ПП, можно сократить длину ГП (что особенно будет эффективно для программ достаточно сложных) и улучшить ее структуру и наглядность. Закодировать формулы, по которым производятся преобразования (см., например, [27] и другие учебники ТОЭ), не составит большого труда.

Однако существуют причины, по которым целесообразно использовать ПП с более сложной структурой по сравнению с теми, что были использованы ранее. Во-первых, при обмене информацией между ГП и ПП необходимо пересылать большое количество входных и выходных величин. При использовании языка БЕЙСИК эти процедуры требуют особого внимания: в целях повышения их эффективности (увеличения скорости расчета, уменьшения длины текста) следует организовать их

особым образом. Во-вторых, необходимо строго описывать и соблюдать порядок размещения результатов вычислений. Особенно это относится к случаю преобразования многолучевая звезда – многоугольник, когда число лучей звезды больше трех, что приводит к значительному увеличению количества выходной информации.

Поэтому были разработаны такие варианты ПП эквивалентных преобразований, которые позволяют создать некоторую единую систему, построенную по единому плану с унифицированным способом передачи данных из ГП в ПП и обратно, что упрощает процедуры обмена.

Все эти ПП предназначены для преобразования схем как на постоянном, так и на переменном токе (для различения этих режимов введен ключ: $Z\phi = 1$ для первого случая и $Z\phi = 2$ для второго случая). Они в свою очередь используют ПП, выполняющие операции с комплексными числами (см. ПП4210, ПП4220 и др. в гл. 5).

Общая особенность ПП состоит в том, что исходные данные для них (сопротивления ветвей преобразуемой схемы) должны быть размещены в двумерном массиве Z , который при $Z\phi = 1$ содержит два столбца (в первом записывается сопротивление, во втором – упакованные по два номера узлов), а при $Z\phi = 2$ массив имеет три столбца (в первом и втором из них размещаются действительная и мнимая части комплексного сопротивления, в третьем – номера узлов). Введение специального массива Z для исходных данных, промежуточных и конечных результатов позволяет упростить обращение к ПП, отказавшись от многочисленных присвоений и пересылок, что особенно важно при расчетах на переменном токе для схем, имеющих достаточно сложную топологию. При этом сокращается длина программы, а быстроедействие ее возрастает.

Задание ветви схемы номерами смежных узлов и упаковка этих номеров производятся так, как это описано в гл. 1 в ПП 2000.

Необходимость иметь дополнительный массив Z (как известно, с помощью ПП 2000 исходные данные вводятся в массив P) с иной структурой размещения информации, отличной от массива P , обусловлена стремлением сохранить исходные данные и повысить эффективность программ.

Перейдем к описанию ПП эквивалентных преобразований.

```

4000 REM *****
4001 REM * ВЫДЕЛЕНИЕ НОМЕРОВ УЗЛОВ *
4002 REM *****
4010 LET U=Z0+1
4012 LET U1=Z[Z1,U]
4014 LET U2=Z[Z2,U]
4016 LET U3=FNA(U1)
4018 LET U4=(U3-FNA(U2))*(U3-FNB(U2))

```



```

4020 LET U4=SGN(ABS(U4))
4022 LET U3=U3*(1-U4)+FNB(U1)*U4
4024 LET U1=FNA(U1)+FNB(U1)-U3
4026 LET U2=FNA(U2)+FNB(U2)-U3
4028 IF U1<U2 THEN 4036
4030 LET U4=U1
4032 LET U1=U2
4034 LET U2=U4
4036 RETURN
4037 REM *****
4038 REM * ПОСЛЕДОВАТЕЛЬНОЕ СОЕДИНЕНИЕ *
4039 REM *****
4040 GOSUB 4000
4042 LET Z[Z3,U]=U1+U2/NO
4044 GOSUB 4210
4046 RETURN
4047 REM *****
4048 REM * ПАРАЛЛЕЛЬНОЕ СОЕДИНЕНИЕ *
4049 REM *****
4050 LET Z[Z3,Z0+1]=Z[Z1,Z0+1]
4052 GOSUB 4290
4054 RETURN
4055 REM *****
4056 REM * ПРЕОБРАЗОВАНИЕ ТРЕУГОЛЬНИК-ЗВЕЗДА *
4057 REM * И МНОГОЛУЧЕВАЯ ЗВЕЗДА-МНОГУГОЛЬНИК *
4058 REM *****
4060 GOSUB 4320
4062 LET K=Z1-1
4064 LET K1=3*(1-K)+Z2*K
4066 LET K2=Z3+(K1-1)*K1/2-K1+1
4068 LET K3=K2
4070 LET K4=Z4-K2-1
4072 IF K4 >= 0 THEN 4078
4074 PRINT "В МАССИВЕ Z НЕ ХВАТАЕТ ";ABS(K4);" СТОК."
4076 STOP
4078 LET Z2=Z4
4080 LET Z[Z2,1]=0
4082 IF Z0=1 THEN 4086
4084 LET Z[Z2,2]=0
4086 FOR K=1 TO K1
4088 LET Z1=K[K]

```

```

4090 IF Z5=1 THEN 4098
4092 LET Z3=Z4-1
4094 GOSUB 4270
4096 LET Z1=Z3
4098 LET Z3=Z2
4100 GOSUB 4210
4102 NEXT K
4104 FOR K=K1 TO 2 STEP -1
4106 FOR KO=K-1 TO 1 STEP -1
4108 LET Z1=K[K]
4110 LET Z2=K[KO]
4112 LET K5=K1+K2-27
4114 IF K2 <= 27 THEN 4124
4116 LET Z3=K2
4118 IF K2=K3 THEN 4126
4120 LET K[K5]=K2
4122 GOTO 4126
4124 LET Z3=K[K5]
4126 GOSUB 4164
4128 GOSUB 4220
4130 LET Z1=Z3
4132 LET Z2=Z4
4134 IF Z5=2 THEN 4140
4136 GOSUB 4240
4138 GOTO 4142
4140 GOSUB 4220
4142 LET K2=K2-1
4144 NEXT KO
4146 NEXT K
4148 LET Z[K[1],U]=Z[K3,U]
4150 LET Z[K[1],1]=Z[K3,1]
4152 IF Z0=1 THEN 4156
4154 LET Z[K[1],2]=Z[K3,2]
4156 GOSUB 4330
4158 LET Z2=K1*(K1-1)/2
4160 LET Z3=K3-1
4162 RETURN
4164 GOSUB 4010
4166 IF Z5=2 THEN 4174
4168 LET U1=U3
4170 LET U2=Z6

```

```
4172 GOSUB 4028
4174 LET Z[Z3,U]=U1+U2/NO
4176 RETURN
```

▼ ПП 4040 предназначена для замены двух последовательно соединенных ветвей одной эквивалентной. До обращения к ней помимо $Z\emptyset$ следует указать номера строк $Z1$ и $Z2$ массива Z , в которых содержится информация об объединяемых ветвях, а в переменной $Z3$ указать номер строки массива Z , в которую заносятся результаты вычислений.

▼ ПП 4050 позволяет заменить две параллельно соединенные ветви схемы. При обращении к ней кроме параметров $Z\emptyset - Z3$, имеющих тот же смысл, что и в ПП 4040, нужно в $Z4$ указать номер вспомогательной строки для хранения промежуточного результата в массиве Z .

▼ ПП 4060 имеет более универсальный характер и предназначена для преобразования треугольник – звезда (параметр $Z1 = 1$) и многолучевая звезда – многоугольник ($Z1 = 2$).

При $Z1 = 1$ перед вызовом ПП следует указать в $Z2$ номер центрального узла звезды, в $Z3$ – номер последней занятой строки массива Z , а в $Z4$ – общее число строк этого массива. Кроме того, в одномерном массиве K нужно перечислить номера тех строк массива Z , в которых содержится информация о ветвях треугольника, подлежащего преобразованию. По окончании работы ПП результаты преобразования (сопротивления лучей звезды и номера узлов для этих лучей) помещаются в массив Z на место исходных данных.

Для преобразования многолучевая звезда – многоугольник ($Z1 = 2$) нужно положить $Z2$ равным числу лучей звезды, а $Z3$ и $Z4$, задать так же, как и в предыдущем случае. В массиве K указываются номера строк исходных данных.

▼ По окончании работы ПП результаты (сопротивления сторон многоугольника и номера их узлов) размещаются на месте исходных данных, а также последовательно после последней занятой строки массива Z (т.е. начиная со строки $Z3 + 1$). Напомним, последнее необходимо в связи с тем, что при числе лучей звезды больше трех число сторон многоугольника больше числа лучей звезды. В массив K заносятся номера строк, в которых размещены результаты. $Z2$ содержит число сторон многоугольника, $Z3$ – номер последней занятой строки в массиве Z .

Если в процессе работы размерность массива Z , предусмотренная в тексте программы, окажется недостаточной для размещения результатов, то ПП 4060 выдает сообщение

В МАССИВЕ Z НЕ ХВАТАЕТ ... СТРОК

и выполнение программы прерывается. Тогда нужно увеличить размерность массива Z и снова запустить программу на выполнение. Такое

диагностическое сообщение позволяет быстрее обнаружить ошибки и ускоряет выбор нужной размерности массива.

Все ПП используют ПП 4010, которая производит распаковку номеров узлов и другие операции с ними.

Применяемые в ПП функции FNA и FNB должны быть определены в ГП так, как это сделано, например, в МКТ1 (см. гл. 2).

Используя все перечисленные процедуры для расчета тестовой схемы, получаем новую программу ПРЭС2, которая является аналогом ПРЭС1, но отличается более совершенной структурой. О том, как ее испытать, см. § 3.4.



```
990 REM *****
992 REM * ПРЕОБРАЗОВАНИЕ ЭЛЕКТРИЧЕСКОЙ СХЕМЫ *
994 REM * МЕТОДОМ РЕДУКЦИИ. ПЕРЕМЕННЫЙ ТОК. *
996 REM * НАЗВАНИЕ: "ПРЭС2" . *
998 REM *****
1000 DIM Z[10,3],K[10],N[5],R[5],L[5],C[5]
1010 DEF FNA(A)=INT(ABS(A))
1020 DEF FNB(B)=INT((ABS(B)-FNA(B))*NO+.5)
1030 GOSUB 2000
1040 PRINT "F(Гц)=";
1050 INPUT F
1060 GOSUB 1500
1070 PRINT "RE(Z)=";Y1,"IM(Z)=";Y2
1080 PRINT "РАСЧЕТ ПРИ ДРУГОЙ ЧАСТОТЕ ПОВТОРИТЬ"
1090 PRINT "(1-ДА,0-НЕТ) ?"
1100 INPUT Y
1110 IF Y=1 THEN 1040
1120 STOP
1490 REM *****
1492 REM * ВЫЧИСЛЕНИЕ RE(Z) И IM(Z) *
1494 REM * ДВУХПОЛЮСНИКА ПРИ ЧАСТОТЕ F *
1496 REM *****
1500 LET W=2*3.14159*F
1510 FOR I=1 TO 5
1520 LET Z[I,1]=R[I]
1530 LET Z[I,2]=W*L[I]
1540 IF C[I]*W=0 THEN 1560
1550 LET Z[I,2]=Z[I,2]-1/W/C[I]
1560 LET Z[I,3]=N[I]
1570 NEXT I
1580 LET ZO=2
```

```

1590 LET Z1=2
1600 LET Z2=3
1610 LET Z3=5
1620 LET Z4=10
1630 LET K[1]=1
1640 LET K[2]=3
1650 LET K[3]=5
1660 GOSUB 4060
1670 LET Z1=1
1680 LET Z2=4
1690 LET Z3=1
1700 GOSUB 4050
1710 LET Z1=2
1720 LET Z2=3
1730 LET Z3=2
1740 GOSUB 4050
1750 LET Z1=2
1760 LET Z2=1
1770 LET Z3=1
1780 GOSUB 4040
1790 LET Z1=1
1800 LET Z2=5
1810 LET Z3=1
1820 GOSUB 4050
1830 LET Y1=Z[1,1]
1840 LET Y2=Z[1,2]
1850 RETURN
1990 REM *****
1992 REM * ВВОД ПАРАМЕТРОВ СХЕМЫ *
1994 REM *****
2000 DATA 0,2,6.6,0
2010 DATA 0,1,2,2,0
2020 DATA 1,2,1,1,0
2030 DATA 1,3,5,5,0
2040 DATA 2,3,4,4,0
2050 LET NO=100
2060 FOR I=1 TO 5
2070 READ K,L,R[I],L[I],C[I]
2080 LET N[I]=K*NO+L
2090 NEXT I
2100 RETURN
9999 END

```

3.3. Автоматизированные методы расчета

▼ Как уже отмечалось ранее, программы, реализующие метод редукции, являются по необходимости уникальными – для каждой новой цепи с конкретной топологией нужно разрабатывать новую программу. А это связано с большими трудозатратами на кодирование, отладку и тестирование программ. Второй их недостаток состоит в том, что при числе ветвей цепи более 15–20 усложняется процесс составления алгоритма программы свертывания схемы, так как нужно предусмотреть безошибочную подготовку промежуточных данных для передачи их в очередную подпрограмму эквивалентного преобразования, а после получения результатов ее работы следует правильно разместить их в соответствующем месте массива. Затруднения вызывают и некоторые другие детали процесса вычислений. Такое положение малопривлекательно. Можно считать, что без преодоления перечисленных трудностей метод редукции не может быть положен в основу достаточно совершенного математического обеспечения.

▼ Создание более совершенных программ расчета методом редукции возможно на пути стандартизации ввода-вывода и использования подпрограмм типа ПП 4040 – ПП 4060, в которых организован экономный способ обмена информацией между ГП и ПП. Еще одним направлением совершенствования программ является автоматизация процесса составления последовательности эквивалентных преобразований (алгоритма редукции). Можно выделить три уровня автоматизации. В первом случае пользователь сам решает, в каком порядке осуществлять процесс вычислений, вводит указания о выбранном подмножестве ветвей и код эквивалентного преобразования, которое надо произвести над ними (или номер соответствующей ПП из числа приведенных в § 3.2). В ЭВМ организуются нужные пересылки и присвоения, вызывается указанная ПП, результат ее работы выводится на экран. Далее процедура повторяется. От такой программы нетрудно сделать еще один шаг: после получения списка очередного подмножества преобразуемых ветвей автоматически определить код преобразования, произвести вызов ПП и разместить результаты ее работы в нужном месте рабочего массива. Подобный способ организации вычислений использован в приведенной далее программе ПРЭСЗ.

Наконец, наиболее высокий уровень автоматизации означает наличие алгоритма, полностью исключающего вмешательство пользователя (исключается диалоговый ввод указаний, что существенно сокращает общую длительность работы). В этом случае программа выбирает те или иные группы ветвей, определяет нужные процедуры, выполняет их с учетом установленных условий и ограничений. (Такой алгоритм будет реализован в программе ПРЭС4 – см. далее). В данном параграфе речь будет идти о программах, соответствующих разным уровням автоматизации вычислений.

▼ Описываемая ниже ПРЭСЗ является универсальной — она предназначена для расчетов на постоянном и переменном токах охем с произвольной топологией методом эквивалентных преобразований. Во-первых, такая универсальность достигается за счет использования ПП 2000 (см. гл. 1) для ввода данных. Во-вторых, после ввода параметров ветвей и топологии пользователь в режиме диалога указывает последовательно один этап преобразования за другим. Причем после каждого очередного сеанса ввода указаний эти преобразования тут же выполняются и ему снова предлагается выбрать очередной участок цепи, который нужно свернуть. Процесс принятия решения облегчается при использовании таблицы, которая отображается на экране дисплея и содержит промежуточные результаты — всю информацию о схеме после завершения очередного этапа преобразований (перечень номеров ветвей, узлов и величин сопротивлений).

Эта таблица выводится на экран с помощью ПП 9000.



```

900 REM *****
902 REM * ПРЕОБРАЗОВАНИЕ ЭЛЕКТРИЧЕСКИХ СХЕМ *
904 REM * ПРОИЗВОЛЬНОЙ ТОПОЛОГИИ МЕТОДОМ *
906 REM * РЕДУКЦИИ. ПОСТОЯННЫЙ И ПЕРЕМЕННЫЙ *
908 REM * ТОК. НАЗВАНИЕ: "ПРЭСЗ" . *
910 REM *****
1000 DIM Z[100,3],P[200,4],K[50]
1010 DEF FNA(A)=INT(ABS(A))
1020 DEF FNB(B)=INT((ABS(B)-FNA(B))*NO+.5)
1030 GOSUB 2000
1040 PRINT "УКАЖИТЕ ЗНАЧЕНИЕ ЧАСТОТЫ F В ГЦ"
1050 PRINT "(F=0 ОЗНАЧАЕТ, ЧТО РАСЧЕТЫ ПРОВОДЯТСЯ"
1060 PRINT "ДЛЯ ПОСТОЯННОГО ТОКА)."

```

```

1200 PRINT "ТАБЛИЦА СОПРОТИВЛЕНИЙ ВЕТВЕЙ СХЕМЫ"
1210 GOSUB 9000
1220 PRINT "СОЕДИНЕНИЕ (1-ПОСЛЕДОВАТЕЛЬНОЕ,2-ПАРАЛЛЕЛЬНОЕ,"
1230 PRINT "3-ТРЕУГОЛЬНИКОМ,4-ЗВЕЗДОЙ,0-КОНЕЦ РАБОТЫ) =";
1240 INPUT N
1250 IF N=0 THEN 1710
1260 IF N=1 THEN 1320
1270 IF N=2 THEN 1370
1280 IF N=3 THEN 1490
1290 IF N=4 THEN 1540
1300 PRINT "МОД НЕВЕРЕН. ВВЕДИТЕ ЕЩЕ РАЗ."
1310 GOTO 1220
1320 PRINT "НОМЕРА ДВУХ ПОСЛЕДОВАТЕЛЬНО СОЕДИНЕННЫХ ВЕТВЕЙ =";
1330 INPUT Z1,Z2
1340 LET Z3=Z1
1350 GOSUB 4040
1360 GOTO 1420
1370 PRINT "НОМЕРА ДВУХ ПАРАЛЛЕЛЬНО СОЕДИНЕННЫХ ВЕТВЕЙ =";
1380 INPUT Z1,Z2
1390 LET Z3=Z1
1400 LET Z4=N8+1
1410 GOSUB 4050
1420 LET N8=N8-1
1430 FOR I=Z2 TO N8
1440 FOR J=1 TO Z0+1
1450 LET Z[I,J]=Z[I+1,J]
1460 NEXT J
1470 NEXT I
1480 GOTO 1680
1490 PRINT "НОМЕРА ТРЕХ ВЕТВЕЙ, СОЕДИНЕННЫХ ТРЕУГОЛЬНИКОМ, =";
1500 INPUT K[1],K[2],K[3]
1510 LET Z2=N9+1
1520 LET N9=Z2
1530 GOTO 1630
1540 PRINT "НОМЕР ЦЕНТРАЛЬНОГО УЗЛА ЗВЕЗДЫ =";
1550 INPUT U
1560 LET Z2=0
1570 FOR I=1 TO N8
1580 LET J=Z[I,Z0+1]
1590 IF (FNA(J)-U)*(FNB(J)-U)#0 THEN 1620
1600 LET Z2=Z2+1

```

```

1610 LET K[Z2]=I
1620 NEXT I
1630 LET Z1=N-2
1640 LET Z3=N8
1650 LET Z4=100
1660 GOSUB 40 60
1670 LET N8=Z3
1680 PRINT "ТАБЛИЦА ПОСЛЕ ПРЕОБРАЗОВАНИЯ"
1690 GOSUB 9000
1700 GOTO 1220
1710 PRINT "РАСЧЕТ ДЛЯ ТОЙ ЖЕ СХЕМЫ ПОВТОРИТЬ (1-ДА,0-НЕТ)";
1720 INPUT N
1730 IF N=0 THEN 1760
1740 IF F=0 THEN 1080
1750 GOTO 1040
1760 PRINT "ВВЕСТИ НОВУЮ СХЕМУ (1-ДА,0-НЕТ)";
1770 INPUT N
1780 IF N=1 THEN 1030
1790 STOP
8900 REM *****
8902 REM * ПП ПЕЧАТЬ ТАБЛИЦЫ *
8904 REM *****
9000 IF Z0=2 THEN 9030
9010 PRINT "N ВЕТВИ", "N1.N2", "R"
9020 GOTO 9040
9030 PRINT "N ВЕТВИ", "N1.N2", "RE(Z)", "IM(Z)"
9040 PRINT "-----"
9050 FOR I=1 TO N8
9060 IF Z0=2 THEN 9090
9070 PRINT I, Z[I,2], Z[I,1]
9080 GOTO 9100
9090 PRINT I, Z[I,3], Z[I,1], Z[I,2]
9100 NEXT I
9110 PRINT "-----"
9120 RETURN
9999 END

```

Если в распоряжении пользователя имеется такая версия языка БЕЙСИК, которая предусматривает применение средств машинной графики, то целесообразно заменить ПП 9000 на ПП, которая на каждом этапе свертывания выводила бы на экран преобразованную (свернутую)

схему. Такой подход обеспечит большую наглядность результатов и ускорит принятие решения при выборе очередного шага, а следовательно, сократит и общее время вычислений.

После выбора пользователем очередного участка схемы и типа преобразования (что также должно быть введено в диалоговом режиме) программа вызывает соответствующие процедуры, производит вычисления в автоматическом режиме (в том числе и все пересылки входной и выходной информации для каждой ПП), размещает результаты преобразований в памяти и выдает сообщение о завершении данного этапа расчета. Процесс циклически повторяется и может быть остановлен в любой момент, что позволяет свертывать цепь не только до двухполюсника, но в случае необходимости (такие случаи часты в практике расчетов) и до многополюсника.

Для обеспечения работоспособности программы важное значение имеет правильное задание величины массивов. Описание размерности массивов в неявной форме для расчета на постоянном токе задается строкой

```
1000 DIM P(M1,2), Z(M2,2), K(M2)
```

и для переменного тока

```
1000 DIM P(M1,4), Z(M2,3), K(M2)
```

В эти строки при вводе программы с клавиатуры следует подставить нужные численные значения: $M1$ – число ветвей схемы, $M2 = 2 \times M1$.

Тестирование программы целесообразно провести по методам, изложенным в § 3.4.

Программа ПРЭСЗ сокращает долю ручного труда, значительно повышает комфортность работы пользователя и, наконец, является универсальной – применимой к схемам произвольной топологической сложности.

При многократном использовании ее для расчета – многовариантного анализа цепи с одной и той же топологией, но с разными параметрами ветвей (оптимизация, учет статистических факторов, расчет при разных частотах) – можно рекомендовать после проведения однократного расчета вывести на машинный носитель последовательность этапов свертывания и при последующих вычислениях использовать ее, исключив повторение каждый раз диалога. Той же цели можно достигнуть, записав последовательность свертывания в операторе DATA и несколько изменив текст программ.

▼ Перейдем к описанию программы ПРЭС4, основная особенность которой заключается в осуществлении в автоматическом режиме выбора алгоритма редукции без вмешательства оператора.

Ввод данных в ней осуществляется с помощью ПП 2000 (см. гл. 1). После запуска программы и ввода параметров ветвей и топологии це-

пи следует, отвечая на соответствующий запрос, ввести список узлов, которые не должны участвовать в преобразованиях. Если, например, ставится задача представить исходную схему двухполюсником, то вводятся номера только двух узлов, которые в процессе вычислений не сокращаются.

Алгоритм той части программ, которая производит автоматическое сворачивание схемы, построен следующим образом.

1. Выявляются последовательно соединенные ветви, и в каждом случае они заменяются одной эквивалентной ветвью с помощью ПП 4040 (при этом один или более узлов выпадают из дальнейшего рассмотрения).

2. Выделяются все параллельные ветви, и каждая группа параллельных ветвей заменяется с помощью ПП 4050 одной эквивалентной ветвью при сохранении общих узлов.

3. Процедуры по п. 1 и 2 выполняются до тех пор, пока не будут выявлены и эквивалентированы все подобные соединения.

4. После выполнения п. 1–3 проводится проверка условия окончания работы: достигнут ли заданный список узлов, относительно которых производится эквивалентирование. Если это условие выполняется, то работа по свертыванию заканчивается – производится переход на п. 7.

5. Если условие окончания работы не выполняется, то выявляется звезда с минимальным числом лучей (выявляется узел с минимальной степенью связности), которая преобразуется в многоугольник с помощью ПП 4060.

6. Производится переход на п. 1, и все действия повторяются.

7. При выходе из п. 4 работа по преобразованию схемы завершается, и выводятся результаты расчета.

Особенностью алгоритма является то, что в нем не используется преобразование треугольник–звезда. Это объясняется тем, что при использовании таблично представляемой информации о топологии схемы звезду выбрать легче, чем треугольник, так как во втором случае возникают значительные трудности с определением ветвей, образующих треугольник. Кроме того, встречаются цепи, расчет которых при использовании обоих преобразований (треугольник – звезда и звезда – треугольник) приводит к заикливанию программы.

Программа ПРЭС4 работает в режиме диалога, и пользователю остается лишь отвечать на вопросы, выдаваемые программой после ее иницирования.



```
980 REM *****
982 REM * ПРЕОБРАЗОВАНИЕ ЭЛЕКТРИЧЕСКИХ СХЕМ *
984 REM * ПРОИЗВОЛЬНОЙ ТОПОЛОГИИ МЕТОДОМ *
986 REM * РЕДУКЦИИ. ПОСТОЯННЫЙ И ПЕРЕМЕННЫЙ *
988 REM * ТОК. НАЗВАНИЕ: "ПРЭС4" . *
990 REM *****
```

```

1000 DIM Z[80,3],P[80,4],N[30],U[3]
1002 REM-----
1004 REM ВВОД ИСХОДНЫХ ДАННЫХ
1006 REM-----
1008 GOSUB 2000
1010 PRINT "ВВЕДИТЕ НОМЕРА УЗЛОВ (U), ОТНОСИТЕЛЬНО"
1012 PRINT "КОТОРЫХ НУЖНО ПРЕОБРАЗОВАТЬ СХЕМУ."
1014 PRINT "U=-1 ПОЗВОЛЯЕТ ИСПРАВИТЬ ОШИБКУ,"
1016 PRINT "ДОПУЩЕННУЮ НА ПРЕДЫДУЩЕМ ШАГЕ, А U=-2"
1018 PRINT "ОЗНАЧАЕТ СКОНЧАНИЕ ВВОДА."
1020 LET N3=0
1022 PRINT "ШАГ N";N3+1," U=";
1024 INPUT U
1026 IF U<0 THEN 1034
1028 LET N3=N3+1
1030 LET U[N3]=U
1032 GOTO 1022
1034 IF U=-2 THEN 1040
1036 LET N3=N3-1+SGN(1-SGN(N3))
1038 GOTO 1022
1040 PRINT "ЧИСЛО ВВЕДЕННЫХ УЗЛОВ РАВНО ";N3
1042 IF N3>1 THEN 1048
1044 PRINT "ОШИБКА: ЧИСЛО УЗЛОВ ДОЛЖНО БЫТЬ >1."
1046 GOTO 1010
1048 IF N9=2 THEN 1056
1050 PRINT "УКАЖИТЕ ГРАНИЦЫ ЧАСТОТЫ И ШАГ ПО ЧАСТОТЕ (В ГЦ)."

```

```

1082 NEXT I
1084 PRINT "-----"
1086 IF N9=2 THEN 1102
1088 LET F=F+F3
1090 IF F <= F2 THEN 1056
1092 PRINT "РАСЧЕТ ЗАКОНЧЕН."
1094 PRINT "НУЖНО ЛИ ЕГО ПОВТОРИТЬ ДЛЯ ДРУГОГО"
1096 PRINT "ДИАПАЗОНА ЧАСТОТ (1-ДА,0-НЕТ) ?"
1098 INPUT Z
1100 IF Z=1 THEN 1050
1102 STOP
9999 END

1490 REM *****
1492 REM * ВЫЧИСЛЕНИЕ RE(Z) И IM(Z) ВЕТВЕЙ *
1494 REM * МНОГОПОЛДНИКА ПРИ ЧАСТОТЕ F. *
1496 REM *****
1500 DEF FNA(A)=INT(ABS(A))
1502 DEF FNB(B)=INT((ABS(B)-FNA(B))*NO+.5)
1504 IF N9=3 THEN 1510
1506 LET F=0
1508 LET F1=0
1510 IF F>F1 THEN 1518
1512 PRINT "УКАЖИТЕ ЧИСЛО СТРОК МАССИВА Z."
1514 INPUT N8
1516 LET ZO=N9-1
1518 LET N2=N2+1
1520 FOR N=1 TO N2
1522 LET N[N]=0
1524 NEXT N
1526 FOR N=1 TO N3
1528 LET N[U[N]+1]=2
1530 NEXT N
1532 LET W=2*3.14159*F
1534 FOR N=1 TO N1
1536 LET P=P[N,1]
1538 LET N5=FNA(P)+1
1540 LET N6=FNB(P)+1
1542 LET N[N5]=N[N5]+1/NO
1544 LET N[N6]=N[N6]+1/NO
1546 LET Z[N,1]=P[N,2]
1548 IF N9=2 THEN 1556

```

```

1550 LET Z[N,2]=P[N,3]*W
1552 IF P[N,4]=0 THEN 1556
1554 LET Z[N,2]=Z[N,2]-1/W/P[N,4]
1556 LET Z[N,N9]=P[N,1]
1558 IF N5<N6 THEN 1562
1560 LET Z[N,N9]=N6-1+(N5-1)/NO
1562 NEXT N
1564 LET N5=N1
1566 LET N6=N2
1568 LET N7=0
1570 LET N4=0
1572 GOTO 1622
1574 LET K8=Z[K9+1,N9]-Z[K9,N9]
1576 IF K8>0 THEN 1622
1578 IF K8<0 THEN 1606
1580 LET Z1=K9
1582 LET Z2=Z1+1
1584 LET Z3=Z1
1586 LET Z4=N8
1588 GOSUB 4050
1590 LET Z=Z[K9,N9]
1592 LET K6=FNA(Z)+1
1594 LET K7=FNB(Z)+1
1596 GOSUB 1710
1598 LET Z[Z2,N9]=N2+N7
1600 LET N7=N7+1
1602 LET N4=K9
1604 GOTO 1624
1606 FOR N=1 TO N9
1608 LET Z=Z[K9,N]
1610 LET Z[K9,N]=Z[K9+1,N]
1612 LET Z[K9+1,N]=Z
1614 NEXT N
1616 IF K9=1 THEN 1622
1618 LET K9=K9-1
1620 GOTO 1574
1622 LET N4=N4+1
1624 LET K9=N4
1626 IF N4<N5 THEN 1574
1628 IF Z[N5,N9]<N2 THEN 1634
1630 LET N5=N5-1

```

```

1632 GOTO 1628
1634 IF N6=N3 THEN 1696
1636 LET N7=0
1638 LET N4=0
1640 LET K8=1
1642 IF N[K8]>1 THEN 1672
1644 LET K9=FNB(N[K8])
1646 IF K9<2 THEN 1672
1648 IF K9>2 THEN 1666
1650 GOSUB 1724
1652 LET Z1=K[1]
1654 LET Z2=K[2]
1656 LET Z3=Z1
1658 GOSUB 4040
1660 LET Z[Z2,N9]=(N2+N7)*(1+1/NO)
1662 LET N7=N7+1
1664 GOTO 1638
1666 IF N4=0 THEN 1670
1668 IF FNB(N[N4]) <= K9 THEN 1672
1670 LET N4=K8
1672 LET K8=K8+1
1674 IF K8 <= N2 THEN 1642
1676 IF N4*(1-SCN(N7))=0 THEN 1570
1678 LET K8=N4
1680 GOSUB 1724
1682 LET Z1=2
1684 LET Z2=K5
1686 LET Z3=N5
1688 LET Z4=N8
1690 GOSUB 4060
1692 LET N5=Z3
1694 GOTO 1570
1696 IF N5>1 THEN 1706
1698 LET Y1=Z[N5,1]
1700 LET Y2=0
1702 IF N9=2 THEN 1706
1704 LET Y2=Z[N5,2]
1706 LET N2=N2-1
1708 RETURN
1710 IF N[K6]>1 THEN 1716
1712 LET N[K6]=N[K6]-1/NO

```

```

1714 LET N6=N6-1+SGN(FNB(N[K6]))
1716 IF N[K7]>1 THEN 1722
1718 LET N[K7]=N[K7]-1/NO
1720 LET N6=N6-1+SGN(FNB(N[K7]))
1722 RETURN
1724 LET K5=FNB(N[K8])
1726 LET K4=K8-1
1728 LET K3=0
1730 LET K2=1
1732 LET K1=2[K2,N9]
1734 IF (FNA(K1)-K4)*(FNB(K1)-K4)*0 THEN 1750
1736 LET K3=K3+1
1738 LET K[K3]=K2
1740 LET K=FNA(K1)
1742 IF K=K4 THEN 1746
1744 LET K=FNB(K1)
1746 IF N[K+1]>1 THEN 1750
1748 LET N[K+1]=N[K+1]+(K5-2)/NO
1750 LET K2=K2+1
1752 IF K3=K5 THEN 1732
1754 LET N[K8]=0
1756 LET N6=N6-1
1758 RETURN

```

Описание всех массивов должно быть включено пользователем в текст программы до начала ее работы в том случае, если приведенное в строке 1000 определение размерностей окажется недостаточным.

Используемый алгоритм не позволяет в данной программе однозначно оценить связь топологической сложности анализируемой схемы с размерностью необходимых при вычислениях массивов. В общем случае можно рекомендовать следующие размерности: при расчете на постоянном токе

```
1000 DIM P(M1,2), Z(M2,2), N(M1), U(M3)
```

а при переменном токе так:

```
1000 DIM P(M1,4), Z(M2,3), N(M1), U(M3)
```

В этих строках вместо M1, M2, M3 можно подставить численные значения: M1 – число ветвей схемы, M2 = 2M1, M3 – число узлов схемы.

Проверить работоспособность программы можно по методике, изложенной в § 3.4.

▼ Программа ПРЭС4 обеспечивает полностью автоматизированный расчет на постоянном и переменном токах электрических цепей произволь-

ной топологии. Ее целесообразно усовершенствовать с учетом специфики вычислений. Так, отметим, что автоматический выбор последовательности эквивалентных преобразований требует значительных затрат времени (особенно для сложных схем с большим числом ветвей). Поэтому при многовариантном расчете одной и той же цепи целесообразно дополнить алгоритм следующими процедурами:

- 1) при первом расчете цепи запомнить всю найденную последовательность преобразований (разместить в оперативной памяти или на машинных носителях);
- 2) при последующих расчетах использовать эту последовательность в качестве готового алгоритма свертывания, организовав с этой целью считывание ее из памяти или с диска.

Большие возможности открываются и при использовании средств машинной графики, однако на них мы останавливаться не будем.

В заключение отметим, что для организации практических работ по анализу электрических цепей желательно иметь в библиотеке обе программы – и ПРЭС3, и ПРЭС4. Необходимость первой кажется не столь очевидной, так как ПРЭС4 обеспечивает более высокий уровень автоматизации вычислений. Однако практика проведения электрических расчетов показывает, что работа инженера-электрика содержит значительное число стадий, которые носят эвристический характер. В таких ситуациях более пригодными являются программы типа ПРЭС3, позволяющие корректировать процесс вычислений и менять направление их выполнения.

3.4. Методы тестирования

▼ В данном параграфе описаны методы тестирования программ расчета по методу редукции.

В настоящее время теория верификации программ (назначение которой – создание методов доказательства их правильности) не дает общих методов обоснования верности программ [36]. Одним из способов анализа работоспособности и повышения надежности математического обеспечения является так называемое многовариантное, или, в частном случае, дуальное, программирование. Оно состоит в том, что создаются два или более независимо написанных вариантов программы [37], в автоматическом режиме производится сравнение вырабатываемых ими выходных данных и принимается соответствующее решение. Аналогом такого подхода было применявшееся в гл. 2 сопоставление результатов расчета цепей одной и той же конфигурации с использованием программ для методов КТ и УП. (Пользователь может продолжить этот путь – включить, если позволяют объем оперативной памяти ПЭВМ, процедуру автоматического сравнения результатов двух программ с соответствующей диагностикой.)

Имея две программы (например, ПРЭС0 и ПРЭС1), возможно проверить вторую (как более сложную), рассчитав с ее помощью топологию

чески эквивалентную цепь на постоянном токе. Результаты таких расчетов должны совпадать в пределах точности вычислений для данной ЭВМ. А несовпадение результатов будет свидетельствовать о необходимости дальнейшей отладки.

Применим описанный метод к программе ПРЭС1. Возьмем в качестве исходных данных $R_1 = 6$ Ом, $R_2 = 2$ Ом, $R_3 = 1$ Ом, $R_4 = 5$ Ом, $R_5 = 4$ Ом, а $L_k = C_k = 0$ (рис. 3.1, а), заменим строки 1020–1060 на одну

```
1020 DATA 6,0,0,2,0,0,1,0,0,5,0,0,4,0,0
```

Проводя вычисления, отмечаем совпадение с результатом, полученным с помощью ПРЭС0.

▼ Перейдем ко второму способу тестирования. Он, как отмечалось ранее в § 2.4, состоит в том, что при подготовке исходных данных значения частоты и параметров ветвей выбираются так, чтобы для каждой k -й ветви выполнялось соотношение

$$\operatorname{Re}(Z_k) = \operatorname{Im}(Z_k). \quad (3.2)$$

Комплексное сопротивление ветви (если обозначить мнимую и действительную части как A_k) равно $A_k(1 + j)$.

В этом случае тестовый расчет существенно упрощается. Так, например, эквивалентное сопротивление двух последовательно соединенных ветвей $Z = (A_1 + A_2)(1 + j)$, а параллельно соединенных ветвей

$$Z = \frac{A_1 A_2}{A_1 + A_2} (1 + j).$$

Можно показать, что и в случае преобразования треугольник — звезда и многолучевая звезда — многоугольник вычисления производятся над коэффициентами A_k , а результат затем получается умножением на комплексное число $1 + j$.

Алгоритм редукции есть последовательность перечисленных эквивалентных преобразований. Поэтому, если известен результат расчета цепи на постоянном токе (обозначим его R_3), то нетрудно предсказать результат расчета на переменном токе

$$Z_3 = R_3(1 + j)$$

при условии, что элементы каждой ветви удовлетворяют (3.2). А последнее связывает четыре параметра: круговую частоту ω , активное сопротивление ветви R_k , индуктивность L_k и емкость C_k :

$$R_k = \omega L_k - 1/(\omega C_k). \quad (3.3)$$




Рис. 3.2. Тестовая схема в форме квадратной решетки

Три из них могут быть выбраны произвольно. Положим $C_K = 0$, круговую частоту $\omega = 1$ (т. е. $f = 1/(2\pi) = 0,159155$ Гц), и тогда $R_K = L_K$. Таким образом, достаточно выбрать только значение сопротивления, чтобы по нему найти все остальные параметры ветви.

С целью проверить ПРЭС1 принимаем $R_1 = 6$ Ом, $R_2 = 2$ Ом, $R_3 = 1$ Ом, $R_4 = 5$ Ом, $R_5 = 4$ Ом.

Расчет при этих данных дает $\text{Re}(Z_3) = 3,79775$ Ом и $\text{Im}(Z_3) = 3,79775$ Ом.

Если результаты тестирования оказались отрицательными, то следующий шаг — локализация ошибки. Для этого достаточно включить печать (вывод на экран) промежуточных результатов после каждой процедуры эквивалентных преобразований и определить тот этап расчета, после которого впервые возникает неравенство действительной и мнимой частей комплексного сопротивления — признак ошибки на этом этапе.

При тестировании программы ПРЭС3 целесообразно производить многократный счет для одной и той же схемы, вводя каждый раз иную последовательность этапов свертывания, а затем сравнить полученные результаты. При тестировании ПРЭС4 желательно ту же схему просчитать и с помощью ПРЭС3.

▼ Проверить правильность работы программ можно, проведя расчеты для схемы, эквивалентное сопротивление которой можно заранее подсчитать. В качестве такой рекомендуем использовать квадратную планарную решетку (рис. 3.2), все ветви которой имеют одинаковые активные и реактивные сопротивления. Эквивалентное сопротивление определяется относительно точек A и B на диагонали.

Исходные данные для такой тестовой схемы ввиду их большого количества лучше подготовить с помощью небольшой подпрограммы "ТЕСТ".



```

1994 REM *****
1996 REM * ТЕСТ (РЕШЕЧКАТАЯ СХЕМА) *
1998 REM *****
2000 PRINT "ВВЕДИТЕ ЧИСЛО УЗЛОВ РЕШЕТКИ"
2002 PRINT "КВАДРАТНОЙ ФОРМЫ, РАСПОЛОЖЕННЫХ"
2004 PRINT "ВДОЛЬ ОДНОГО ИЗ ОСНОВАНИЙ "
2006 PRINT "КВАДРАТА."

```

```

2008 INPUT N2
2010 LET NO=1000
2012 PRINT "ЗАДАЙТЕ ЗНАЧЕНИЯ ЭЛЕМЕНТОВ"
2014 PRINT "R,L,C ОДНОЙ ИЗ ВЕТВЕЙ РЕШЕТКИ."
2016 INPUT R,L,C
2018 LET N9=2+SGN(ABS(L)+ABS(C))
2020 LET N=0
2022 FOR K1=1 TO N2
2024 FOR K2=1 TO N2
2026 LET K3=(K1-1)*N2+K2-1
2028 LET K4=K3+1
2030 IF K2=N2 THEN 2034
2032 GOSUB 2050
2034 IF K1=N2 THEN 2040
2036 LET K4=K3+N2
2038 GOSUB 2050
2040 NEXT K2
2042 NEXT K1
2044 LET N1=N
2046 LET N2=N2*N2-1
2048 RETURN
2050 LET N=N+1
2052 LET P[N,1]=K3+K4/NO
2054 LET P[N,2]=R
2056 IF N9=2 THEN 2062
2058 LET P[N,3]=L
2060 LET P[N,4]=C
2062 RETURN

```

Ее следует присоединить к ГП вместо ПП 2000, описанной в гл. 1. Обращение к ней также производится строкой GOSUB 2000. Программа "ТЕСТ" запрашивает размер стороны решетки (т.е. число узлов на одной стороне), а затем вводит R , L , C одной ветви. После этого заполняется матрица P , в которой и хранятся все исходные данные. В ней же содержится указание, что свертывание производится относительно точек A и B , расположенных на диагонали (см. рис. 3.2). Сопротивление решетки равно $Z_N = K_N Z$, где Z — сопротивление (активное или комплексное) одной ветви решетки, коэффициент $K_N = 1; 1,5; 1,8571; 2,1470$ при $N=1, 2, 3, 4$ (N — число ветвей в стороне квадрата).

Описанная тестовая электрическая цепь может рассматриваться как верхняя граница топологической сложности электрических цепей с близким числом ветвей, так как у нее средняя степень вершины быстро

приближается с ростом размерности к 4, что заведомо больше средней степени вершины графов, которые встречаются в практических расчетах (2,5–3).

Результаты расчетов целесообразно также контролировать различными приближенными методами, в том числе с помощью методик, описанных в [38–39].

3.5. Частотные характеристики

Описанные в предыдущих параграфах программы позволяют решить ряд задач частотного анализа электрических цепей.

Если имеется программа для расчета эквивалентного сопротивления цепи, то для проведения частотного анализа достаточно к ней добавить: а) цикл по частоте; б) ввод начальной и конечной частоты, шага по частоте; в) вывод результатов в виде таблиц и графиков; г) некоторые ПП, например определения модуля и угла комплексного числа.

В программе ПРЭС4 предусмотрена возможность организации цикла по частоте, а ввод данных с помощью ПП 2000 перенесен в ГП. Если в этой программе, а также в ПРЭС3 предусмотреть запоминание порядка эквивалентных преобразований, то это позволит сократить время расчетов.

Однако целесообразно иметь универсальную программу, предназначенную для расчета частотных характеристик. Текст такой программы с названием ПРЭС5 приведен ниже.



```

490 REM *****
492 REM * ВЫЧИСЛЕНИЕ АФЧХ. ПРЭС5. *
494 REM * ПЕЧАТЬ ГРАФИКОВ И ТАБЛИЦ. *
496 REM * ОПРЕДЕЛЕНИЕ РЕЗОНАНСОВ. *
498 REM *****
500 DIM X[40,4],Y[4,2],K[60],U[2]
502 REM-----
504 REM НАСТРОЙКА ПРОГРАММЫ
506 REM-----
510 PRINT "ПРОГРАММА НАСТРОЕНА (1-ДА,0-НЕТ) ?"
512 INPUT X
514 IF X=1 THEN 544
516 PRINT "1. ЗАГРУЗИТЕ ПП 6550, 7210, 7510 И 2000."
518 PRINT "2. ВВЕДИТЕ ПОД НОМЕРАМИ 6571 И 6597"
520 PRINT " ОПЕРАТОР ' <N> GOSUB 950 ', "
522 PRINT " ). ВВЕДИТЕ ПОД НОМЕРАМИ 7215, 7223 И"
524 PRINT " 7259 ОПРАТОР ' <N> LET M9=X(M8,M7)' ."

```

```

526 PRINT "4. ПОДГОТОВЬТЕ ИП 1500, В КОТОРОЙ ВЫЧИСЛЯЕТСЯ"
528 PRINT " КОМПЛЕКСНОЕ СОПРОТИВЛЕНИЕ ДВУХПОЛЮСНИКА"
530 PRINT " ПРИ ЧАСТОТЕ F (ГЦ). РЕЗУЛЬТАТЫ НУЖНО"
532 PRINT " РАЗМЕЩАТЬ В Y1 (RE(Z)) И Y2 (IM(Z))."
534 PRINT " ПРЕОБРАЗОВЫВАТЬ СХЕМУ СЛЕДУЕТ ОТНОСИТЕЛЬНО"
536 PRINT " УЗЛОВ U(1) И U(2)."

```

```

608 REM-----
610 REM ВЫБОР ПРОЦЕДУРЫ
612 REM-----
614 PRINT "УКАЖИТЕ ВИД РАБОТЫ: 1-ПЕЧАТЬ ТАБЛИЦЫ,"
616 PRINT "2-ПЕЧАТЬ ГРАФИКОВ, 3-ПОИСК РЕЗОНАНСОВ,"
618 PRINT "0-ВВОД НОВЫХ ИСХОДНЫХ ДАННЫХ ИЛИ"
620 PRINT "ЗАВЕРШЕНИЕ РАБОТЫ."
622 INPUT X4
624 IF X4=0 THEN 786
626 IF X4=1 THEN 642
628 IF X4=2 THEN 682
630 IF X4=3 THEN 736
632 PRINT "НЕВЕРНО УКАЗАН КОД. ПОВТОРИТЕ ВВОД."
634 GOTO 622
636 REM-----
638 REM ПЕЧАТЬ ТАБЛИЦЫ
640 REM-----
642 PRINT "ВВЕДИТЕ НОМЕР ТАБЛИЦЫ:"
644 PRINT "1. ПЕЧАТЬ RE(Z),IM(Z),MOD(Z),ARG(Z),"
646 PRINT "2. ПЕЧАТЬ RE(Y),IM(Y),MOD(Y),ARG(Y)."

```

```

690 GOSUB 900
692 IF X5=5*INT(X5/5) THEN 712
694 LET MO=X5-5*SGN(1+SGN(X5-6))
696 LET M1=1
698 LET M2=X2
700 LET M3=1
702 LET M4=7
704 LET M5=15
706 LET M6=30
708 GOSUB 7210
710 GOTO 614
712 LET K1=X2
714 LET K2=41
716 LET K3=27
718 LET K4=1
720 GOSUB 7510
722 FOR X=1 TO X2
724 LET X[X,3]=SQR(X[X,1]*X[X,1]+X[X,2]*X[X,2])
726 NEXT X
728 GOTO 614
730 REM-----
732 REM ПОИСК РЕЗОНАНСОВ
734 REM-----
736 PRINT "УКАЖИТЕ ВИД РЕЗОНАНСОВ:"
738 PRINT "1.РЕЗОНАНСЫ НАПРЯЖЕНИЙ, 2.РЕЗОНАНСЫ ТОКОВ."
740 INPUT X5
742 GOSUB 900
744 PRINT "ЗАДАЙТЕ ТОЧНОСТЬ ОПРЕДЕЛЕНИЯ "
746 PRINT "РЕЗОНАНСНОЙ ЧАСТОТЫ FO В ГЦ."
748 INPUT YO
750 LET F=F1
752 LET X8=SGN(X[2,3]-X[1,3])
754 FOR X7=2 TO X2-1
756 LET X9=SGN(X[X7+1,3]-X[X7,3])
758 IF X8 >= X9 THEN 772
760 LET Y[1,1]=F
762 LET Y[2,1]=F+F3
764 LET Y[1,2]=X[X7-1,3]
766 LET Y[2,2]=X[X7+1,3]
768 GOSUB 6554
770 PRINT "FO =";Y1

```

```

772 LET X8=X9
774 LET F=F+P3
776 NEXT X7
778 GOTO 614
780 REM-----
782 REM ИЗМЕНЕНИЕ ВХОДНЫХ ПАРАМЕТРОВ
784 REM-----
786 PRINT "ВВЕДИТЕ НОМЕР ЗАДАНИЯ:"
788 PRINT "1. ИЗМЕНЕНИЕ СХЕМЫ,"
790 PRINT "2. ВВОД НОВЫХ ВЕРШИН ДВУХПОЛЮСНИКА,"
792 PRINT "3. ЗАДАНИЕ НОВОГО ДИАПАЗОНА ЧАСТОТ,"
794 PRINT "0. ЗАВЕРШЕНИЕ РАБОТЫ."
796 INPUT X
798 IF X=1 THEN 556
800 IF X=2 THEN 558
802 IF X=3 THEN 564
804 IF X=0 THEN 9999
806 PRINT "НЕВЕРНО УКАЗАН НОМЕР. ПОВТОРИТЕ ВВОД."
808 GOTO 796
890 REM *****
892 REM * ПЕРЕХОД ОТ Z К Y *
894 REM *****
900 LET X6=X5-1.5
902 LET X6=SGN(X6-4*(1-SGN(ABS(X4-2))))
904 LET X7=X0*X6
906 IF X7>0 THEN 924
908 FOR X=1 TO X2
910 LET X7=X[X,1]*X[X,1]+X[X,2]*X[X,2]
912 LET X[X,1]=X[X,1]/X7
914 LET X[X,2]=-X[X,2]/X7
916 LET X[X,3]=SQR(X[X,1]*X[X,1]+X[X,2]*X[X,2])
918 LET X[X,4]=ATN(X[X,2]/X[X,1])*X3
920 NEXT X
922 LET X0=-X0
924 RETURN
940 REM *****
942 REM * ВСПОМОГАТЕЛЬНАЯ ПП, *
944 REM * ВЫЗЫВАЕМАЯ ИЗ ПП 6510. *
946 REM *****
950 LET F=Y1
952 GOSUB 1500

```

```

954 LET Y2=SQR(Y1*Y1+Y2*Y2)
956 RETURN
1490 REM *****
1492 REM * ВЫЧИСЛЕНИЕ RE(Z) И IM(Z) *
1494 REM * ДВУХПОЛЮСНИКА ПРИ ЧАСТОТЕ F *
1496 REM *****
1500 LET W=2*3.14159*F
1502 LET Y1=R
1504 LET Y2=W*L
1506 IF C=0 THEN 1510
1508 LET Y2=Y2-1/W/C
1510 RETURN
1990 REM *****
1992 REM * ВВОД ПАРАМЕТРОВ СХЕМЫ *
1994 REM *****
2000 DATA 1,1.00000E-02,1.00000E-02
2002 READ R,L,C
2004 RESTORE
2006 RETURN

```

Опишем ее особенности. Ввод исходных данных, кроме топологии и R , L , C (параметров ветвей), включает указание номеров двух узлов $U(1)$ и $U(2)$, которые выбраны в качестве полюсов (относительно них сворачивается схема). Вводится также интервал частот, в котором проводится частотный анализ, и шаг по частоте.

Расчет схемы производится в ПП 1500. В ней методом редукции вычисляются при некоторой частоте F действительная и мнимая части комплексного сопротивления, размещаемые в $Y1$ и $Y2$. Затем эти результаты пересылаются в массив X .


Ввод информации осуществляется или в виде таблиц, или в виде графиков. В первом случае на печать (или дисплей) выводится полная информация об эквивалентном сопротивлении или проводимости (действительная и мнимая компоненты, модуль и аргумент в градусах). Во втором случае можно напечатать график частотной зависимости одной из четырех перечисленных величин или вывести годограф, дающий картину изменения сопротивления $Z = Z(F)$ или проводимости $Y = Y(F)$ двухполюсника в зависимости от частоты на комплексной плоскости. В связи с тем что точки на годографе располагаются неравномерно (например, наблюдается сгущение точек в области низких и высоких частот), при вычислениях с целью получить графическую картину можно задавать несколько частотных интервалов с разными шагами по частоте.

Кроме того, программа позволяет определять резонансные частоты в заданном частотном интервале: находятся частоты резонансов по напряжению или по току (локальные минимумы модуля эквивалентного сопротивления или проводимости). Для этого используется метод золотого сечения (см. ПП6554, ПП7210, ПП7510 в гл. 5).

Последний применяется следующим образом. Перед вызовом ПП определяется интервал частот, в котором модуль комплексного сопротивления (или проводимости) достигает минимального значения. Границы интервала находятся с помощью пошагового перебора элементов массива X (где в третьем столбце записаны значения модуля).

Погрешность определения резонансной частоты вводится с клавиатуры после появления на дисплее соответствующего запроса. В большинстве практических применений достаточно погрешность 0,1 Гц. Повышение требований к точности приводит к значительному возрастанию времени счета, что особенно заметно при сложных схемах с большим числом ветвей.

В целях проверки работы программы ПРЭС5 рассмотрим простейший пример расчета для одной ветви с последовательно соединенными R , L , C элементами ($R = 1$ Ом, $L = 0,01$ Гн, $C = 0,01$ Ф). Сначала была найдена зависимость модуля и аргумента комплексного сопротивления от частоты в интервале от 1 до 30 Гц с шагом в 1 Гц, затем определена частота резонанса, равная 15,9128 Гц (точное значение — 15,9155 Гц). После этого выводился годограф комплексной проводимости в том же интервале частот. В данном случае точки годографа должны лежать на окружности с радиусом $R/2$ и координатами центра $(R/2, 0)$.



УКАЖИТЕ ВИД РАБОТЫ: 1-ПЕЧАТЬ ТАБЛИЦЫ,
2-ПЕЧАТЬ ГРАФИКОВ, 3-ПОИСК РЕЗОНАНСОВ,
0-ВВОД НОВЫХ ИСХОДНЫХ ДАННЫХ ИЛИ
ЗАВЕРШЕНИЕ РАБОТЫ.

?2

ВВЕДИТЕ НОМЕР ЗАВИСИМОСТИ:

1.RE(Z), 2.IM(Z), 3.MOD(Z), 4.ARG(Z), 5.Z=Z(F),
6.RE(Y), 7.IM(Y), 8.MOD(Y), 9.ARG(Y), 10. Y=Y(F).
?10

Y(MAX)= .499649


Y(MIN)=-.499887

X(MAX)= .999888

X(MIN)= 3.96342E-03

DY= 3.84437E-02

DX= 2.48981E-02



УКАЖИТЕ ВИД РАБОТЫ: 1-ПЕЧАТЬ ТАБЛИЦЫ,
 2-ПЕЧАТЬ ГРАФИКОВ, 3-ПОИСК РЕЗОНАНСОВ,
 0-ВВОД НОВЫХ ИСХОДНЫХ ДАННЫХ ИЛИ
 ЗАВЕРШЕНИЕ РАБОТЫ.
 ?0

ВВЕДИТЕ НОМЕР ЗАДАНИЯ:
 1. ИЗМЕНЕНИЕ СХЕМЫ,
 2. ВВОД НОВЫХ ВЕРШИН ДВУХПОЛЮСНИКА,
 3. ЗАДАНИЕ НОВОГО ДИАПАЗОНА ЧАСТОТ,
 0. ЗАВЕРШЕНИЕ РАБОТЫ.
 ?0

READY

Вообще особенность анализа поведения электрических цепей в частотной области состоит в том, что вычисления по программе производятся многократно (не один десяток раз). Получив результаты расчета, пользователь анализирует их, выделяет наиболее интересные участки (диапазоны частот). Затем в выделенных областях расчет повторяется для той же схемы с иным шагом по частоте. И если требуется выделить или обнаружить какие-то специфические детали частотных характеристик, то расчеты многократно повторяются при разных шагах по частоте и в разных диапазонах. Кроме того, часто меняют один или несколько параметров ветвей, изменяют топологию схемы замещения (добавляя или уда-

ля отдельные ветви, что, например, соответствует присоединению или отключению фильтров или компенсирующих конденсаторных батарей в электроэнергетических системах. Такая методика хорошо сочетается с диалоговым режимом вычислений на ЭВМ, и именно на такой режим ориентированы описанные выше программы.

3.6. Длинные линии

▼ Цепи с распределенными параметрами – важный объект электротехнических расчетов. Например, при анализе стационарных режимов электроэнергетических систем необходимо определять параметры линий передач. В данном параграфе описаны ПП, нужные для подобных расчетов.

Предварительно приведем ПП для перехода от одной формы представления параметров, характеризующих линии, к другой.

▼ Вычисление параметров длинных линий основано на использовании операций с комплексными числами (арифметические операции, вычисление гиперболических функций комплексного переменного и др.). Эти ПП приведены в гл. 5 и используются при разработке описанных ниже программ.

Длинные линии характеризуются погонными или первичными параметрами R_0 , L_0 , G_0 и C_0 – удельным сопротивлением (Ом/км), удельной индуктивностью (Гн/км), удельной проводимостью (См/км) и удельной емкостью (Ф/км). Другой формой их представления являются вторичные или характеристические параметры: волновое сопротивление

$$Z_B = \sqrt{(R_0 + j\omega L_0) / (G_0 + j\omega C_0)} \quad (3.4)$$

и коэффициент распространения

$$\gamma = \sqrt{(R_0 + j\omega L_0)(G_0 + j\omega C_0)}. \quad (3.5)$$

Формулы (3.4) и (3.5) позволяют переходить от первичных параметров длинной линии ко вторичным и реализованы в ПП 4600.



```

4590  RFM *****
4592  REM * ВЫЧИСЛЕНИЕ ВТОРИЧНЫХ ПАРАМЕТРОВ ЛИНИИ *
4594  REM * ПО ЗАДАНЫМ ПЕРВИЧНЫМ ПАРАМЕТРАМ И *
4596  REM * ЧАСТОТЕ. *
4598  REM *****
4600  LET W=2*3.14159*F
4602  LET Z1=K[3]
4604  LET Z2=Z1+1
4606  LET Z[Z1,1]=Z[K[1],1]
4608  LET Z[Z1,2]=W*Z[K[1],2]
    
```

```

4610 LET Z[Z2,1]=Z[K[1]+1,1]
4612 LET Z[Z2,2]=W*Z[K[1]+1,2]
4614 LET Z3=K[2]
4616 GOSUB 4246
4618 LET Z3=Z3+1
4620 GOSUB 4226
4622 LET Z1=Z3
4624 GOSUB 4410
4626 LET Z1=Z1-1
4628 LET Z3=Z1
4630 GOSUB 4410
4632 RETURN
4640 REM *****
4642 REM * ВЫЧИСЛЕНИЕ ПЕРВИЧНЫХ ПАРАМЕТРОВ ЛИНИИ *
4644 REM * ПО ЗАДАНЫМ ВТОРИЧНЫМ ПАРАМЕТРАМ И *
4646 REM * ЧАСТОТЕ . *
4648 REM *****
4650 LET W=2*3.14159*F
4652 LET W1=1-SGN(ABS(W))+W
4654 LET Z1=K[1]+1
4656 LET Z2=K[1]
4658 LET Z3=K[2]
4660 GOSUB 4226
4662 LET Z[Z3,2]=Z[Z3,2]/W1
4664 LET Z3=Z3+1
4666 GOSUB 4246
4668 LET Z[Z3,2]=Z[Z3,2]/W1
4670 RETURN
4672 REM *****
4674 REM * ВЫЧИСЛЕНИЕ СОПРОТИВЛЕНИЙ ХОЛОСТОГО ХОДА *
4676 REM * И КОРОТКОГО ЗАМЫКАНИЯ ЛИНИИ ПРИ *
4678 REM * ИЗВЕСТНЫХ ВТОРИЧНЫХ ПАРАМЕТРАХ И *
4680 REM * ДЛИНЕ ЛИНИИ . *
4682 REM *****
4690 LET Z1=K[2]
4692 LET Z[Z1,1]=Z[K[1]+1,1]*D
4694 LET Z[Z1,2]=Z[K[1]+1,2]*D
4696 LET Z3=Z1+1
4698 LET Z2=K[3]
4700 LET Z4=Z2+1
4702 GOSUB 4510

```

```

4704 LET Z1=K[1]
4706 LET Z2=K[2]+1
4708 LET Z3=K[2]
4710 GOSUB 4246
4712 LET Z3=Z3+1
4714 GOSUB 4226
4716 RETURN
4718 REM *****
4720 REM * ВЫЧИСЛЕНИЕ ВТОРИЧНЫХ ПАРАМЕТРОВ ЛИНИИ *
4722 REM * ПРИ ИЗВЕСТНЫХ ЗНАЧЕНИЯХ СОПРОТИВЛЕНИЙ *
4724 REM * ХОЛОСТОГО ХОДА И КОРОТКОГО ЗАМЫКАНИЯ. *
4726 REM * УКАЗЫВАЕТСЯ ТАКЖЕ ДЛИНА ЛИНИИ, *
4728 REM * ЧАСТОТА И ОРИЕНТИРОВОЧНОЕ ЗНАЧЕНИЕ *
4730 REM * ФАЗОВОЙ СКОРОСТИ . *
4732 REM *****
4740 LET Z1=K[1]+1
4742 LET Z2=K[1]
4744 LET Z3=K[2]
4746 GOSUB 4226
4748 LET Z3=Z3+1
4750 GOSUB 4246
4752 LET Z1=Z3
4754 GOSUB 4410
4756 GOSUB 4530
4758 LET Z[Z3,1]=Z[Z3,1]/D
4760 LET P=3.14159
4762 LET K=INT(2*F*D/V-Z[Z3,2]/P+.5)
4764 LET Z[Z3,2]=(Z[Z3,2]+P*K)/D
4766 LET Z3=Z3-1
4768 LET Z1=Z3
4770 GOSUB 4410
4772 RETURN

```

До вызова ПП следует записать удельные параметры линии в двух произвольных смежных строках массива Z, причем R_0 и L_0 нужно разместить в первой из них (R_0 в первом столбце, L_0 во втором), а G_0 и C_0 — во второй строке. Кроме того, нужно занести в массив K номера строк исходных данных, вспомогательных строк и строк массива Z, в которых будут размещаться результаты вычислений:

K(1) — номер первой строки исходных данных в массиве Z;

K(2) — номер первой из двух строк результата;

K(3) — номер первой из двух вспомогательных строк.

Таблица 3.1

№ строки массива Z	Элемент массива Z	
	I столбца	II столбца
K(1)	R_0	L_0
K(1) + 1	G_0	C_0
...
K(2)	$\text{Re}\{Z_B\}$	$\text{Im}\{Z_B\}$
K(2) + 1	$\text{Re}\{\gamma\}$	$\text{Im}\{\gamma\}$
...
K(3)	Промежуточные результаты	
K(3) + 1	Промежуточные результаты	

Таблица 3.2

№ строки массива Z	Элемент массива Z	
	I столбца	II столбца
K(1)	$\text{Re}\{Z_B\}$	$\text{Im}\{Z_B\}$
K(1) + 1	$\text{Re}\{\gamma\}$	$\text{Im}\{\gamma\}$
...
K(2)	R_0	L_0
K(2) + 1	G_0	C_0

По окончании работы ПП в строке с номером K(2) будут размещены действительная и мнимая части волнового сопротивления (1-й столбец – действительная часть), а в строке K(2) + 1 – коэффициент распространения.

Информация в массиве Z будет размещаться так, как это показано в табл. 3.1.

В (3.4) и (3.5) входит круговая частота, поэтому перед вызовом ПП 4600 надо присвоить переменной F значение частоты в герцах.

Приведенный выше способ размещения информации является универсальным и используется во всех остальных описываемых в данном параграфе программах. Независимое размещение информации в разных строках массива Z обеспечивает большую гибкость программ.

▼ Обратный переход от вторичных параметров к первичным можно осуществить, если основываться на уравнениях

$$R_0 + j\omega L_0 = Z_B \gamma; \quad G_0 + j\omega C_0 = \gamma / Z_B.$$

Для реализации этого перехода применяется ПП 4650. Перед ее вызовом следует произвести присвоения аналогично тому, как это описано выше для ПП 4600, однако в этом случае вспомогательные строки в массиве Z не используются.

Размещение информации в массиве Z должно соответствовать показанному в табл. 3.2.

Иногда длинная линия представляется комплексными сопротивлениями холостого хода и короткого замыкания

$$Z_X = Z_B \text{cth}(\gamma x); \quad Z_K = Z_B \text{th}(\gamma x),$$

где x – длина линии.

Вычислить эти параметры можно с помощью ПП 4690, для которой исходными данными служат Z_B и γ , записанные в массиве Z так, как

Таблица 3.3

№ строки массива	Содержание элементов	
	I столбца	II столбца
K(1)	$\text{Re}\{Z_B\}$	$\text{Im}\{Z_B\}$
K(1) + 1	$\text{Re}\{\gamma\}$	$\text{Im}\{\gamma\}$
...
K(2)	$\text{Re}\{Z_X\}$	$\text{Im}\{Z_X\}$
K(2) + 1	$\text{Re}\{Z_K\}$	$\text{Im}\{Z_K\}$
...
K(3)	Промежуточные результаты	

Таблица 3.4

№ строки массива	Содержание элементов	
	I столбца	II столбца
K(1)	$\text{Re}\{Z_X\}$	$\text{Im}\{Z_X\}$
K(1) + 1	$\text{Re}\{Z_k\}$	$\text{Im}\{Z_k\}$
...
K(2)	$\text{Re}\{Z_B\}$	$\text{Im}\{Z_B\}$
K(2) + 1	$\text{Re}\{\gamma\}$	$\text{Im}\{\gamma\}$

это показано в табл. 3.3, а также длина линии в километрах, которая заносится в *D*.

Для решения обратной задачи — определения γ и Z_B — недостаточно знать только Z_X и Z_K , а также x . Помимо этого следует использовать значение частоты и ориентировочное значение фазовой скорости, что обусловлено неоднозначностью решения уравнений, в которые входят гиперболические функции комплексных переменных. Поэтому при применении ПП 4740 в качестве исходных данных должна фигурировать также фазовая скорость, которая присваивается переменной *V*.

Размещение информации для этого случая показано в табл. 3.4.

Все три вида описания длинных линий являются эквивалентными, и их выбор определяется лишь постановкой задачи расчета. Например, из опытов холостого хода и короткого замыкания можно получить Z_X и Z_K , а затем рассчитать удельные параметры линии.

▼ В энергетике часто производят вычисления со схемами замещения для цепей, содержащих длинные линии, к которым присоединена нагрузка. В целях получения схемы замещения можно линию вместе с нагрузкой заменить эквивалентным сопротивлением по формуле

$$Z = Z_X \frac{Z_H + Z_K}{Z_H + Z_X},$$

где Z_H — сопротивление нагрузки. Соответствующие вычисления реализуются с помощью ПП 4790.



```

4774 REM *****
4776 REM * ВЫЧИСЛЕНИЕ ВХОДНОГО СОПРОТИВЛЕНИЯ ЛИНИИ *
4778 REM * ПРИ ЗАДАННЫХ СОПРОТИВЛЕНИЯХ ХОЛОСТОГО *
4780 REM * ХОДА, КОРТОКОГО ЗАМЫКАНИЯ И НАГРУЗКИ, *
4782 REM * ПОДКЛЮЧЕННОЙ В КОНЦЕ ЛИНИИ . *

```

```

4784 REM *****
4790 LET Z1=K[3]
4792 LET Z2=Z1+1
4794 LET Z[Z1,1]=Z[K[1]+1,1]+Z[K[1]+2,1]
4796 LET Z[Z1,2]=Z[K[1]+1,2]+Z[K[1]+2,2]
4798 LET Z[Z2,1]=Z[K[1],1]+Z[K[1]+2,1]
4800 LET Z[Z2,2]=Z[K[1],2]+Z[K[1]+2,2]
4802 LET Z3=K[2]
4804 GOSUB 4246
4806 LET Z1=K[1]
4808 LET Z2=Z3
4810 GOSUB 4226
4812 RETURN
4814 REM *****
4816 REM * ВЫЧИСЛЕНИЕ СОПРОТИВЛЕНИЙ Т-ОБРАЗНОЙ СХЕМЫ *
4818 REM * ЗАМЕЩЕНИЯ ЛИНИИ ПО ИЗВЕСТНЫМ ВТОРИЧНЫМ *
4820 REM * ПАРАМЕТРАМ И ДЛИНЕ ( ГИПЕРВОЛИЧЕСКИЕ *
4822 REM * ФУНКЦИИ ) . *
4824 REM *****
4830 LET KO=1
4832 LET Z1=K[2]
4834 LET Z[Z1,1]=Z[K[1]+1,1]*D
4836 LET Z[Z1,2]=Z[K[1]+1,2]*D
4838 LET Z2=Z1+1
4840 LET Z3=K[3]
4842 GOSUB 4470
4844 LET Z3=Z3+1
4846 GOSUB 4500
4848 LET Z[Z3,1]=Z[Z3,1]-1
4850 LET Z1=K[1]
4852 LET Z2=K[3]
4854 LET Z3=K[2]+2-KO
4856 IF KO=2 THEN 4862
4858 GOSUB 4246
4860 GOTO 4864
4862 GOSUB 4226
4864 LET Z1=Z3
4866 LET Z2=K[3]+1
4868 LET Z3=K[2]-1+KO
4870 IF KO=2 THEN 4876
4872 GOSUB 4226

```

```

4874 GOTO 4878
4876 GOSUB 4246
4878 RFTURN
4880 REM *****
4882 REM * ВЫЧИСЛЕНИЕ СОПРОТИВЛЕНИЙ П-ОБРАЗНОЙ СХЕМЫ *
4884 REM * ЗАМЕЩЕНИЯ ЛИНИИ ПО ИЗВЕСТНЫМ ВТОРИЧНЫМ *
4886 REM * ПАРАМЕТРАМ И ДЛИНЕ ( ГИПЕРБОЛИЧЕСКИЕ *
4888 REM * ФУНКЦИИ ) . *
4890 REM *****
4900 LET KO=2
4902 GOSUB 4832
4904 RETURN

```

Здесь исходные данные записываются в массив Z в порядке, показанном в табл. 3.5.

Таблица 3.5

№ строки массива Z	Элемент массива Z	
	I столбец	II столбец
K(1)	Re { Z _X }	Im { Z _X }
K(1) + 1	Re { Z _K }	Im { Z _K }
K(1) + 2	Re { Z _H }	Im Z _H
...
K(2)	Re { Z }	Im { Z }
...
K(3)	Промежуточные результаты	
K(3) + 1	Промежуточные результаты	

Таблица 3.6

№ строки массива Z	Элемент массива Z	
	I столбец	II столбец
K(1)	Re { Z _B }	Im { Z _B }
K(1) + 1	Re { γ }	Im { γ }
...
K(2)	Re { Z ₁ }	Im { Z ₁ }
K(2) + 1	Re { Z ₂ }	Im { Z ₂ }
...
K(3)	Промежуточные результаты	
K(3) + 1	Промежуточные результаты	

Если линию необходимо представить T- или П-образными схемами замещения, то следует воспользоваться ПП 4830 и ПП 4900.

Для T-образной схемы замещения, изображенной на рис. 3.3, а,

$$Z_1 = Z_B (\operatorname{ch}(\gamma x) - 1) / \operatorname{sh}(\gamma x); \quad (3.6)$$

$$Z_2 = Z_B / \operatorname{sh}(\gamma x), \quad (3.7)$$

где γ , Z_B и x соответствуют приведенным выше обозначениям.

Вычисления по (3.6) и (3.7) реализуются подпрограммой ПП 4830. До ее вызова следует заполнить (как и раньше) некоторые строки массивов Z и K в соответствии с табл. 3.6.




Рис. 3.3. Т- и П-образные схемы замещения длинных линий

Для П-образной схемы замещения, изображенной на рис. 3.3, б,

$$Z_1 = Z_B \operatorname{sh}(\gamma x); \quad Z_2 = Z_B \operatorname{sh}(\gamma x) / (\operatorname{ch}(\gamma x) - 1).$$

Для нее используется ПП 4900, обращение к которой требует подготовки массива Z аналогично тому, как это показано в табл. 3.6.

Пример использования описанных в данном параграфе подпрограмм приведен в приложении 1.

Глава четвертая

СТАТИСТИЧЕСКИЙ АНАЛИЗ КАЧЕСТВА

4.1. Метод статистического моделирования

▼ В данной главе описывается применение программ на языке БЕЙСИК для реализации метода статистического моделирования (статистических испытаний или метод Монте-Карло). Вообще при моделировании на ПЭВМ, как отмечается в [40, 41], этот язык программирования широко используется, так как обеспечивает удобство работы пользователя, наглядное отображение получаемой информации, легкость изменения хода вычислений и структуры модели, сокращает время проведения расчетов.

Метод статистического моделирования (для краткости в дальнейшем будем называть его метод МК) применяется при исследовании объектов, в описании которых играет существенную роль статистическая вариация и которые включают в себя вероятностные процессы. Большинство электротехнических устройств, изделий, аппаратов и машин могут быть отнесены к этому классу. Так, например, свойства электротехнических материалов (сопротивление, индукция, электрическая прочность и другие физические величины, характеризующие их) имеют значительный статистический разброс [42, 45, 58]. Электрические аппараты и машины выпускаются в серийном производстве с большими колебаниями выходных параметров (таких, как тяговое усилие, время срабатывания, пусковой момент, ток холостого хода и др.). Электрические сети

и системы, как в целом, так и их отдельные элементы, также характеризуются значительными масштабами колебаний нагрузок, параметров, режимов работы, вероятностная природа которых уже давно привлекала внимание исследователей [43]. Причем чем более сложна, точна, чем более совершенна конструкция электротехнических изделий, чем в более сложных условиях они работают, чем выше требования к ним, тем нагляднее, явственнее проявляются случайные колебания их свойств. Все это обостряет внимание и интерес разработчиков, изготовителей и исследователей к особенностям формирования и учета влияния статистических закономерностей свойств, параметров и характеристик электротехнических изделий.

▼ Одним из средств анализа статистических явлений является метод МК. В данной главе он применяется для исследования статистической вариации характеристик электротехнических объектов на примере анализа разброса параметров электрических цепей, хотя область его применения, без сомнения, много шире.

Метод МК как прием анализа качества серийной продукции и расчета неоднородности ее свойств пока мало применяется в электротехнике в отличие, например, от электроники и радиоэлектроники, где статистический анализ схем активно используется при решении большого круга практических и теоретических задач и где этот анализ рассматривается как целое самостоятельное направление [44]. Поэтому приведем сначала краткое описание этого метода (его можно рассматривать лишь как введение в метод), а затем в последующих параграфах опишем программы для его реализации на языке БЕЙСИК.

Различия в свойствах "одинаково" изготовленных изделий проявляются в конечном итоге в производственном браке. По [45] среди малогабаритных трансформаторов из стали 3421 (Э340) брак достигает 3,5%, а импульсных трансформаторов — 9%. Значительно разнятся тяговые характеристики электромагнитных реле, изготовленных по одинаковой технологии. Разброс выходных характеристик электрических машин постоянного тока и маломощных асинхронных двигателей достигает 2—10% номинальных значений, значительны также колебания КПД для электродвигателей разного исполнения и назначения.

▼ Неоднородность выпускаемых партий изделий по техническим характеристикам, параметрам, показателям качества возникает в силу действия ряда причин. Во-первых, при изготовлении любых электротехнических изделий неизбежными являются колебания режимов технологического процесса. Во-вторых, в условиях массового производства неизбежны также колебания качества сырья, полуфабрикатов и комплектующих. Все это формирует значительный разброс свойств, неоднородность партий продукции по качеству (которые не всегда могут быть выявлены ограниченным выходным контролем). Неповторяемость, возникающая на стадии технологии, отражается не только в разбросе средних

значений параметров, но и в различном поведении изделий при эксплуатации. Даже если режимы эксплуатации изделий одинаковы, а сами изделия при выпуске заводом-изготовителем имеют одинаковые номиналы, тем не менее у них могут быть разные коэффициенты старения. А это приводит к тому, что партия однородных изделий со временем становится неоднородной, возникают различия в сопротивлении, емкости, токах холостого хода, времени срабатывания и т.п. В-третьих, сами условия эксплуатации крайне разнообразны, и по этой причине также создаются условия для неодинакового изменения технических характеристик и параметров, для отклонения их от номиналов.

Для того чтобы регламентировать определенный уровень некоторого свойства Y , в производстве устанавливают предельные допустимые отклонения для него – нижнюю Y_1 и верхнюю Y_2 границы. Они определяют ширину поля допуска $Y_2 - Y_1$ и его полуширину $\Delta Y = (Y_2 - Y_1)/2$, и требования к уровню Y задаются в виде $Y_{\text{ном}} \pm \Delta Y$, где $Y_{\text{ном}}$ – номинальное значение данного свойства. Если Y попадает в поле допуска, изделие признается годным, в противном случае оно бракуется. (Возможно много частных случаев, которые мы не рассматриваем.)

Для установления численных значений границ Y_1 и Y_2 в производственных условиях проводится достаточно большое число измерений, из полученного массива выбираются минимальное и максимальное значения, которые и назначаются в качестве границ допуска. Эта весьма упрощенная процедура может быть улучшена, если считать, что распределение Y близко к нормальному: тогда находится среднеквадратичное отклонение σ и принимают $\Delta Y = k\sigma$. Так, например, при $k = 2$ он равен 95,45%. Выбор k не может быть произвольным и, с одной стороны, нуждается в комплексном обосновании с учетом и технических, и экономических сторон вопросов. С другой стороны, сама процедура эмпирического определения допуска более сложна.

Однако чаще допуск требуется установить еще на стадии проектирования изделия. Один из методов основан на положениях теории вероятностей. Принимается, что свойство (выходное свойство) Y есть функция n независимых аргументов (факторов)

$$Y = f(X_1, \dots, X_n), \quad (4.1)$$

каждый из которых в частном случае имеет нормальное распределение. Находят дисперсию

$$D(Y) = \sum_{i=1}^n \left(\frac{\partial Y}{\partial X_i} \right)^2 D(X_i), \quad (4.2)$$

выраженную через дисперсии $D(X_i)$ аргументов и частные производные $\partial Y/\partial X_i$.

Затем определяется $D(Y)$ и выбирается коэффициент k . Так, например, в [45] рассматривалось в качестве выходного контролируемого параметра напряжение на вторичной обмотке малогабаритного трансформатора

$$U = 4,44f wBS, \quad (4.3)$$

где f – частота (она принимается постоянной, не подверженной случайным вариациям); w – число витков; B – индукция; S – поперечное сечение витого магнитопровода:

$$S = mhb, \quad (4.4)$$

здесь m – число витков магнитной ленты; b и h – толщина и ширина ленты.

Далее на основании экспериментально определенных значений дисперсии аргумента находится по (4.2) дисперсия U и устанавливается допуск.

▼ Изложенный аналитический метод расчета допуска (описанный во многих источниках – см., например, [44, 46]) применим при наложении определенных ограничений как на функцию (4.1), так и к форме кривых распределения независимых факторов, и эти условия далеко не всегда выполняются при проведении практических расчетов. В отличие от этого метод МК в приложении к данной задаче свободен от таких ограничений и позволяет проводить расчеты для произвольных функций (4.1) и произвольных видов распределения аргументов и, следовательно, охватить более широкий класс объектов.

Поясним его содержание на примере рассмотренного выше малогабаритного трансформатора. Подставим в (4.3) численное значение частоты (как и раньше, она может рассматриваться как постоянная величина) и случайные значения w , B и S , которые от изделия к изделию изменяются под воздействием многих неуправляемых и неучтенных причин. Вычисление по (4.3) дает первое значение U_1 . Затем новое сочетание f , w , B , S подставляется в (4.3) и находится U_2 . Расчет повторяется много раз (иногда 1000 и более), каждое его повторение имитирует процесс изготовления одного трансформатора, технологического процесса, где каждое изделие имеет свои, "индивидуальные" значения числа витков, индукции сердечника, толщины и ширины ленточного материала, из которого изготавливается витой магнитопровод. Эти величины (технологические факторы) независимы, они подвержены случайным колебаниям, и поэтому контролируемая результирующая величина – напряжение на вторичной обмотке, по которой оценивается качество изделия (разумеется, это не единственная характеристика качества трансформатора), – также подвержена случайным вариациям – наблюдается разброс U между U_{min} и U_{max} . А эти величины могут использоваться для оценки поля допуска.

Таким образом, в данном случае метод МК состоит в многократном повторении одинакового расчета – в вычислении по одной и той же

формуле (называемой моделью), но каждый раз набор аргументов принимается разным, берутся случайные их комбинации. (О том, как последнее осуществляется, рассказано в § 4.2 и 4.3.)

Этот метод является, как говорят, "чисто машинным" — он может быть реализован только на ЭВМ, так как при том огромном объеме вычислений, который требуется для его выполнения, расчеты вручную осуществить невозможно. Следующие параграфы поэтому посвящены описанию алгоритмов и программ на языке БЕЙСИК для вычислений по методу МК.

4.2. Программы для метода МК

▼ Имея программу расчета электрической цепи, можно задать вопрос: как изменяются токи в ветвях, напряжения в узлах или общее ее сопротивление при изменении одного из параметров элементов ветви, например, на 1, 5 или 10%. Например, для схемы рис. 2.1 выясним, как изменится ее общее сопротивление R при увеличении сопротивления отдельных резисторов R_i на 10%. Составим небольшую программу ПВЧ, используя программу ПРЭСО из гл. 3 в качестве подпрограммы для расчета этой схемы на постоянном токе. Будем изменять каждый раз на 10% только один параметр, тогда как сопротивления остальных резисторов будут иметь постоянные номинальные значения, задаваемые оператором DATA (см. строку 9000 программы ПВЧ).

```

100 REM *****
102 REM * ОЦЕНКА ВЛИЯНИЯ ПРИРАЩЕНИЯ *
104 REM * СОПРОТИВЛЕНИЙ ОТДЕЛЬНЫХ ВЕТВЕЙ СХЕМЫ *
106 REM * НА ЭКВИВАЛЕНТНОЕ СОПРОТИВЛЕНИЕ. *
108 REM * НАЗВАНИЕ: 'ПВЧ' . *
110 REM *****
120 DIM R[5]
122 READ R[1],R[2],R[3],R[4],R[5]
124 GOSUB 1000
126 LET RO=R
128 PRINT "RO=";R
130 PRINT "ВВЕСТИ ПРИРАЩЕНИЕ АРГУМЕНТА (В %)"
132 INPUT N1
134 PRINT "N АРГ.", " R", "(R-RO)/RO*100"
136 LET N=5
138 FOR I=1 TO N
140 LET R1=R[I]
142 LET R[I]=R[I]*(1+N1/100)
144 GOSUB 1000

```



```

146 LET K=(R-RO)/RO*100
148 PRINT I,R,K
150 LET R[1]=R1
152 NEXT I
154 STOP
1000 REM *****
1002 REM * III 'ПРӨСО' . *
1004 REM *****
1010 LET R6=R[1]+R[3]+R[1]*R[3]/R[5]
1020 LET R7=R[3]+R[5]+R[3]*R[5]/R[1]
1030 LET R8=R[5]+R[1]+R[5]*R[1]/R[3]
1040 LET R9=R[2]*R6/(R[2]+R6)+R[4]*R7/(R[4]+R7)
1050 LET R=R8*R9/(R8+R9)
1060 RETURN
8900 REM --- ИСХОДНЫЕ ДАННЫЕ ---
9000 DATA 6,2,1,5,4
9999 END

```

В результате расчетов по ПВЧ получаем коэффициенты чувствительности или коэффициенты влияния — относительное изменение в процентах (от его номинального значения) при увеличении каждого отдельно взятого сопротивления R_i . Это позволяет оценить влияние последних на R (табл. 4.1).

Таблица 4.1

№ аргумента I	$\Delta R, \%$	№ аргумента I	$\Delta R, \%$
1	1,16	4	2,79
2	2,66	5	2,82
3	0,16		

В общем случае с помощью ЭВМ можно анализировать, как "чувствует" изменение тех или иных величин объект в целом, т.е. насколько сильно влияет изменение аргумента на выходную величину [46, 47], причем изменения аргумента могут быть или детерминированными (как в приведенном выше случае), или случайными. Реализуем второй случай — напишем программу: для схемы рис. 2.1 вместо номинального значения $R_3 = 1$ Ом будет задавать случайные равномерно распределенные значения в интервале от 0,5 до 1,5 Ом. И каждый раз будем определять общее сопротивление цепи R .

▼ Такой расчет приведем с помощью программы ПМК0.

```

100 REM *****
102 REM * ОЦЕНКА ВЛИЯНИЯ СЛУЧАЙНЫХ КОЛБЕБАНИЙ R3 *
104 REM * ПО РАВНОМЕРНОМУ ЗАКОНУ В ИНТЕРВАЛЕ .5-1.5 *
106 REM * НА ЭКВИВАЛЕНТНОЕ СОПРОТИВЛЕНИЕ ЦЕПИ R. *
108 REM * НАЗВАНИЕ: 'ПМК0' . *
110 REM *****
120 DIM R[5]
122 READ R[1],R[2],R[3],R[4],R[5]
124 GOSUB 1000
126 LET RO=R
128 PRINT "RO=";RO
130 PRINT "ВВЕДИТЕ ЧИСЛО ИСПЫТАНИЙ (ОБЪЕМ ВЫБОРКИ)."

```



В ней все факторы, т.е. сопротивления резисторов, кроме одного, — R_1 , R_2 , R_4 и R_5 — принимают значения, равные номиналам 6, 2, 5 и 4 Ом соответственно, тогда как R_3 благодаря строке 138 программы ПМК0, где используется датчик случайных чисел, изменяется случайным образом. И для каждого набора $R(1)$ вычисляется общее сопротивление

Таблица 4.2

№ п/п.	Случайные значения		$\Delta R, \%$
	$R_3, \text{ Ом}$	$R, \text{ Ом}$	
1	0,500015	3,76398	-0,889
2	1,000009	3,79776	$1,38 \times 10^{-4}$
.....			
199	1,04103	3,82051	0,599
200	1,003342	3,79796	$5,48 \times 10^{-4}$

ние R . Числа, вырабатываемые функцией БЕЙСИКА RND, как и любым программно реализованным датчиком случайных чисел (ДСЧ), следует, если быть точным, именовать псевдослучайными. Их совокупность можно рассматривать лишь как некоторую модель, с определенной степенью точности приближения имитирующую множество случайных чисел. В данной главе они называются случайными только для краткости. Более подробно о псевдослучайных числах, их генерации и свойствах см. в [48, 49].

Запустив программу ПМК0 на выполнение, получаем табл. 4.2, содержащую 200 случайных значений R .

Если просмотреть все 200 случайных значений R_3 , то можно отметить, что R_3 колеблется в пределах 0,500015–1,489999 Ом, чему соответствует варьирование R в пределах $-0,889 \div +0,721\%$ номинала $R_{\text{ном}} = 3,79775$ Ом, найденного при $R_3 = 1$ Ом.

По полученному массиву можно найти также дисперсию и другие статистические характеристики. Таким образом, если ПВЧ выявляет роль фактора при его детерминированном изменении, то ПМК0 позволяет оценить влияние случайных изменений отдельного фактора на повторяемость свойств изделия.

▼ Рассмотрим случай, когда не один, а все факторы имеют случайные колебания и моделируются программой ПМК1. В первой ее части вводятся пределы варьирования (нижняя и верхняя границы) для каждого резистора и объем выборки – число повторений расчета. Номинальные значения записаны в операторе DATA. Вторая часть (строки 150–158) циклически обращается к ПП 3000 и ПП 4000 и выводит результаты после каждого испытания (расчета, опыта). ПП 3000 при каждом обращении к ней присваивает пяти переменным случайные значения, подчиняющиеся равномерному закону. А последняя часть – это ПРЭС0, которая должна быть взята из гл. 3 и размещена в строках 1000–1020. Она реализует модель анализируемой цепи.

```

100 REM *****
102 REM * КОЛЕБАНИЯ СОПРОТИВЛЕНИЯ ЦЕПИ ПРИ СЛУЧАЙНЫХ *
104 REM * КОЛЕБАНИЯХ СОПРОТИВЛЕНИЙ В ВЕТВЯХ, *
106 REM * РАСПРЕДЕЛЕННЫХ ПО РАВНОМЕРНОМУ ЗАКОМУ. *
108 REM * НАЗВАНИЕ: 'ПМК1' . *
110 REM *****
120 DIM A[5],B[5],C[5],R[5]
122 REM M - ЧИСЛО АРГУМЕНТОВ
124 LET M=5
126 PRINT "УКАЖИТЕ ГРАНИЦЫ ИНТЕРВАЛОВ."
128 FOR J=1 TO M
130 PRINT "ДЛЯ АРГУМЕНТА N";J
132 INPUT A[J],B[J]
134 LET C[J]=B[J]-A[J]
136 NEXT J
138 PRINT "ВВЕДИТЕ ОБЪЕМ ВЫБОРКИ."
140 INPUT N
142 READ R[1],R[2],R[3],R[4],R[5]
144 GOSUB 1000
146 LET RO=R
148 PRINT "RO=";RO
150 FOR I=1 TO N
152 GOSUB 3000
154 GOSUB 1000
156 PRINT I;R;(R-RO)/RO*100
158 NEXT I
160 STOP
1000 REM *****
1002 REM * ПП 'ПРЭСО' . *
1004 REM *****
1010 REM . . .
1020 RETURN
2900 REM *****
2992 REM * ПРИСВОЕНИЕ СЛУЧАЙНЫХ ЗНАЧЕНИЙ АРГУМЕНТАМ *
2994 REM *****
3000 FOR J=1 TO M
3002 LET R[J]=A[J]+C[J]*RND(1)
3004 NEXT J
3006 RETURN
8900 REM --- ИСХОДНЫЕ ДАННЫЕ ---
9000 DATA 6,2,1,5,4
9999 END

```



При запуске программы пользователь вводит пределы варьирования для каждого фактора. Так, например, в протоколе расчета для всех переменных указаны колебания в пределах $\pm 50\%$ номинального значения.



ПРОТОКОЛ РАБОТЫ ПРОГРАММЫ ПМК1

УКАЖИТЕ ГРАНИЦЫ ИНТЕРВАЛА

ДЛЯ АРГУМЕНТА № 1 ? 3,9

ДЛЯ АРГУМЕНТА № 2 ? 1,3

ДЛЯ АРГУМЕНТА № 3 ? 0.5,1,5

ДЛЯ АРГУМЕНТА № 4 ? 2.5,7.5

ДЛЯ АРГУМЕНТА № 5 ? 2,7

ВВЕДИТЕ ОБЪЕМ ВЫБОРКИ

? 40

$\zeta 0 = 3.79775$

1 2.33698 -38.4642 %

2 4.23991 11.6425 %

...

39 3.46468 -8.77031 %

40 4.26486 12.2996 %

Программа ПМК1 по сравнению с ПМК0 более точно имитирует реальную ситуацию, с которой сталкиваются проектировщики и производственники, так как в ней предусмотрено одновременное случайное варьирование всех пяти независимых факторов. Однако в ней используется датчик случайных чисел, распределенных по равномерному закону. Следующим шагом приближения является использование ДСЧ с распределениями, встречающимися в практике. Об этом будет сказано в § 4.3.

Недостатком ПМК1 является то, что в практике применения статистических испытаний редко ставится задача — оценить только пределы варьирования функции отклика под действием случайных колебаний аргументов (факторов).

▼ Обычно задача ставится более широко — получить статистическое распределение функции и оценить ее параметры или иные вероятностные характеристики. Поэтому программа ПМК1 нуждается в дальнейшей

модификации. В частности, можно с ее помощью определять процент годных изделий P_r , определяемый по числу изделий, у которых выходной параметр попадает в поле допуска. Рассмотрим относительно простую задачу: вычислить методом статистических испытаний P_r , исходя из того, что качество изделий оценивается только по одной контролируемой величине и однозначно связано со свойствами комплектующих изделий (их параметрами или техническими характеристиками), причем свойства эти не зависят друг от друга.

Для решения этой задачи необходимо знать:

а) как распределены параметры (входные величины, свойства элементов, входящих в устройство) — $X_i, i = 1, \dots, n$;

б) как связаны эти величины с выходными (контролируемыми) свойствами Y готового изделия, по которым оценивается выход годных, т.е. зависимость $Y(X_1, \dots, X_n)$;

в) номинальное значение $Y_{ном}$, нижнюю Y_1 и верхнюю Y_2 границы для Y , такие, что при выполнении условия $Y_1 \leq Y \leq Y_2$ изделие признается годным.

Для схемы рис. 3.1, а, которую рассматриваем в качестве примера, X_i — это сопротивления пяти резисторов ($n = 5$). Функциональная связь в нашем случае задается алгоритмом вычисления, записанным в виде программы ПРЭСО.

Алгоритм вычислений состоит из следующих шагов:

- 1) выбрать пятерку случайных значений X_1, \dots, X_5 ;
- 2) вычислить и запомнить Y ;
- 3) повторить п. 1 и п. 2 N раз;
- 4) подсчитать число случаев, когда выполняется условие $Y_1 \leq Y \leq Y_2$.

Отсюда находится P_r .

Наиболее сложным является выполнение п. 1. Для этого надо знать закон распределения каждого X_i и иметь ДСЧ, выдающие величины, соответствующие этим законам.

Для простоты будем исходить из предположения, что все X_i имеют одинаковый — равномерный закон распределения со своими интервалами варьирования для каждой переменной. (Более сложные случаи рассмотрим позже.) Кроме того, примем, что все они варьируются в пределах $\pm 10\%$ своего номинального значения. Наконец, примем, что $Y_{ном} = 3,79775$ Ом (см. стр. 90), а в 5%-ный допуск на выходную величину Y дает границы $Y_1 = 3,60787$ Ом и $Y_2 = 3,98764$ Ом.

Описанный алгоритм вычислений реализуется программой ПМК2.

```

100 REM *****
102 REM * ПОДСЧЕТ ЧИСЛА ГОДНЫХ С УЧЕТОМ *
104 REM * КОЛЕБАНИЙ СОПРОТИВЛЕНИЙ В ВЕТВЯХ. *
106 REM * РАВНОМЕРНОЕ РАСПРЕДЕЛЕНИЕ. *
108 REM * НАЗВАНИЕ: 'ПМК2' . *

```



```

110 REM *****
120 DIM A[5],C[5],R[5],Y[250]
122 REM M - ЧИСЛО АРГУМЕНТОВ
124 LET M=5
126 REM N1 - ДОПУСК НА ФУНКЦИЮ ОТКЛИКА (%)
128 LET N1=5
130 REM N2 - ДОПУСК НА АРГУМЕНТЫ (%)
132 LET N2=10
134 REM ВВОД НОМИНАЛОВ И ПОДСЧЕТ ГРАНИЦ
136 FOR J=1 TO M
138 READ R[J]
140 LET A[J]=(100-N2)/100*R[J]
142 LET C[J]=2*N2/100*R[J]
144 NEXT J
146 GOSUB 1000
148 LET RO=R
150 LET R1=(100-N1)/100*RO
152 LET R2=(100+N1)/100*RO
154 PRINT "ГРАНИЦЫ ДЛЯ R:" ,R1,R2
156 PRINT "ВВЕДИТЕ ОБЪЕМ ВЫБОРКИ N"
158 PRINT "(N=0 ПОЗВОЛЯЕТ ЗАБЕРИТЬ ВЫЧИСЛЕНИЯ)."

```

```

3010 REM . . .
3020 RETURN
3090 REM *****
3092 REM * ПОДСЧЕТ ЧИСЛА ГОДНЫХ *
3094 REM *****
3100 LET T=0
3110 FOR I=1 TO N
3120 LET T=T+SGN(1-SGN(Y[I]-R1)*SGN(Y[I]-R2))
3130 NEXT I
3140 RETURN
8900 REM --- ИСХОДНЫЕ ДАННЫЕ ---
9000 DATA 6,2,1,5,4
9999 END

```

В ней используются ПП 1000 и ПП 3000, аналогичные тем, что приведены в тексте ПМК1. Строки 120–200 содержат ввод и подготовку данных. В структуре программы выделена ПП 3100 для подсчета частоты появления годных изделий, а в общем случае это может быть ПП для любой статистической обработки результатов моделирования.

Проведя вычисления по программе ПМК2 при указанных выше исходных данных при $N = 200$ получим $P_T = 89,2\%$.

▼ В состав математического обеспечения для метода МК должны входить: 1) программа – модель объекта или процесса; 2) датчики случайных чисел; 3) программы обработки результатов.

Математическая модель – это функциональная связь между выходной величиной (будем еще использовать термины “функция” и “функция отклика”) и одной или несколькими входными величинами (как синоним–аргументы, независимые факторы). Для схемы на рис. 2.1 аргументы – это сопротивления ветвей, функция – общее сопротивление схемы.

Математическая модель может быть задана в виде явной функции (4.1) или в виде алгоритма ее вычисления. (Последнее даже для простых объектов часто удобнее, чем использовать явную функцию.)

Случайные числа в соответствии с исходной информацией о распределении аргументов вырабатываются разными ДСЧ. При обращении к ним передаются параметры закона распределения, а на выходе получаем значение случайной величины.

Программы обработки результатов осуществляют накопление массивов, оценку статистических характеристик, частотные распределения и их печать.

Описанным частям МО должен предшествовать блок ввода данных (информация о законе распределения аргументов и объем выборки). Алгоритм программ достаточно прост: вырабатываются случайные чис-

ла, по ним находится значение функции, процесс циклически повторяется с накоплением массива, который по достижении нужного объема подвергается статистической обработке.

В большинство известных применений метода МК ставится задача вычисления математического ожидания случайной величины [48]. В отличие от этого при статистическом анализе (расчете) качества методом МК определяются процент годных, коэффициент вариации, дисперсия – величины, характеризующие разброс, вариативность, повторяемость.

К особенностям программ для метода МК относятся повышенные затраты машинного времени, и соответственно большее внимание при их разработке уделяется быстродействию и длительности вычислений, объему требуемой памяти (так как необходимо накапливать значительные объемы информации). О некоторых способах экономии машинных ресурсов см. в § 4.3.

В заключение отметим, что с более строгими определениями понятий и постановкой задачи метода МК можно ознакомиться по [44, 48, 51].

4.3. ДСЧ и обработка данных

▼ В данном параграфе остановимся лишь на некоторых типах ДСЧ, более детально с ними можно познакомиться по [48, 49].

Алгоритмы, на которых строятся ДСЧ, основаны на некоторых операциях с числами, равномерно распределенными в интервале (0,1). В языке БЕЙСИК во всех его более чем 40 версиях имеется функция RND – датчик случайных чисел, равномерно распределенных в интервале от 0 до 1. Не останавливаясь на способах проверки вероятностных характеристик вырабатываемых им чисел, укажем, что для целей технических расчетов, описываемых в данном параграфе, он вполне приемлем и удовлетворяет большинство предъявляемых требований.

Для того чтобы получить случайную выборку чисел равномерно распределенных в интервале от А до В следует ввести простое преобразование

$$X = A + (B - A) * \text{RND}(1)$$

▼ Наиболее часто в расчетах применяются нормально распределенные числа. На практике это обосновывается следующими соображениями. Все электротехнические свойства изделий формируются под влиянием большого числа факторов (до нескольких десятков), поэтому естественно предполагать, что в этом случае применима центральная предельная теорема теории вероятностей, а свойства изделий подчиняются нормальному закону. Разумеется, эти соображения требуют проверки с привлечением различных критериев согласия (хи-критерий, критерий Колмогорова и др.) по методикам, описанным в литературе по математической

статистике (см., например, [50] и др.). И хотя не во всех случаях предположение это проверяется и тем более подтверждается достаточно строго, однако нормальное распределение остается наиболее популярным в приложениях.

Получение нормально распределенных чисел возможно по различным алгоритмам. В общем случае [51] существуют два пути преобразования случайных равномерных чисел в нормальные:

- а) использование условий предельной теоремы теории вероятностей;
- б) прямое преобразование.

В первом случае используется центральная предельная теорема теории вероятностей: распределение суммы n независимых случайных величин с одинаковым законом распределения стремится к нормальному при возрастании n (более точные и строгие формулировки теоремы см. в [20, 51]). Если слагаемые X_i распределены равномерно в интервале $(0,1)$, то сумма таких чисел

$$S = \sum_{i=1}^n x_i$$

имеет распределение с математическим ожиданием $M = n/2$ и среднеквадратичным отклонением $\sigma = \sqrt{n/2}$.¹ Применяв преобразование по формуле

$$t = (m - S) / \sigma,$$

получим величины t , которые будут распределены по закону, приближающемуся с ростом n к стандартному нормальному с нулевым математическим ожиданием и единичной дисперсией.

Описанный алгоритм часто используется при расчетах на ЭВМ. Он реализован в ПП 6020.

```

6000 REM *****
6002 REM * ДАТЧИК СЛУЧАЙНЫХ ЧИСЕЛ, *
6004 REM * РАСПРЕДЕЛЕННЫХ ПО НОРМАЛЬНОМУ ЗАКОНУ *
6006 REM * С МАТЕМАТИЧЕСКИМ ОЖИДАНИЕМ M=0 И *
6008 REM * ДИСПЕРСИЕЙ D=1. *
6010 REM * НАЗВАНИЕ: 'ДСЧ1' . *
6012 REM *****
6020 LET N7=6
6022 LET M7=1/SQR(N7/12)
6024 LET S7=0
6026 FOR L7=1 TO N7
6028 LET S7=S7+RND(1)
6030 NEXT L7
6032 LET S7=(S7-N7/2)*M7

```



```

6034 RETURN
6040 REM: *****
6042 REM * ДАТЧИК СЛУЧАЙНЫХ ЧИСЕЛ, *
6044 REM * РАСПРЕДЕЛЕННЫХ ПО НОРМАЛЬНОМУ ЗАКОНУ. *
6046 REM * НАЗВАНИЕ: 'ДСЧ2' . *
6048 REM *****
6050 PRINT "ВВЕСТИ СРЕДНЕЕ И СР. КВ. ОТКЛОНЕНИЕ"
6052 INPUT M8,S8
6054 LET N7=6
6056 LET M7=S8/SQR(N7/12)
6058 LET S7=0
6060 FOR L7=1 TO N7
6062 LET S7=S7+RND(1)
6064 NEXT L7
6066 LET S7=M8+(S7-N7/2)*M7
6068 RETURN

```

В данной ПП нормальные числа получаются суммированием шести равномерно распределенных в интервале (0,1) случайных чисел. При ее использовании в программах для расчетов по методу МК строки 6020 и 6022 целесообразно вынести в ГП, что ускорит работу ПП. Этой же цели служит замена строк 6024–6030 на одну строку

```
6024 LET S7 = RND(1) + RND(1) + RND(1) + RND(1) + RND(1) + RND(1),
```

явно "нерациональный" вид которой очевиден лишь в языке БЕЙСИК, так как такая запись сокращает затраты времени на выработку одного случайного числа на 50% и более.

В датчике число слагаемых $N7$ может быть изменено так: а) при стремлении получить распределение, более близкое к нормальному, $N7$ увеличивают до 10–12; б) при желании сократить время счета $N7$ уменьшают (обычно до 4).

▼ В практике расчетов используют не стандартное нормальное распределение, а такое, у которого произвольные параметры — ненулевое математическое ожидание и дисперсия, не равная 1. Для этой цели можно использовать ПП ДСЧ2, построенную на основе ПП ДСЧ1. В ней применяется преобразование, позволяющее переходить от стандартного нормального распределения для чисел X к произвольному нормальному для чисел Y по формуле $Y = M8 + X * S8$, где $M8$ и $S8$ — математическое ожидание и среднеквадратичное отклонение распределения Y .

При единичном, разовом выполнении ПП ДСЧ2, отвечая на появляющийся запрос, следует ввести значения $M8$ и $S8$. А при включении ПП ДСЧ2 в ГП для вычислений по методу МК следует: а) исключить

строки 6050 и 6052, введя в ГП операции присвоения значений M8 и S8; б) строки 6052 и 6054 перенести в ГП; в) строки 6058–6064 заменить одной строкой, как это показано для ПП ДСЧ1.

Хотя нормальное распределение является одним из самых популярных, однако прежде чем его использовать (впрочем, это относится к любому теоретическому закону), необходимо провести довольно трудоемкую работу по проверке статистической гипотезы о том, что данное физическое свойство распределяется по закону, не отличающемуся от нормального. Имеются лишь отдельные случаи, когда этот вопрос решен достаточно убедительно. Для магнитных материалов, например, в [42] показано, что если для электротехнической стали индукция B_{25} распределена нормально, то свойства постоянных магнитов и ферритов имеют существенные отличия от нормального закона. В [58] показано, что большинство интересных для практических применений параметров полупроводниковых тиристоров подчиняется нормальному закону. ▼ Для получения случайных чисел Y_i с заданной плотностью вероятности $f(Y)$ следует решить относительно Y_i уравнение

$$\int_{-\infty}^{Y_i} f(Y) dY = X_i, \quad (4.5)$$

где X_i – равномерно распределенные в интервале (0,1) числа.

В [51] приведены точные решения для экспоненциального и релеевского законов. Однако в большинстве случаев таких решений нет, поэтому опишем способ, основанный на кусочной аппроксимации [44].

Будем исходить из того, что экспериментально получен массив частот F , каждый элемент которого F_i дает относительную частоту попадания Y в интервал (Y_i, Y_{i+1}) .

Задача состоит в том, что нужно разработать генератор случайных чисел, вероятность попадания которых в те или иные интервалы соответствует массиву F , т.е. при увеличении объема выборки получаемые частоты чисел, вырабатываемых этим генератором, асимптотически приближаются к заданному частотному распределению. Рассмотрим ее решение на трех примерах.

Когда число интервалов равно двум, им соответствуют частоты F_1 и F_2 , причем $F_1 + F_2 = 1$. Для того чтобы реализовать случайное появление одного из двух событий A и B (A – попадание в первый интервал, B – попадание во второй) с заданными частотами, достаточно иметь ДСЧ с равномерно распределенными числами в интервале (0,1) и весь этот интервал разбить на отрезки, равные F_1 и $1 - F_1$. Тогда фрагмент программы

```
10 INPUT F(1)
20 LET T = RND(1)
30 IF T > F(1) THEN GOTO
```




Рис. 4.1

```

40 PRINT "СОБЫТИЕ А"
50 GO TO 70
60 PRINT "СОБЫТИЕ В"
70 REM

```

реализует (разыгрывает) события A и B с заданными частотами. На рис. 4.1 приведена графическая иллюстрация этого алгоритма.

Если интервалов три и им соответствуют частоты $F_1 - F_3$, то надо по гистограмме рис. 4.2 (нижний график) построить дискретное интегральное распределение – найти накопленные частоты $X_1 = F_1$, $X_2 = F_1 + F_2$, $X_3 = F_1 + F_2 + F_3 = 1$.

Далее процедура та же, что и раньше: вырабатывается равномерно распределенное число, определяется, в какой из трех интервалов $(0, X_1)$, (X_1, X_2) и (X_2, X_3) оно попадает, на основании этого делается вывод о том, в какой из интервалов (Y_1, Y_2) , (Y_2, Y_3) или (Y_3, Y_4) попало число Y . Тогда Y приписывается значение полусуммы границ данного интервала.

Аналогично разыгрывается появление случайного числа Y , если его частотное распределение задано большим числом интервалов (на рис. 4.3 их пять).

▼ Таким образом, можно сформулировать общий алгоритм для программы, генерирующей случайные числа с заданным дискретным частот-





Рис. 4.2





Рис. 4.3

ным распределением: а) ввести интервалы Y и частоты для них; б) подсчитать нарастающие частоты (получить дискретное интегральное распределение); в) получить одно равномерно распределенное число X ; г) определить, в какой интервал попало это число (обозначим номер этого интервала k); д) приписать Y значение середины интервала с тем же номером.

Полученная величина $(Y_k + Y_{k+1})/2$ и будет искомым случайным значением для Y .

Текст программы, реализующей данный алгоритм, — ПП ДСЧЗ — приведен ниже.



```

900 REM *****
902 REM * ПОЛУЧЕНИЕ ВЫБОРКИ ИЗ N СЛУЧАЙНЫХ ЧИСЕЛ, *
904 REM * РАСПРЕДЕЛЕННЫХ В СООТВЕТСТВИИ С ЗАДААННЫМИ *
906 REM * ОТНОСИТЕЛЬНЫМИ ЧАСТОТАМИ F(I). *
908 REM * НАЗВАНИЕ: 'ПМКЗ' . *
910 REM *****
1000 DIM F[25],S[250],X[25],Y[25]
1002 PRINT "ВВЕДИТЕ ОБ'ЕМ ВЫБОРКИ (N)."

```

```

6078 REM * НАЗВАНИЕ: 'ДСЧЗ' . *
6080 REM *****
6090 LET T=RND(1)
6092 FOR I=1 TO N1
6094 IF T<X[I] THEN 6098
6096 NEXT I
6098 LET Y1=Y[I]-D2
6100 RETURN
9999 END

```

Программа ПМКЗ, вызывающая ПП ДСЧЗ, позволяет реализовать описанный выше алгоритм. В ГП сначала вводятся число интервалов, их ширина и нижняя граница Y_i , вычисляются верхние границы интервалов $Y_i = Y_i + kD_1$, где i – номер интервала. Затем вводятся частоты F_i для отдельных интервалов и вычисляются накопленные частоты:

$$X_{i+1} = X_i + F_i .$$

После N -кратного обращения к ПП ДСЧЗ случайные значения Y накапливаются в массиве S .

Для проверки работы ПМКЗ решим задачу получить выборку из 200 чисел, распределенных по нормальному закону с диапазоном варьирования $(-3,5, 3,5)$, математическим ожиданием $M = 0$ и дисперсией $D = 1$. Для этого вводится число интервалов $N1 = 10$, ширина интервала $0,7$ и нижняя граница $Y(1) = -3,5$. Затем вводятся частоты для каждого интервала, определенные по формуле

$$F_i = \frac{1}{\sqrt{2\pi}} \int_{Y_i}^{Y_{i+1}} e^{-t^2/2} dt.$$

Они приведены в табл. 4.3. По ним находятся накопленные частоты (массив X), соответствующие интегралу

$$X_i = \Phi_i(Y) = \frac{1}{\sqrt{2\pi}} \int_{-\infty}^{Y_{i+1}} e^{-t^2/2} dt.$$

Случайная равномерно распределенная величина T сравнивается последовательно со значениями X_i , и таким образом находится номер интервала, в котором находится число Y . Затем полученный массив Y подвергается частотному анализу – подсчитываются частоты K_i попадания Y в тот или иной интервал. Результаты для объема выборки $N = 200$ и $N = 1000$ приведены в двух последних столбцах табл. 4.3.

Программа ДСЧЗ является универсальной – она вырабатывает дискретные числа для произвольно заданного частотного распределения.

Таблица 4.3

№ интервала	Границы интервала		$F_i, \%$	K_i при	
				$N = 200$	$N = 1000$
1	-3,5	-2,8	0,2555	1,5	0,3
2	-2,8	-2,1	1,531	2,0	1,2
3	-2,1	-1,4	6,289	3,5	5,2
4	-1,4	-0,7	16,1	16,5	19,1
5	-0,7	0	25,8	29,5	26,2
6	0	-0,7	25,8	27,0	24,3
7	0,7	1,4	16,1	13,5	17,1
8	1,4	2,1	6,289	4,5	6,0
9	2,1	2,8	1,531	2,0	1,4
10	2,8	3,5	0,2555	0	0,2

Причем последнее может быть как эмпирического происхождения, так и может быть получено путем расчета по известному теоретическому закону распределения.

Отметим, что в описанном алгоритме использована кусочно-постоянная аппроксимация вероятностного распределения. Более точной является кусочно-линейная аппроксимация, которая используется в некоторых известных системах моделирования. Однако рассматривать ее не будем.

▼ Как отмечалось в § 4.2, математическое обеспечение для метода МК должно содержать программы обработки полученных в результате моделирования данных. Это чаще всего отдельные ПП для статистической обработки – вычисление процента годных, среднего (\bar{X}), дисперсии (D) и среднеквадратичного отклонения (σ). Алгоритмы этих ПП общеизвестны, поэтому далее приводятся их тексты с минимальными пояснениями.

ПП СТАТ1 вычисляет \bar{X} , D и σ для массива X длиной N (объем выборки равен N), подготовленного в ГП.



```

6110 REM *****
6112 REM * ВЫЧИСЛЕНИЕ СРЕДНЕГО, ДИСПЕРСИИ И *
6114 REM * СРЕДНЕКВАДРАТИЧНОГО ОТКЛОНЕНИЯ *
6116 REM * НАЗВАНИЕ: 'СТАТ1' . *
6118 REM *****
6120 LET S=0
6122 LET D=0
6124 FOR I=1 TO N
6126 LET S=S+X[I]
6128 NEXT I

```

```

6130 LET S=S/N
6132 FOR I=1 TO N
6134 LET D=D+(S-X[I])*(S-X[I])
6136 NEXT I
6138 LET D=D/(N-1)
6140 LET S1=SQR(D)
6142 PRINT " N", "СРЕДНЕЕ", "ДИСПЕРСИЯ", "СРЕДНЕКВ. ОТКЛ."
6144 PRINT N, S, D, S1
6146 RETURN

```

Весьма часто возникает необходимость в определении частотного распределения полученного массива результатов моделирования. Для подсчета частот по отдельным интервалам, на которые делится весь диапазон варьирования случайной величины X , используется ПП СТАТ2.

Она получает несгруппированный массив X из ГП, запрашивает нужное число интервалов (обычно 10–15), на которое надо разбить этот диапазон, находит нижнюю и верхнюю границы этого диапазона (минимальное и максимальное значения для массива X), ширину одного интервала и, наконец, подсчитывает частоты. На печать выводятся номер интервала, абсолютная и относительная (в процентах) частоты.

```

6150 REM *****
6152 REM * ОПРЕДЕЛЕНИЕ ЧАСТОТНОГО РАСПРЕДЕЛЕНИЯ . *
6154 REM * НАЗВАНИЕ: 'СТАТ2' . *
6156 REM *****
6160 PRINT "ВВЕДИТЕ ЧИСЛО ИНТЕРВАЛОВ."
6162 INPUT K1
6164 REM ПОИСК МАХ И МИН
6166 LET U=X[1]
6168 LET W=X[2]
6170 FOR I=1 TO N
6172 IF X[I]>U THEN 6176
6174 LET U=X[I]
6176 IF X[I]<W THEN 6180
6178 LET W=X[I]
6180 NEXT I
6182 LET D1=(W-U)/K1
6184 REM ОПРЕДЕЛЕНИЕ ЧАСТОТ
6186 FOR I=1 TO K1
6188 LET K[I]=0
6190 NEXT I
6192 FOR I=1 TO N

```



```

6194 LET M=INT((X[I]-U)/D1)+SGN(W-X[I])
6196 LET K[M]=K[M]+1
6198 NEXT I
6200 REM ПЕЧАТЬ АБСОЛЮТНЫХ И ОТНОСИТЕЛЬНЫХ ЧАСТОТ
6202 FOR I=1 TO K1
6204 PRINT I;K[I];K[I]/N*100;"(%)"
6206 NEXT I
6208 RETURN

```

▼ Задачи, приводящие к многовариантным расчетам, — нахождение оптимума, определение частотных характеристик, оценка разброса методом МК — являются наиболее трудоемкими, точнее ресурсоемкими, поскольку требуют и значительных затрат машинного времени, и много места в оперативной памяти. Поэтому немаловажное значение имеют различные способы экономии машинных ресурсов. Упомянем некоторые из них.

Метод МК требует постоянного обращения к ДСЧ. Каждое такое обращение к ПП, реализующей генератор случайных чисел, занимает на разных ЭВМ около 5–15 мс, если учитывать только вызов ПП и возвращение из нее в ГП. Число таких вызовов для задач средней сложности (например, при пяти аргументах в математической модели) и при объеме выборки около 1000 равно 5000 для получения только одной выборочной характеристики. Сократить время, затрачиваемое на эту процедуру, можно путем накопления случайных чисел в некотором вспомогательном массиве (100–250 шт.) и постепенного их использования по мере надобности. Это усложняет структуру программы, но позволяет несколько снизить затраты времени.

Того же можно достичь, если использовать при моделировании предварительное частичное замещение схем (метод подсхем) [52, 53]. Из схемы выделяется некоторая часть, для которой известно, что вероятностное распределение соответствующего параметра Y_1 или параметры этого распределения не меняются со временем. Для этой части — подсхемы — реализуется алгоритм МК и находится распределение (гистограмма) некоторой промежуточной величины Y_1 . Затем модель заменяется на упрощенную, и для нее снова проводятся статистические испытания. Таким образом, модель $Y = f(X_1, \dots, X_5)$ заменяется двумя: $Y_1 = f_1(X_1, X_2, X_3)$ и $Y_2 = f_2(Y_1, X_4, X_5)$, для каждой из них реализуется алгоритм МК, результаты, полученные на первой подмодели, многократно используются при моделировании на второй. При достаточно продуманном разделении и большом числе факторов этот прием позволяет снизить затраты времени на порядок, так как каждый раз, когда нужно очередное значение Y_1 , оно получается обращением к одному и тому же массиву, где записаны случайные значения Y_1 , или к ДСЧ,

моделирующему распределение чисел, соответствующее гистограмме Y_1 .

Можно повысить эффективность вычислительного процесса, если более рационально организовать моделирование, изменяя его ход в режиме диалога, производить адаптацию программ к специфическим свойствам решаемых задач на различных этапах проектирования в различные моменты времени [54].

При большом числе факторов допустимо предварительно оценить вклад каждого из них в суммарную дисперсию или вариацию функции отклика. Выделяются те из них, которые вносят незначительный вклад (порядка 1–2% каждый или 5–10% вместе). Предполагая, что такие факторы не подвержены статистическим флюктуациям (т. е. полагая, что их дисперсии равны нулю), можно без ущерба для точности сократить общий объем испытаний (а следовательно, и время расчетов) более чем на порядок. Такой подход требует предварительных пробных вычислений для оценки значимости аргументов. Это достаточно удобно провести, используя диалоговый режим расчета на ПЭВМ.

▼ Кроме больших затрат времени метод МК требует и накопления значительных массивов статистической информации для ее последующей обработки и анализа. Чтобы избежать этого, обычно используют интегральные характеристики (частота, моменты и др.). Однако в этом случае все-таки приходится затрачивать определенные усилия на размещение и выборку информации. Для того чтобы несколько сократить размерности используемых массивов, применяют (что особенно важно при работе на ПЭВМ с ограниченной оперативной памятью) прием декомпозиции процесса вычисления на части: после накопления $N = 100 \div 250$ значений функции отклика моделирование прерывается, вызывается ПП обработки данных, которая находит среднее, дисперсию и т. п. Затем весь процесс повторяется. А после каждого такого шага выборочные статистики усредняются.

Если этот прием использовать при $N = 1$, т. е. обновлять значения выборочных характеристик после каждого нового значения функции, то получим так называемый "пошаговый алгоритм вычисления статистик" (ПАВС), когда статистики вычисляются на каждом шаге машинного эксперимента.

ПАВС реализован в ПП СТАТЗ.

```
6210 REM *****
6212 REM * ВЫЧИСЛЕНИЕ СРЕДНЕГО, ДИСПЕРСИИ И *
6214 REM * СРЕДНЕКВАДРАТИЧНОГО ОТКЛОНЕНИЯ. *
6216 REM * ПОШАГОВЫЙ АЛГОРИТМ. *
6218 REM * НАЗВАНИЕ: 'СТАТЗ' . *
6220 REM *****
6230 LET S1=0
```



```

6232 LET S2=0
6234 FOR I=1 TO N
6236 LET S1=S1+X[I]
6238 LET S2=S2+X[I]*X[I]
6240 NEXT I
6242 LET S=S1/N
6244 LET D=(S2-N*S*S)/(N-1)
6246 LET S1=SQR(D)
6248 PRINT " N", "СРЕДНЕЕ", "ДИСПЕРСИЯ", "СРЕДНЕЕ ОТКЛ."
6250 PRINT N,S,D,S1
6252 RETURN

```

При вызове этой ПП в ГП подготавливается очередное значение некоторой величины X_i , для которой надо определить среднее, дисперсию и среднеквадратичное отклонение, это значение X_i передается в ПП СТАТЗ, где накапливаются суммы и печатаются текущие значения выборочных характеристик для i -го шага.

Применение ПАВС позволяет полностью отказаться от хранения массивов большой размерности (известно, что введение массивов большой размерности затруднительно на программах на языке БЕЙСИК). Вместо них нужны всего две ячейки памяти для накопления текущих сумм S_1 и S_2 . А перенеся строки 6266–6267 в ГП, можно сократить затраты времени на вызов ПП и возвращение из нее. Экономия времени счета можно получить так же, перенеся строки 6230, 6232, 6236 и 6238 в ГП. При выполнении вычислений по методу статистических испытаний желательно иметь возможность графически отображать получаемую информацию — как гистограммы (характеризующие форму кривой распределения полученной статистической величины), так и графики (зависимости величины, по которой оценивается распределение в целом от объема выборки и других аргументов). Это обусловлено сложностью, а зачастую невозможностью априорной оценки результатов расчетов. Кроме того, в ряде случаев не существуют объективные надежные методы оценки вычислительных моделей [41], а выявление аппаратных сбоев или различных ошибок, допущенных пользователем, — достаточно сложная задача.

Наблюдая график изменения параметра (доля, среднее и т. п.) "во времени", т. е. по мере наращивания числа испытаний, можно оценить момент возможного прекращения статистического моделирования ввиду достижения удовлетворительной точности. Тогда можно прекратить опыт, перейти к новому набору аргументов или следующему этапу исследования.

4.4. Примеры применения метода МК

▼ С помощью программ, аналогичных по структуре ПМК2, можно решать широкий круг задач статистического анализа качества [44]. Среди них укажем некоторые:

- а) оценка разброса свойств изделия при заданных распределениях входных параметров;
- б) оценка его влияния на процент годных в производстве;
- в) анализ влияния отдельных факторов.

Приведем пример решения первой задачи.

В [55] исследовался разброс максимального M_{max} и пускового $M_{п}$ моментов асинхронного двигателя. В качестве математических моделей были приняты

$$M_{max} = f_1(w_1, d_{пр}, \rho, l_1, \delta, l_2);$$

$$M_{п} = f_2(w_1, d_{пр}, l_1, \delta, l_2),$$

где w_1 – число витков обмотки статора; $d_{пр}$ – диаметр провода обмотки статора; ρ – удельное сопротивление материала короткозамкнутой обмотки ротора; l_1 – длина пакета статора; δ – величина воздушного зазора; l_2 – длина пакета ротора.

В моделях следовало бы учитывать и среднюю длину витка обмотки статора, однако вариация ее столь мала по сравнению с разбросом других факторов, что ею можно пренебречь.

Для всех факторов было принято нормальное распределение, параметры которого для каждой величины приведены в табл. 4.4. С учетом того что все факторы являются независимыми, проводилось статистическое моделирование: вырабатывались шесть псевдослучайных нормально распределенных чисел, они преобразовывались к заданным параметрам распределения, подставлялись в модель и находились одно значение M_{max} и одно значение $M_{п}$. Эти величины запоминались и весь процесс повторялся 1000 раз. По этим двум накопленным выборкам находились (табл. 4.5) среднее значение и характеристики

Таблица 4.4

№ фактора	1	2	3	4	5	6
Единица измерения	шт.	мм	$\frac{\text{Ом} \cdot \text{мм}^2}{\text{м}}$	см	мм	см
\bar{X}	135	1,35	0,065	14,5	0,45	15
σ	0,9	0,0108	0,0075	0,203	0,028	0,18

Таблица 4.5. Результаты моделирования при $N=1000$

Параметр	\bar{X} , Н · м	D , Н · м ²	σ , Н · м	V , %
M_{max}	201,5	0,793	2,81	13,82
M_{Π}	201	2,21	4,69	23,45

разброса: дисперсия, среднееквадратичное отклонение σ , коэффициент вариации V .

▼ Приведенные в таблице данные позволяют уже на стадии проектирования электрической машины оценить ожидаемый разброс ее выходных параметров и допуски на них. Так, например, примем в качестве предельно допустимого отклонения для моментов трехсигмовые пределы. Тогда для максимального момента получаем $M_{max} = 201,5 \pm \pm 1,41$ Н · м (допуск $\pm 7\%$). Такой допуск приведет к почти 100%-ной годности изделий, поскольку обеспечит попадание 99,73% всех значений в поле допуска. Однако, если рассматривать его во взаимосвязи с другими технико-экономическими показателями, то допуск придется уменьшить.

▼ При анализе технологического процесса (с целью повысить однородность, воспроизводимость свойств, сократить брак) возникает задача определить, какой из технологических факторов оказывает наибольшее влияние на качество. Таким образом, возникает задача проранжировать факторы, оценить вклад каждого из них в дисперсию функции.

В [45] исследовано методом МК влияние различных факторов на среднееквадратичное отклонение тока в цепи нагрузки $\sigma(I_H)$ дросселя с подмагничиванием постоянным током. В качестве модели, содержащей девять факторов, используется соотношение

$$I_H = U / \sqrt{(R_H + R_p)^2 + (X_H + X_p)^2}, \quad (4.11)$$

где U – напряжение в цепи нагрузки; R_H, R_p, X_H и X_p – активное и реактивное сопротивления нагрузки и рабочей обмотки:

$$X_p = 2\pi fL; \quad L = w_p^2 \mu_r S/l;$$

здесь f – частота; w_p – число витков рабочей обмотки; S и l – сечение и средняя длина силовой линии магнитопровода; μ_r – относительная начальная проницаемость материала.

По методу МК был произведен расчет среднееквадратичного отклонения I_H , основанный на определенных из эксперимента данных по разбросу всех аргументов, входящих в (4.11). Закон распределения для них принимался соответствующим нормальному. Объем выборки для получения одной точки зависимости равнялся 1000.

Были получены зависимости типа $\sigma(Y) = f(\sigma_i)$, т. е. определялось влияние дисперсии отдельных факторов на разброс Y . Установлено, что среди последних важную роль играет однородность начальной проницаемости магнитного материала: сокращение $\sigma(\mu)$ на 20% приводит к уменьшению $\sigma(I_n)$ на 18%. Отсюда следует рекомендация добиться снижения разброса проницаемости или путем более точного регулирования технологии изготовления магнитопроводов, или путем внедрения промежуточного контроля по этой величине.

Там же исследовались зависимости типа $\sigma(Y) = f(X_i)$. Расчеты показали, что разброс выходной величины зависит также и от номинальных (средних) значений аргументов: некоторые эти зависимости оказались статистически значимыми. В частности, было установлено, что с ростом \bar{X}_p разброс I_n уменьшается, а повышение номинального значения начальной проницаемости магнитопровода на 30% позволяет сократить $\sigma(I_n)$ на 18%. Это свидетельствовало о том, что для данного изделия были неверно установлены номинальные значения для X_p и μ .

▼ Еще одну задачу оценки влияния дисперсии (или иной характеристики разброса) независимых факторов можно решить с помощью программ типа ПМК2: определить, как изменяется процент годных P_r в зависимости от ширины поля допуска на один из факторов X_i или от его номинала. Такое исследование ведется так, что все факторы, кроме заданного X_i , варьируют случайным образом в соответствии с теми значениями параметров распределения, которые для них установлены, а для фактора X_i берется сначала одно значение параметра (например, дисперсии), потом следующее – увеличенное на некоторый шаг. И таким образом строится зависимость $P_r = f(\sigma_i)$ или $P_r = f(X_i)$.

В [55] исследована зависимость дисперсии максимального момента асинхронного двигателя от дисперсии факторов, определяющих его величину. С применением метода МК были получены результаты, которые в виде зависимостей $D(M_{max}) = f(D_i)$ обрабатывались с применением регрессионного анализа: определялись коэффициенты регрессии и их значимость для линейного уравнения

$$D(M_{max}) = a_i + b_i D_i,$$

где D_i – дисперсия i -го фактора.

Одна часть входных величин имела незначимые коэффициенты регрессии, две величины – значимые, но их влияние было признано небольшим, а дисперсия одного фактора – воздушного зазора – оказывает решающее влияние на разброс максимального момента двигателя. Таким образом, коэффициенты регрессии в данном случае играют роль коэффициентов влияния, определяемых и исследуемых в теории чувствительности, и позволяют произвести ранжировку факторов.

Отметим, что выявление наиболее важного фактора — это прямое указание на то, в каком направлении следует совершенствовать технологию данного двигателя — в данном случае в направлении резкого повышения точности обработки, чтобы обеспечить минимальные колебания воздушного зазора.

▼ Метод МК позволяет не только наметить пути улучшения технологии и контроля, но и произвести оптимизацию конструкции изделия. Дело в том, что нет никаких препятствий для определения еще одной группы зависимостей: $D(Y) = f(\bar{X}_1, \dots, \bar{X}_n)$, т. е. выяснить, как влияют на дисперсию выходного свойства не дисперсия входных величин, как было показано выше, а их математические ожидания, величина которых собственно и есть формальное выражение конструкции изделия.

Для приводившегося выше примера было установлено, что на дисперсию пускового момента наибольшее влияние оказывает среднее значение удельного сопротивления короткозамкнутой обмотки ротора: увеличение его на 10% вызывает уменьшение дисперсии $D(M_{п})$ на 41%. А на втором месте число витков обмотки статора: их увеличение позволяет снизить разброс пускового момента.

Такие результаты позволяют ставить вопрос об оценке проектируемых электротехнических изделий по ожидаемому разбросу их выходных технических характеристик и о проведении оптимизации ("статистическая оптимизация") конструкций с учетом влияния дополнительного критерия — дисперсии выходных величин, коэффициента вариации или процента годных. В качестве примеров таких работ в электронике можно указать [44, 56 и др.].

В краткой сводке задач анализа качества электротехнических изделий с применением метода МК, рассмотренных в данном параграфе, нами не рассмотрены такие важные случаи, как оценка качества по нескольким выходным критериям, или учет корреляционной связи между входными факторами.

Как уже отмечалось ранее, метод статистического моделирования пока не нашел широкого применения в практике проектирования электротехнических изделий и в их производстве. Однако увеличение доступности вычислительной техники и происходящее в настоящее время превращение ЭВМ (и особенно ПЭВМ) в рабочий инструмент инженера-электротехника позволяют надеяться, что преимущества и новые интересные возможности применения метода Монте-Карло будут оценены по достоинству и он не только будет использоваться в дальнейшем в системах автоматического проектирования, но и найдет свое место и в заводской практике.

ОСНОВНЫЕ ПОДПРОГРАММЫ

5.1. Решение систем линейных алгебраических уравнений

▼ В различных электротехнических задачах (расчет и анализ электрических цепей, теория управления и др.) возникает необходимость в составлении и решении систем алгебраических линейных уравнений. Кроме того, в ряде вычислительных алгоритмов требуется обращаться матрицы и вычислять определители. В данном параграфе приведено несколько подпрограмм, предназначенных для подобных вычислений. В литературе по прикладной вычислительной математике как широко описаны сами алгоритмы, так и проведен анализ их точности и освещены другие вопросы их применения при расчетах на ЭВМ. Можно рекомендовать пособия и учебники [59–63].

Отсутствие в БЕЙСИКе локальных переменных и возможности передачи данных с помощью списка переменных вынуждает искать иные возможности: данные объединяются в массив, а затем в ПП передается только адрес – номер элемента, строки или столбца, с которых начинается данная группа данных [66]. Этот прием используется во всех ПП данной главы.

▼ В данном параграфе описаны три подпрограммы: одна из них (ПП 5200) предназначена для обращения матриц, а ПП 5600 и ПП 5800 позволяют решать системы линейных уравнений и находить детерминант матрицы. ПП 5200 и ПП 5600 производят вычисления с действительными, а ПП 5800 – с комплексными матрицами.

```

5190 REM *****
5192 REM * ПП ОБРАЩЕНИЯ R-МАТРИЦ МЕТОДОМ *
5194 REM * ОБМЕНА ( N-КРАТНОГО ПЕРЕСЧЕТА *
5196 REM * ЭЛЕМЕНТОВ, N - ПОРЯДОК МАТРИЦЫ ) *
5198 REM *****
5200 LET K0=10~(INT(LOG(A1)/LOG(10))+1)
5202 LET K1=A1+1
5204 FOR K=1 TO A1
5206 LET K2=K
5208 LET K3=K
5210 LET K4=ABS(A[K2,K3])
5212 FOR K5=K TO A1
5214 FOR K6=K TO A1
5216 LET K7=ABS(A[K5,K6])
5218 IF K7 <= K4 THEN 5226
5220 LET K2=K5
5222 LET K3=K6

```



```

5224 LET K4=K7
5226 NEXT K6
5228 NEXT K5
5230 LET A4=SGN(1+SGN(K4-A3))
5232 IF A4=0 THEN 5276
5234 LET K4=A[K2,K3]
5236 LET A[K,K1]=K2+K3/K0
5238 GOSUB 5278
5240 FOR K5=1 TO A1
5242 IF K5=K THEN 5254
5244 LET K6=A[K5,K]/K4
5246 FOR K7=1 TO A1
5248 LET A[K5,K7]=A[K5,K7]-A[K,K7]*K6
5250 NEXT K7
5252 LET A[K5,K]=K6
5254 NEXT K5
5256 FOR K5=1 TO A1
5258 LET A[K,K5]=-A[K,K5]/K4
5260 NEXT K5
5262 LET A[K,K]=1/K4
5264 NEXT K
5266 FOR K=A1 TO 1 STEP -1
5268 LET K3=INT(A[K,K1])
5270 LET K2=INT((A[K,K1]-K3)*K0+.5)
5272 GOSUB 5278
5274 NEXT K
5276 RETURN
5278 IF K=K2 THEN 5290
5280 FOR K5=1 TO A1
5282 LET K6=A[K,K5]
5284 LET A[K,K5]=A[K2,K5]
5286 LET A[K2,K5]=K6
5288 NEXT K5
5290 IF K=K3 THEN 5302
5292 FOR K5=1 TO A1
5294 LET K6=A[K5,K]
5296 LET A[K5,K]=A[K5,K3]
5298 LET A[K5,K3]=K6
5300 NEXT K5
5302 RETURN

```

Общей особенностью всех этих ПП является высокая степень унификации обращения к ним. Все коэффициенты матрицы должны быть размещены в двумерном массиве А, причем, если решается система уравнений, столбец свободных членов размещается в крайнем правом столбце массива А. При работе с комплексными матрицами следует зарезервировать два массива: А — для действительных компонент и В — для мнимых. Перед обращением к ПП следует также указать в идентификаторе А1 порядок матрицы (или число уравнений), а в А3 — точность вычислений. Последний параметр необходим для контроля вырожденности матрицы: если в процессе вычислений значение наибольшего по модулю элемента (главного элемента) становится меньше А3, то дальнейшие операции прекращаются и осуществляется выход из ПП со значением А4 = 0. При естественном завершении работы ПП А4 ≠ 0. Таким образом, А4 указывает на то, как завершена работа. Поэтому прежде чем использовать результаты, полученные с помощью данной ПП, нужно проверить состояние А4. Сделать это можно, например, записав следующую последовательность операторов:

```
1090 IF A4 (<) 0 THEN 1130
1100 PRINT "МАТРИЦА ВЫРОЖДЕНА"
1110 PRINT "ВЫЧИСЛЕНИЯ ПРЕКРАЩАЮТСЯ"
1120 STOP
1130 REM ПРОДОЛЖЕНИЕ
```

которые следует вставить в ГП сразу же после обращения к ПП.

Параметр А2 задается только при вызове ПП 5600 и ПП 5800, в основу которых положен метод Гаусса [59–63].

```
5590 REM *****
5592 REM * РЕШЕНИЕ СИСТЕМ ЛИНЕЙНЫХ УРАВНЕНИЙ *
5594 REM * И ВЫЧИСЛЕНИЕ ДЕТЕРМИНАНТА МЕТОДОМ *
5596 REM * ГАУССА ( R-КОЭФФИЦИЕНТЫ ) *
5598 REM *****
5600 LET K1=A1+A2
5602 LET A4=1
5604 FOR K=1 TO A1
5606 LET K2=K
5608 LET K3=K
5610 LET K4=ABS(A[K2,K3])
5612 FOR K5=K TO A1
5614 FOR K6=K TO A1
5616 LET K7=ABS(A[K5,K6])
5618 IF K7 <= K4 THEN 5626
5620 LET K2=K5
5622 LET K3=K6
```



```

5624 LET K4=K7
5626 NEXT K6
5628 NEXT K5
5630 LET A4=SGN(1+SGN(K4-A3))*A4
5632 LET A5=A4
5634 IF A4=0 THEN 5666
5636 IF K=A1 THEN 5656
5638 LET KO=K
5640 GOSUB 5700
5642 FOR K5=K+1 TO A1
5644 LET K6=-A[K5,K]/A[K,K]
5646 FOR K7=K+1 TO K1
5648 LET A[K5,K7]=A[K5,K7]+K6*A[K,K7]
5650 NEXT K7
5652 NEXT K5
5654 LET A[K+1,1]=K3
5656 NEXT K
5658 IF A2=0 THEN 5668
5660 FOR K=1 TO A1
5662 LET A5=A5*A[K,K]
5664 NEXT K
5666 RETURN
5668 FOR K=A1 TO 1 STEP -1
5670 FOR K2=A1+1 TO K1
5672 FOR K3=K+1 TO A1
5674 LET A[K,K2]=A[K,K2]-A[K,K3]*A[K3,K2]
5676 NEXT K3
5678 LET A[K,K2]=A[K,K2]/A[K,K]
5680 NEXT K2
5682 NEXT K
5684 LET KO=A1+1
5686 FOR K3=A1 TO 2 STEP -1
5688 LET K=K3-1
5690 LET K2=A[K3,1]
5692 GOSUB 5714
5694 NEXT K3
5696 LET A4=1
5698 RETURN
5700 IF K=K3 THEN 5714
5702 FOR K5=1 TO A1
5704 LET K6=A[K5,K]

```

```

5706 LET A[K5,K]=A[K5,K3]
5708 LET A[K5,K3]=K6
5710 NEXT K5
5712 LET A4=-A4
5714 IF K=K2 THEN 5728
5716 FOR K5=K0 TO K1
5718 LET K6=A[K,K5]
5720 LET A[K,K5]=A[K2,K5]
5722 LET A[K2,K5]=K6
5724 NEXT K5
5726 LET A4=-A4
5728 RETURN

```

Параметр A2 служит ключом, позволяющим выбирать один из двух режимов работы. Если $A2 = 0$, то вычисляется детерминант матрицы, значение которого размещается в A5 (действительная часть при работе с ПП 5600) и в A5 и A6 (действительная и мнимая части при работе с ПП 5800). Если $A2 = 1$, то решается система линейных уравнений (ее порядок равен A1) и в столбце под номером A1 + 1 (крайний правый) массива коэффициентов размещаются свободные члены.

```

5790 REM *****
5792 REM * РЕШЕНИЕ СИСТЕМ ЛИНЕЙНЫХ УРАВНЕНИЙ *
5794 REM * И ВЫЧИСЛЕНИЕ ДЕТЕРМИНАНТА МЕТОДОМ *
5796 REM * ГАУССА ( С-КОЭФФИЦИЕНТЫ ) *
5798 REM *****
5800 LET K1=A1+A2
5802 LET A4=1
5804 FOR K=1 TO A1
5806 LET K2=K
5808 LET K3=K
5810 LET K4=A[K2,K3]*A[K2,K3]+B[K2,K3]*B[K2,K3]
5812 FOR K5=K TO A1
5814 FOR K6=K TO A1
5816 LET K7=A[K5,K6]*A[K5,K6]+B[K5,K6]*B[K5,K6]
5818 IF K7 <= K4 THEN 5826
5820 LET K2=K5
5822 LET K3=K6
5824 LET K4=K7
5826 NEXT K6
5828 NEXT K5
5830 LET A4=SGN(1+SGN(SQR(K4)-A3))*A4

```



```

5832 LET A5=A4
5834 IF A4=0 THEN 5880
5836 IF K=A1 THEN 5864
5838 LET KO=K
5840 GOSUB 5922
5842 LET K8=A[K,K]/K4
5844 LET K9=-B[K,K]/K4
5846 FOR K5=K+1 TO A1
5848 LET K2=-A[K5,K]*K8+B[K5,K]*K9
5850 LET K4=-A[K5,K]*K9-B[K5,K]*K8
5852 FOR K7=K+1 TO K1
5854 LET A[K5,K7]=A[K5,K7]+A[K,K7]*K2-B[K,K7]*K4
5856 LET B[K5,K7]=B[K5,K7]+A[K,K7]*K4+B[K,K7]*K2
5858 NEXT K7
5860 NEXT K5
5862 LET A[K+1,1]=K3
5864 NEXT K
5866 IF A2>0 THEN 5882
5868 LET A6=0
5870 FOR K=1 TO A1
5872 LET K2=A[K,K]*A5-B[K,K]*A6
5874 LET A6=A[K,K]*A6+B[K,K]*A5
5876 LET A5=K2
5878 NEXT K
5880 RETURN
5882 FOR K=A1 TO 1 STEP -1
5884 LET K4=A[K,K]*A[K,K]+B[K,K]*B[K,K]
5886 FOR K2=A1+1 TO K1
5888 FOR K3=K+1 TO A1
5890 LET A[K,K2]=A[K,K2]-A[K,K3]*A[K3,K2]+B[K,K3]*B[K3,K2]
5892 LET B[K,K2]=B[K,K2]-A[K,K3]*B[K3,K2]-A[K3,K2]*B[K,K3]
5894 NEXT K3
5896 LET K5=(A[K,K2]*A[K,K]+B[K,K2]*B[K,K])/K4
5898 LET B[K,K2]=(B[K,K2]*A[K,K]-A[K,K2]*B[K,K])/K4
5900 LET A[K,K2]=K5
5902 NEXT K2
5904 NEXT K
5906 LET KO=A1+1
5908 FOR K3=A1 TO 2 STEP -1
5910 LET K=K3-1
5912 LET K2=A[K3,1]

```

```

5914 GOSUB 5942
5916 NEXT K3
5918 LET A4=1
5920 RETURN
5922 IF K=K3 THEN 5942
5924 FOR K5=1 TO A1
5926 LET K6=A[K5,K]
5928 LET A[K5,K]=A[K5,K3]
5930 LET A[K5,K3]=K6
5932 LET K6=B[K5,K]
5934 LET B[K5,K]=B[K5,K3]
5936 LET B[K5,K3]=K6
5938 NEXT K5
5940 LET A4=-A4
5942 IF K=K2 THEN 5962
5944 FOR K5=K0 TO K1
5946 LET K6=A[K,K5]
5948 LET A[K,K5]=A[K2,K5]
5950 LET A[K2,K5]=K6
5952 LET K6=B[K,K5]
5954 LET B[K,K5]=B[K2,K5]
5956 LET B[K2,K5]=K6
5958 NEXT K5
5960 LET A4=-A4
5962 RETURN

```

Необходимость задания $A2 > 1$ возникает, например, при определении методом Гаусса обратной матрицы, когда одновременно нужно решить $A1$ систем линейных уравнений со свободными членами, представляющими собой векторы-столбцы единичной матрицы [60].

Подпрограмма ПП 5200 обращает действительную матрицу методом N -кратного пересчета элементов [59, 60]. Ее применение требует выделения в массиве A дополнительного $A1 + 1$ столбца для промежуточных результатов.

По окончании работы ПП 5200 на месте исходной матрицы размещается обратная, поэтому следует предусмотреть копирование первой, если в дальнейшем она будет нужна.

▼ В качестве примера использования ПП 5200 найдем матрицу, обратную следующей:

1	2
-3	4

Подготовим исходные данные с помощью фрагмента

```
1 DIM A(2,3)
10 DATA 1,2,-3,4
20 LET A1 = 2
30 LET A3 = 1E - 20
40 FOR I = 1 TO A1
50 FOR J = 1 TO A1
60 READ A(I, J)
70 NEXT J
80 NEXT I
```

Вызов ПП осуществляем строкой

```
90 GOSUB 5200
```

а по окончании вычислений результаты выводим следующими строками:

```
100 PRINT "ОБРАТНАЯ МАТРИЦА"
110 FOR I = 1 TO A1
120 FOR J = 1 TO A1
130 PRINT A(I, J);
140 NEXT J
150 PRINT
160 NEXT I
9999 END
```

(укажем, что для краткости здесь не производится контроль содержания A4). Обратная матрица имеет вид

```
0,4  -0,2
0,3   0,1
```

▼ ПП 5600 и ПП 5800 используют метод Гаусса с выбором главного элемента. Проиллюстрируем работу ПП 5600 на следующем примере:

$$\begin{aligned} X + 2Y &= 7 \\ -3X + 4Y &= 9 \end{aligned}$$

Процедура подготовки исходных данных осуществляется тем же фрагментом, что был приведен выше (строки 1–80), в котором надо изменить строки

```
10 DATA 1,2,7,-3,4,9
50 FOR J = 1 TO A1 + 1
```

и добавить

```
25 LET A2 = 1
```

Вызов ПП производим строкой

```
90 GOSUB 5600
```

Результат вычислений выводим с помощью строк

```
100 PRINT "КОРНИ РАВНЫ"
```

```
110 FOR I = 1 TO A1
```

```
120 PRINT A(I,A1+1)
```

```
130 NEXT I
```

Он равен: $X_1 = 1$ и $X_2 = 3$.

Для вычисления детерминанта достаточно изменить лишь строку

```
25 LET A2 = 0
```

и вместо строк 100–160 записать

```
100 PRINT "ДЕТЕРМИНАНТ ="; A5
```

Единственное отличие для ПП 5800 состоит в том, что предварительно нужно заполнить два массива: А и В, разместив, как было отмечено выше, в массиве В мнимые компоненты коэффициентов матрицы и свободных членов.

▼ В заключение отметим, что во всех ПП используется ограниченное число идентификаторов простых переменных, начинающихся с букв А и К, а также идентификаторы двумерных массивов А и В, которые необходимо описывать в ГП следующим образом:

```
DIM A(A1, A1+1)
```

при вызове ПП 5200 и

```
DIM A(A1, A1+ A2), B(A1, A1+ A2)
```

при вызове ПП 5600 и ПП 5800 (напомним, массив В используется и поэтому описывается только при вызове ПП 5400 и ПП 5800). Вместо А1, А2 и арифметических выражений следует в описании применять их числовые эквиваленты.

5.2. Действия с комплексными числами

▼ Из всех операций с комплексными числами для программирования были выбраны те, которые чаще всего используются в электротехнике, — комплексная арифметика, вычисление степени и гиперболических функций. Они, как показал опыт практических расчетов, наиболее употребительны в конкретных задачах. Всего было разработано 16 ПП (из них 2 вспомогательные), список которых приведен в табл. 5.1.

Для упрощения использования ПП приняты определенные ограничения, которые следует учитывать при разработке математического

Таблица 5.1

Номер первого оператора ПП	Назначение ПП (выполняемая операция)	Номера строк входных параметров	Вызываемые ПП
4210	$C_3^* = C_1 + C_2$	Z0, Z1, Z2	—
4220	$C_3 = C_1 C_2$	Z0, Z1, Z2	—
4240	$C_3 = C_1 / C_2$	Z0, Z1, Z2	4220
4270	$C_3 = 1 / C_1$	Z0, Z1	—
4290	$C_3 = \frac{C_1 C_2}{C_1 + C_2}$	Z0, Z1, Z2, Z4	4220, 4240 4320, 4330
4320	Запоминание входных параметров	Z1, Z2, Z3, Z5, Z6, Z7	—
4330	Восстановление входных параметров	Z1, Z2, Z3, Z5, Z6, Z7	—
4350	$A_3 = \text{mod} \{C_1\}$ $B_3 = \text{arg} \{C_1\}$	Z1	—
4390	$A_3 = \text{Re} \{C_1\}$ $B_3 = \text{Im} \{C_1\}$ $C_1 = A_1 \exp \{JB_1\}$	Z1	—
4410	$C_3 = C_1^{0,5}$	Z1	4350, 4390
4430	$C_3 = C_1^2$	Z1, Z2	4220, 4320 4330, 4350 4390
4450	$C_3 = \exp \{C_1\}$	Z1	—
4470	$C_3 = \text{SH} \{C_1\}$	Z1, Z2	4320, 4330 4450
4500	$C_3 = \text{CH} \{C_1\}$	Z1, Z2	4470
4510	$C_3 = \text{TH} \{C_1\}$	Z1, Z2, Z4	4240, 4330 4470, 4500
4530	$C_3 = \text{ARC TH} \{C_1\}$	Z1	4350

* $C_k = A_k + JB_k$, A_k содержится в первом столбце массива Z, а B_k — во втором.

обеспечения с использованием операций с комплексными числами. В ПП применяется ограниченное число идентификаторов — двухмерный базовый массив $Z(N, 2)$, переменные Z0, Z1, . . . , Z9 и C0, C1, . . . , C9. Эти идентификаторы желательно не использовать в ГП.

Все комплексные числа (как исходные данные, так и результаты вычислений) располагаются в базовом массиве $Z(N, 2)$ по строкам: в первом столбце — действительная часть, во втором столбце — мнимая. Базовый массив Z описывается в ГП, число строк в нем (N) зависит от степени сложности программы. Благодаря принятой системе обозначений достигается экономия времени и длины текста.

При вызове той или иной ПП нет необходимости организовать пересылку исходных данных и результатов — достаточно указать лишь номера строк, в которых они размещаются. Для этого в $Z1$ и $Z2$ заносятся номера строк базового массива, содержащих исходные данные, а в $Z3$ — номер строки, в которую будет занесен результат расчета. (В некоторых случаях нужно дополнительно указать в $Z4$ номер вспомогательной строки базового массива, предназначенной для хранения промежуточных данных.)

ПП подразделяются на три группы: ПП алгебраических преобразований (сложение, умножение, деление и обращение комплексных чисел), ПП вычисления гиперболических функций от комплексного переменного и, наконец, группа программ для вычисления квадратного корня, экспоненты, возведения в степень, определения модуля и аргумента, перехода от одной формы записи комплексных чисел к другой. Чтобы сохранить исходные данные, применяются две вспомогательные ПП — для запоминания (ПП 4320) и для восстановления (ПП 4330) информации.

▼ Кратко остановимся на отдельных ПП.

```

4200 REM *****
4201 REM * СЛОЖЕНИЕ *
4202 REM *****
4210 LET Z[Z3,1]=Z[Z1,1]+Z[Z2,1]
4212 IF ZO=1 THEN 4216
4214 LET Z[Z3,2]=Z[Z1,2]+Z[Z2,2]
4216 RETURN
4217 REM *****
4218 REM * УМНОЖЕНИЕ *
4219 REM *****
4220 IF ZO=2 THEN 4226
4222 LET Z[Z3,1]=Z[Z1,1]*Z[Z2,1]
4224 RETURN
4226 LET Z[Z3,1]=Z[Z1,1]*Z[Z2,1]-Z[Z1,2]*Z[Z2,2]
4228 LET Z[Z3,2]=Z[Z1,1]*Z[Z2,2]+Z[Z1,2]*Z[Z2,1]
4230 LET Z[Z3,1]=Z
4232 RETURN
4233 REM *****

```



```

4234 REM * ДЕЛЕНИЕ *
4235 REM *****
4240 IF ZO=2 THEN 4246
4242 LET Z[Z3,1]=Z[Z1,1]/Z[Z2,1]
4244 RETURN
4246 LET Z9=Z[Z2,1]*Z[Z2,1]+Z[Z2,2]*Z[Z2,2]
4248 LET Z[Z2,2]=-Z[Z2,2]
4250 GOSUB 4226
4252 IF Z3=Z2 THEN 4256
4254 LET Z[Z2,2]=-Z[Z2,2]
4256 LET Z[Z3,1]=Z[Z3,1]/Z9
4258 LET Z[Z3,2]=Z[Z3,2]/Z9
4260 RETURN
4261 REM *****
4262 REM * ОБРАЩЕНИЕ *
4263 REM *****
4270 IF ZO=2 THEN 4276
4272 LET Z[Z3,1]=1/Z[Z1,1]
4274 RETURN
4276 LET Z=Z[Z1,1]*Z[Z1,1]+Z[Z1,2]*Z[Z1,2]
4278 LET Z[Z3,1]=Z[Z1,1]/Z
4280 LET Z[Z3,2]=-Z[Z1,2]/Z
4282 RETURN
4283 REM *****
4284 REM * ПРИБЛИЖЕНИЕ  $Z1*Z2/(Z1+Z2)$  *
4285 REM *****
4290 IF ZO=2 THEN 4296
4292 LET Z[Z3,1]=Z[Z1,1]*Z[Z2,1]/(Z[Z1,1]+Z[Z2,1])
4294 RETURN
4296 GOSUB 4320
4298 LET Z3=Z4
4300 GOSUB 4210
4302 LET Z3=Z7
4304 GOSUB 4226
4306 LET Z1=Z3
4308 LET Z2=Z4
4310 GOSUB 4246
4312 GOSUB 4330
4314 RETURN
4315 REM *****
4316 REM * ЗАПОМНАНИЕ И ВОССТАНОВЛЕНИЕ *

```

```

4317 REM * ПАРАМЕТРОВ ПИ *
4318 REM *****
4320 LET Z5=Z1
4322 LET Z6=Z2
4324 LET Z7=Z3
4326 RETURN
4330 LET Z1=Z5
4332 LET Z2=Z6
4334 LET Z3=Z7
4336 RETURN
4337 REM *****
4338 REM * ВЫЧИСЛЕНИЕ МОДУЛЯ И АРГУМЕНТА С-ЧИСЛА, *
4339 REM * ПРЕДСТАВЛЕННОГО В АЛГЕБРАИЧЕСКОЙ ФОРМЕ *
4340 REM *****
4350 LET CO=3.14159
4352 LET C1=Z[Z1,1]
4354 LET C2=Z[Z1,2]
4356 IF ABS(C2) <= ABS(C1) THEN 4364
4358 LET C3=CO/2-ATN(ABS(C1/C2))
4360 LET Z[Z3,1]=ABS(C2)/SIN(C3)
4362 GOTO 4368

4364 LET C3=ATN(ABS(C2/(C1+(1-SGN(ABS(C1))))))
4366 LET Z[Z3,1]=ABS(C1)/COS(C3)
4368 IF C1 >= 0 THEN 4372
4370 LET C3=CO-C3
4372 LET C4=INT(C3*SGN(SGN(C2)+.5)*180000./CO+.5)/1000
4374 LET C4=C4+360*(1-SGN(ABS(C4+180)))
4376 LET Z[Z3,2]=C4*CO/180
4378 RETURN
4379 REM *****
4380 REM * ВЫЧИСЛЕНИЕ ДЕЙСТВИТЕЛЬНОЙ И МИМОЙ *
4381 REM * ЧАСТИ С-ЧИСЛА, ПРЕДСТАВЛЕННОГО В *
4382 REM * ВИДЕ МОДУЛЯ И АРГУМЕНТА *
4383 REM *****
4390 LET C1=Z[Z1,1]
4392 LET C2=Z[Z1,2]
4394 LET Z[Z3,1]=C1*COS(C2)
4396 LET Z[Z3,2]=C1*SIN(C2)
4398 RETURN
4399 REM *****
4400 REM * ВЫЧИСЛЕНИЕ КВАДРАТНОГО КОРНЯ С-ЧИСЛА *
4401 REM *****

```



```

4410 GOSUB 4350
4412 LET Z[Z3,1]=SQR(Z[Z3,1])
4414 LET Z[Z3,2]=Z[Z3,2]/2
4416 LET Z7=Z1
4418 LET Z1=Z3
4420 GOSUB 4390
4422 LET Z1=Z7
4424 RETURN
4425 REM *****
4426 REM * ВОЗВЕДЕНИЕ С-ЧИСЛА В С-СТЕПЕНЬ *
4427 REM *****
4430 GOSUB 4320
4432 GOSUB 4350
4434 LET Z[Z3,1]=LOG(Z[Z3,1])
4436 LET Z1=Z3
4438 GOSUB 4226
4440 LET Z[Z3,1]=EXP(Z[Z3,1])
4442 GOSUB 4390
4444 GOSUB 4330
4446 RETURN
4447 REM *****
4448 REM * ВЫЧИСЛЕНИЕ ЭКСПОНЕНТЫ С-ЧИСЛА *
4449 REM *****
4450 LET C8=EXP(Z[Z1,1])
4452 LET C9=Z[Z1,2]
4454 LET Z[Z3,1]=C8*COS(C9)
4456 LET Z[Z3,2]=C8*SIN(C9)
4458 RETURN
4459 REM *****
4460 REM * ВЫЧИСЛЕНИЕ SH С-ЧИСЛА *
4461 REM *****
4470 LET C7=-1
4472 GOSUB 4320
4474 GOSUB 4450
4476 LET Z3=Z2
4478 LET Z[Z1,1]=-Z[Z1,1]
4480 LET Z[Z1,2]=-Z[Z1,2]
4482 GOSUB 4450
4484 LET Z[Z7,1]=(Z[Z7,1]+C7*Z[Z3,1])/2
4486 LET Z[Z7,2]=(Z[Z7,2]+C7*Z[Z3,2])/2
4488 LET Z[Z1,1]=-Z[Z1,1]

```



```

4490 LET Z[Z1,2]=-Z[Z1,2]
4492 GOSUB 4330
4494 RETURN
4495 REM *****
4496 REM * ВЪЧИСЛЕНИЕ СН С-ЧИСЛА *
4497 REM *****
4500 LET C7=1
4502 GOSUB 4472
4504 RETURN
4505 REM *****
4506 REM * ВЪЧИСЛЕНИЕ ТН С-ЧИСЛА *
4507 REM *****
4510 GOSUB 4470
4512 LET Z[Z4,1]=Z[Z3,1]
4514 LET Z[Z4,2]=Z[Z3,2]
4516 GOSUB 4500
4518 LET Z1=Z4
4520 LET Z2=Z3
4522 GOSUB 4246
4524 GOSUB 4330
4526 RETURN
4527 REM *****
4528 REM * ВЪЧИСЛЕНИЕ ARC ТН С-ЧИСЛА *
4529 REM *****
4530 LET C8=1-Z[Z1,1]
4532 LET C9=Z[Z1,2]
4534 LET C7=C8*C8+C9*C9
4536 LET Z[Z3,1]=2*C8/C7-1
4538 LET Z[Z3,2]=2*C9/C7
4540 LET Z5=Z1
4542 LET Z1=Z3
4544 GOSUB 4350
4546 LET Z[Z3,1]=LOG(Z[Z3,1])/2
4548 LET Z[Z3,2]=Z[Z3,2]/2
4550 LET Z1=Z5
4552 RETURN

```



ПП 4210 позволяет складывать комплексные числа. До ее вызова слагаемые размещаются в строках массива Z с номерами Z1 и Z2. А сумма помещается в строке с номером Z3.

ППП 4220 и ППП 4240 производят умножение и деление. Размещение исходной информации и результатов аналогично предыдущему случаю.

Обратить комплексное число можно с помощью ППП 4270. Оно должно быть занесено в строчку с номером Z1 базового массива, а результат будет, как и в предыдущих случаях, помещен в строку с номером Z3 (параметр Z2 не используется).

Специфика электротехнических расчетов часто требует вычисления выражений типа $C1 \times C2 / (C1 + C2)$, где C1 и C2 – комплексные числа. Эта операция выполняется с помощью ППП 4290, для которой также используется стандартное размещение информации. Кроме того, нужно указать номер Z4 свободной строчки базового массива для размещения промежуточных результатов.

Для всех перечисленных выше программ указывается также значение ключа Z0. При $Z0 = 1$ все операции производятся с действительными числами, при $Z0 = 2$ действия выполняются с комплексными числами.

Гиперболические функции вычисляются по обычным математическим формулам [64]. Во всех ППП этой группы аргумент размещается в строке с номером Z1 базового массива, а результат – в строке с номером Z3.

ППП 4470, ППП 4500, ППП 4510 и ППП 4530 позволяют соответственно вычислять гиперболический синус, косинус, тангенс и арктангенс от комплексного числа.

Размещение промежуточных результатов для первых двух программ требует одной дополнительной строки с номером Z2, а для ППП 4510 – еще одной строки с номером Z4. Укажем также, что мнимая компонента комплексного числа гиперболического арктангенса определяется с точностью $2\pi k$, где $k = 1, 2, \dots$

Следующая группа ППП – это ППП 4350, 4390, 4410 и 4450. ППП 4350 позволяет вычислять модуль и аргумент комплексного числа. В строке с номером Z1 следует поместить действительную и мнимую части комплексного числа, а результат помещается в строке с номером Z3: в первом столбце – значение модуля, во втором – аргумент в радианах.

ППП 4390 предназначается для решения обратной задачи.

ППП 4410 вычисляет главное значение квадратного корня из комплексного числа [65]. Как и ранее, в строках с номерами Z1 и Z3 размещаются исходное число и результат.

ППП 4430 возводит комплексное число в комплексную степень. В строках базового массива с номерами Z1–Z3 помещаются соответственно основание степени, ее показатель и результат. Назначение ППП 4440 – вычисление комплексной экспоненты. И в этом случае в Z1 указывается номер строки с показателем экспоненты, а в Z3 – адрес результата.

5.3. Решение нелинейных уравнений

▼ Многие задачи в электротехнике сводятся к нахождению корней нелинейных уравнений. В данном параграфе приведены две ПП: ПП 6510 для определения действительного корня методом дихотомии, ПП 6550 для нахождения минимума функции методом золотого сечения.

Метод дихотомии (половинного деления) [60] для приближенного определения действительного корня уравнения $f(x) = 0$ реализован в ПП 6510. До обращения к ней необходимо найти интервал $[a, b]$, на границах которого $f(x)$ принимает значения, удовлетворяющие условию $f(a)f(b) < 0$. Последнее неравенство означает, что внутри интервала $f(x)$ пересекает ось абсцисс и, следовательно, имеет хотя бы один корень.

Исходными данными для ПП являются значения a и b , которые предварительно заносятся в $Y(1,1)$ и $Y(1,2)$ соответственно, а также заданная погрешность вычисления корня, указанная в $Y0$. В процессе работы в ПП неоднократно вычисляются значения функции $f(x)$ при различных величинах аргумента. Поскольку вид функции заранее не известен, следует подготовить ПП, вычисляющую $f(x)$ при $x=Y1$ и размещающую результат в $Y2$, и в строках 6514 и 6524 ПП 6510 записать обращение к ней. Однако если функция имеет достаточно простой вид, в указанных строках можно поместить оператор

```
LET Y2 = FNF(Y1)
```

и описать нужную функцию FNF в ГП.

По окончании работы ПП 6510 в $Y1$ будет расположено значение корня. Используемый в ПП массив Y размерности 2×2 следует описать в ГП.

```
6500 REM *****
6502 REM * ОПРЕДЕЛЕНИЕ ТОЧКИ ПЕРЕСЕЧЕНИЯ *
6504 REM * ФУНКЦИИ С ОСЬЮ АБСЦИСС МЕТОДОМ *
6506 REM * ДИХОТОМИИ (ПОЛОВИННОГО ДЕЛЕНИЯ) *
6508 REM *****
6510 FOR Y=1 TO 2
6512 LET Y1=Y[Y,1]
6514 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
6516 LET Y[Y,2]=Y2
6518 NEXT Y
6520 LET Y1=(Y[1,1]+Y[2,1])/2
6522 IF ABS(Y[1,1]-Y[2,1]) <= Y0 THEN 6534
6524 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
6526 LET Y=1+ABS(SGN(SGN(Y2)-SGN(Y[1,2])))
```



```

6528 LET Y[Y,1]=Y1
6530 LET Y[Y,2]=Y2
6532 GOTO 6520
6534 RETURN

```

▼ ПП 6550 решает более общую задачу — она позволяет приближенно найти абсциссу локального минимума функции $f(x)$ методом золотого сечения [60].

Исходные данные — границы интервала $[a, b]$, в котором ищется минимум, и значение погрешности — задаются так же, как и для вычисления значений функции. Обращение к последней ПП нужно разместить в строках 6570 и 6596. По окончании работы в Y1 содержится абсцисса минимума. Вспомогательный массив Y размерностью 4 x 2 следует описать в ГП.



```

6536 REM *****
6538 REM * ВЫЧИСЛЕНИЕ МИНИМУМА ФУНКЦИИ *
6540 REM * МЕТОДОМ ЗОЛОТОГО СЕЧЕНИЯ *
6542 REM *****
6550 LET Y6=1
6552 GOTO 6556
6554 LET Y6=3
6556 LET Y7=ABS(Y[2,1]-Y[1,1])
6558 LET Y8=(SQR(5)-1)/2
6560 LET Y9=Y7*Y8
6562 LET Y[3,1]=Y[2,1]-Y9
6564 LET Y[4,1]=Y[1,1]+Y9
6566 FOR Y=Y6 TO 4
6568 LET Y1=Y[Y,1]
6570 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
6572 LET Y[Y,2]=Y2
6574 NEXT Y
6576 LET Y9=Y9*Y8
6578 IF ABS(Y[2,1]-Y[1,1]) <= Y0 THEN 6602
6580 LET Y1=1+SGN(1+SGN(Y[4,2]-Y[3,2]))
6582 FOR Y=1 TO 2
6584 LET Y[Y1,Y]=Y[2+Y1,Y]
6586 LET Y[2+Y1,Y]=Y[5-Y1,Y]
6588 NEXT Y
6590 LET Y=5-Y1
6592 LET Y[Y,1]=Y[Y1,1]+SGN(1.5-Y1)*Y9

```

```

6594 LET Y1=Y[Y,1]
6596 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
6598 LET Y[Y,2]=Y2
6600 GOTO 6576
6602 LET Y1=(Y[2,1]+Y[1,1])/2
6604 RETURN

```

5.4. Подпрограммы для ввода и вывода информации

▼ Большинство современных версий языка БЕЙСИК обладают возможностями, позволяющими значительно усовершенствовать процедуры ввода и вывода информации, придать ей более наглядный вид, упростить действия пользователя и т. п. В связи с этим опишем ПП, аналогичные по своему назначению ПП 1700 и ПП 2000 (приведены в гл. 1 и 2) и построенные на основе версий языка, реализованных на ПЭВМ типов БК0010, ДВК, ЕС1840 и др.

Номера новых вариантов ПП совпадают с номерами ПП, описанных ранее. Это позволяет не вносить изменений в ГП, где имеется их вызов. Поэтому прежние версии ПП можно без существенных изменений заменить на более совершенные.

Опишем новый вариант ПП 2000 при условии, что данная ПП будет применяться только при расчете схем на постоянном токе. Она предназначена для ввода топологии схемы и параметров ветвей путем заполнения "электронной" ("экранный") таблицы, отображаемой на экране дисплея (табл. 5.2).

Таблица 5.2

Ветвь	Узел 1	Узел 2	R[OM]	I[A]	E[B]
1	1	3	1.23	-10.1E-1	4.2E2
...

```

1994 REM *****
1995 REM * ВВОД СОПРОТИВЛЕНИЙ ВЕТВЕЙ И ТОПОЛОГИИ *
1996 REM * СХЕМЫ С ПОМОЩЬЮ ЭЛЕКТРОННОЙ ТАБЛИЦЫ *
1997 REM *****
2000 REM === ПОДГОТОВКА ===
2010 DIM N(7),Nn(6),H(6)
2012 DEF FNK(K)=INT(P(K,1))
2014 DEF FNL(L)=INT((P(L,1)-FNK(L))*N0+.5)
2016 DATA 1000,0,0,6,7
2018 DATA 1,7,13,19,34,49,64

```



```

2020 DATA ВЕТЬ,УЗЕЛ1,УЗЕЛ2,' R(OM)''
2022 DATA '' J(A)'',' E(B)''
2024 READ N0,N1,N2,N6,N7
2026 FOR I=1 TO N6+1
2028 READ N(I)
2030 NEXT I
2032 FOR I=1 TO N6
2034 READ N0(I)
2036 NEXT I
2038 FOR I=1 TO N
2040 FOR J=1 TO 4
2042 P(I,J)=0
2044 NEXT J
2046 NEXT I
2100 REM === ИЗОБРАЖЕНИЕ ===
2110 GOSUB 2300
2112 N9=1
2114 GOSUB 2380
2120 REM === ВВОД ===
2122 I=0
2124 K=N9+I
2126 IF K>N THEN I=I-1
2128 L=I*2+3
2130 LOCATE N(1),L,1
2132 N0=INKEY0
2134 IF N0=''' THEN 2132
2136 IF N0='3' THEN 2202
2138 IF N0<-'M' THEN 2146
2140 GOSUB 2240
2142 N9=I-1
2144 GOTO 2114
2146 J=ASC(N0)
2148 IF NOT(J=26 OR J=27) THEN 2154
2150 I=I+SGN(J-26.5)-1
2152 GOTO 2194
2154 LOCATE N(1),L
2156 PRINT K
2158 FOR J=2 TO 6
2160 H(2)=FNK(K)
2162 H(3)=FNL(K)
2164 N0='''

```

```

2166 IF J>3 THEN N1=P(K,J-2)
2168 IF J=2 OR J=3 THEN N1=H(J)
2170 LOCATE N(J),L,1
2172 INPUT Nα
2174 IF Nα<>' ' THEN N1=VAL(Nα)
2176 LOCATE N(J),L
2178 PRINT STRINGα(N(J+1)-N(J)-1,' ')
2180 LOCATE N(J),L
2182 PRINT N!
2184 IF J=4 AND N1<K AND N1<>0 THEN N1=K
2186 IF J=3 THEN P(K,J-2)=N!
2188 IF J=2 THEN P(K,1)=N!+H(3)/NO
2190 IF J=3 THEN P(K,1)=H(2)+N!/NO
2192 NEXT J
2194 I=I+1
2196 IF I=0 AND I<N7 THEN 2124
2198 N9=49+N7*SQN(I+.5)
2200 GOTO 2114
2202 FOR I=1 TO N1
2204 IF N2=FNK(I) THEN N2=FNK(I)
2206 IF N2=FNL(I) THEN N2=FNL(I)
2208 NEXT I
2210 RETURN
2230 REM === ИСКЛЮЧЕНИЕ СТРОКИ ===
2240 IF K=N1 THEN RETURN
2242 IF K=N1 THEN 2254
2244 FOR J=K+1 TO N1
2246 FOR L=1 TO 4
2248 P(J-1,L)=P(J,L)
2250 NEXT L
2252 NEXT J
2254 N1=N1-1
2256 RETURN
2300 REM === ЛИНИИ И ЗАГОЛОВКИ ===
2310 CLS
2312 K=(N7+1)*20+4
2314 FOR I=1 TO N6+1
2316 IF I=N6+1 THEN 2322
2318 LOCATE N(I),1
2320 PRINT Nα(I)
2322 J=3+(N(I)-1)*8

```

```

2324 LINE (J,4)-(J,K)
2326 NEXT I
2328 FOR I=0 TO N7+1
2330 K=4+I*20
2332 LINE (3,K)-(J,K)
2334 NEXT I
2336 REM === КОММЕНТАРИЙ ===
2338 LOCATE 0,(N7+1)*2+1
2340 PRINT ''ПОЛОЖИТЕЛЬНОЕ НАПРАВЛЕНИЕ ТОКА В ВЕТВИ''
2342 PRINT ''СОВПАДАЕТ С НАПРАВЛЕНИЕМ ОТ УЗЛА1 К УЗЛУ2''
2344 PRINT TAB(29);''ВИДЫ РЕЖИМОВ:И-ИСКЛЮЧЕНИЕ СТРОКИ''
2346 PRINT TAB(43);''3-ЗАВЕРШЕНИЕ ВВОДА''
2348 PRINT ''ИСПОЛЬЗУЙТЕ КЛАВИШИ С ИЗОБРАЖЕНИЕМ СТРЕЛОК,''
2350 PRINT ''НАПРАВЛЕННЫХ ВВЕРХ И ВНИЗ.''
2352 RETURN
2370 REM === ЧИСЛА ===
2380 IF N9>N1 THEN N9=N1
2382 IF N9<=0 THEN N9=1
2384 FOR I=0 TO N7-1
2386 K=N9+I
2388 L=I*2+3
2390 FOR J=1 TO 6
2392 LOCATE N(J),L
2394 PRINT STRING$(N(J+1)-N(J)-1,' ' ')
2396 NEXT J
2398 IF K>N1 THEN 2412
2400 GOSUB 2420
2402 FOR J=1 TO 6
2404 N1=H(J)
2406 LOCATE N(J),L
2408 IF J<4 OR N1!><0 THEN PRINT N1
2410 NEXT J
2412 NEXT I
2420 REM === ЗАПОЛНЕНИЕ МАССИВА H ===
2430 H(1)=K
2432 H(2)=FNK(K)
2434 H(3)=FNL(K)
2436 FOR J=2 TO 4
2438 H(J+2)=P(K,J)
2440 NEXT J
2442 IF MID$(N$(4),5,1)='R' THEN RETURN
2444 GOTO 1750

```

Она состоит из блоков, функции блоков приведены далее.

В строках 2000–2060 задаются начальные значения некоторым переменным и массивам. Как и раньше, вся вводимая пользователем информация хранится в двухмерном массиве Р, который должен быть описан в ГП:

(номер строки) DIM P(N, 4)

где N – число ветвей схемы. Последнее также должно быть задано в ГП (до вызова ПП 2000). Например, для схемы со 100 ветвями следует написать

(номер строки) LET N=100

В массиве Р, как и прежде, в первом столбце хранятся попарно упакованные номера узлов, во втором столбце – сопротивления ветвей, а в третьем и четвертом столбцах записываются номиналы источников тока и ЭДС.

ПП 2300, очистив экран дисплея, рисует вертикальные и горизонтальные прямые, образующие таблицу, заполняет первую строку наименованиями заголовков столбцов, выводит в свободной нижней части экрана краткие пояснения о правилах работы пользователя с таблицей.

ПП 2380 отображает содержимое массива Р в клетках таблицы. Строки 2120–2200 обеспечивают ввод числовой информации в массив Р с клавиатуры. Две сервисные ПП 2202 и ПП 2230 предназначены для определения числа узлов и исключения лишних строк.

При работе с ПП 2000 пользователь, имеющий перед собой изображение таблицы, вводит числа в те клетки ее, где находится курсор. Признаком окончания ввода числа является нажатие клавиши ВК, после чего курсор перескакивает в следующую клетку. После заполнения последней строки таблицы происходит автоматическая смена кадра: выводятся последняя заполненная строка и шесть следующих пустых строк.

Ввод данных завершается нажатием клавиши "русское З". Это признак завершения работы всей ПП 2000. Однако при вводе постоянно возникает необходимость исправления ошибок, поэтому сначала надо просмотреть и проверить данные, скорректировать их и только потом завершать работу.

ПП 2000 выводит на экран лишь семь строк таблицы. Для полного ее просмотра нужно подвести курсор к последней строке и нажать клавишу "стрелка вниз", что приведет к отображению следующих шести строк вместе с последней (строки № 7–13) и т. д. А для того чтобы "листать" таблицу вверх, нужно подвести курсор к верхней строке и нажать клавишу "стрелка вверх".

Корректировка данных осуществляется следующим образом. Нажав несколько раз клавишу ВК, переместить курсор в крайний ле-

вый столбец. Далее с помощью клавиш со стрелками вверх и вниз переместить его по вертикали до строки, где следует внести изменения. И снова переместить курсор по горизонтали (нажимая ВК) до нужной клетки, внести правильное число и нажать клавишу ВК.

Для исключения целой строки следует установить курсор в левой крайней клетке удаляемой строки и нажать клавишу "русское И".

При включении ПП 2000 в одну из программ МКТ1, МКТ2, МУП1 или МУП2 необходимо предусмотреть заполнение массива U, в который нужно переслать данные из третьего столбца массива P (для МУП)

```
2500 FOR K = 1 TO N1
2520 U(K) = P(K, 3)
2520 NEXT K
2530 RETURN
```

или из четвертого столбца массива P (для метода УП) :

```
2500 FOR K = 1 TO N1
2510 U(K) = P(K, 4)
2520 NEXT K
2540 RETURN
```

▼ Далее опишем ПП 1700, которая позволяет выводить результаты расчета (также на постоянном токе) с использованием "электронной таблицы".



```
1690 REM *****
1691 REM * ПП ВЫВОДА РЕЗУЛЬТАТОВ РАСЧЕТА *
1692 REM * НА ЭКРАН ДИСПЛЕЯ *
1693 REM *****
1700 DIM N%(3)
1702 FOR I=4 TO 6
1704 N%(I-3)=N%(I)
1706 NEXT I
1708 N%(3)='' U1(B)''
1710 N%(4)='' U2(B)''
1712 N%(5)='' I(A)''
1714 GOSUB 2300
1716 LOCATE 0,(N7+1)*2+3
1718 PRINT STRING$(3,' ')
1720 N9=1
1722 GOSUB 2380
1724 N%=INKEY$
1726 IF N%='' THEN 1724
```

```

1728 IF N= '3' THEN 1738
1730 J=ASC(N)
1732 IF NOT(J=26 OR J=27) THEN 1724
1734 N9=N9+SGN(J-26.5)*N7
1736 GOTO 1722
1738 FOR I=4 TO 6
1740 N(I)=N(I-3)
1742 NEXT I
1744 RETURN
1750 REM === ЗАПОЛНЕНИЕ МАССИВА N ===
1752 N(4)=0
1754 IF N(2)>0 THEN N(4)=A(N(2),N3)
1756 N(5)=0
1758 IF N(3)>0 THEN N(5)=A(N(3),N3)
1760 N(6)=0
1762 IF P(K,2)=0 THEN 1766
1764 N(6)=(N(4)-N(5)+P(K,4))/P(K,2)+P(K,3)
1766 RETURN

```

Вначале она вычерчивает таблицу и записывает заголовки. Затем в каждой строке заполняются клетки: номер ветви, номера узлов (во втором и третьем столбцах), потенциалы узлов и ток ветви (табл. 5.3).

Таблица 5.3

Ветвь	Узел 1	Узел 2	U1[V]	U2[V]	I[A]
7	12	5	-3.15E-3	5.173E01	2.5
...

В нижней части экрана выводятся краткие пояснения по использованию таблицы. В тексте программы используются описанные выше ПП 2300 и ПП 2380. Программа выводит семь строк таблицы, начиная с первой. Если требуется посмотреть ее дальше, то нужно нажать клавишу "стрелка вниз", тогда кадр сменится и появятся строки с седьмой по тринадцатую и т. д. Таблицу можно "листать" (т. е. менять кадры) и в обратном порядке, нажимая клавишу "стрелка вверх". Просмотр таблицы завершается нажатием клавиши "русское З".

Приведенный текст ПП 1700 предназначен для расчетов по методу УП. Если же используется метод КТ, то строки 1752-1766 следует

заменить на иные:

```
1750 REM ЗАПОЛНЕНИЕ ВСПОМОГАТЕЛЬНОГО МАССИВА H
1752 LET H(4) = 0
1754 IF H(2) > 0 THEN H(4) = U(H(2))
1756 H(5) = 0
1758 IF H(3) = 0 THEN H(5) = U(H(5))
1760 LET H(6) = Z(K)
1762 RETURN
```

Кроме того, в текст программ МКТ1 или МКТ2 следует добавить следующий фрагмент для вычисления потенциалов узлов по токам:

```
1417 REM x x x x x x x x x x x x x x x x x
1418 REM ВЫЧИСЛЕНИЕ ПОТЕНЦИАЛОВ УЗЛОВ x
1419 REM x x x x x x x x x x x x x x x x x
1420 FOR I = 1 TO N2
1422 P(I, 1) = -P(I, 1)
1424 NEXT I
1426 FOR I = 1 TO N2
1428 IF P(I, 1) > 0 THEN 1456
1430 FOR J = 1 TO N1
1432 IF (FNA(J) - I) * (FNB(J) - I) <> 0 THEN 1454
1434 K = FNA(J) + FNB(J) - I
1436 L = 0
1438 IF K = 0 THEN 1442
1440 L = U(K)
1442 M = 1
1444 IF FNA(J) = I THEN 1448
1446 M = -1
1448 U(I) = L - M * ((Z(J) - P(J, 3)) * P(J, 2) - P(J, 4))
1450 P(I, 1) = -P(I, 1)
1452 GOTO 1456
1454 NEXT J
1456 NEXT I
1458 FOR I = 1 TO N2
1460 IF P(I, 1) < 0 THEN 1426
1462 NEXT I
```

Обычно ПП 2000 и ПП 1700 используются совместно. В случае, когда ПП 1700 пользователю не нужна, следует исключить в тексте ПП 2000 строку 2444.

▼ Опишем далее две простые программы, которые можно реализовать на ПЭВМ, не имеющей графических средств языка БЕЙСИК.

ПП 7210 предназначена для печати графика (под словом "печать" понимается здесь как вывод информации на печатающее устройство, так и отображение ее на экране). Исходными данными служат M0 — номер функции (номер строки, в которой она размещена), M1 и M2 — номера первого и последнего значений функции, M3 — ключ (при M3=1 изображается ось абсцисс, при M3=2 ее нет), M4, M5 и M6 — соответственно ширина зоны номеров точек, числовых значений и графика.



```

7200 REM *****
7202 REM * ПЕЧАТЬ ГРАФИКА ОДНОЙ ФУНКЦИИ *
7204 REM *****
7210 LET M7=M0
7212 LET M8=M1
7214 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
7216 LET L8=M9
7218 LET L9=M9
7220 FOR M8=M1+1 TO M2
7222 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
7224 IF M9 <= L8 THEN 7230
7226 LET L8=M9
7228 GOTO 7234
7230 IF M9 >= L9 THEN 7234
7232 LET L9=M9
7234 NEXT M8
7236 LET L7=2-M3
7238 LET L1=L8*SGN(1+SGN(L8)*L7)
7240 LET L2=L9*SGN(1-SGN(L9)*L7)
7242 LET L3=(L1-L2)/(M6-1+SGN(1-SGN(M6-1)))
7244 IF L3>0 THEN 7250
7246 PRINT "ГРАФИК ФУНКЦИИ - ПРЯМАЯ Y(X)=";L1
7248 RETURN
7250 LET LO=M4+M5+2
7252 LET L4=INT((-L2/L3+1.5)+LO)
7254 GOSUB 7288
7256 FOR M8=M1 TO M2
7258 REM GOSUB <ВЫЧИСЛЕНИЕ ФУНКЦИИ>
7260 LET L5=INT((M9-L2)/L3+1.5)+LO
7262 IF (M3-2)*(L4-L5)=0 THEN 7274
7264 IF L4>L5 THEN 7270
7266 PRINT M8;TAB(M4);M9;TAB(LO);"I";TAB(L4);".";TAB(L5);""
7268 GOTO 7276

```

```

7270 PRINT M8;TAB(M4);M9;TAB(LO);"I";TAB(L5);"";TAB(L4);"."
7272 GOTO 7276
7274 PRINT M8;TAB(M4);M9;TAB(LO);"I";TAB(L5);""
7276 NEXT M8
7278 GOSUB 7288
7280 PRINT "Y(MAX)=";L8,"Y(MIN)=";L9
7282 PRINT "DY=";L3
7284 GOSUB 7288
7286 RETURN
7288 PRINT "-----"
7290 RETURN

```

До начала работы в строки 7214, 7222 и 7258 нужно внести изменения в зависимости от того, как передаются значения функции в ПП. Наиболее простой способ передачи — запись значений функции в некоторый массив X. Пусть, например, в строках 1, 2, 3 этого массива расположены три разные функции U_a , U_b и U_c , а печатать следует U_c с 10-го по 70-й элемент. Тогда присвоим $M0=3$, $M1=10$ и $M2=70$, а в перечисленных строках запишем

```
LET M9 = X(M7, M8)
```

Можно также непосредственно вычислять значения функции, что целесообразно при ее достаточно простом виде. Строки 7214, 7222 и 7258 нужно записать в виде

```
LET M0 = FNA (M8)
```

где функция должна быть описана в ГП с помощью оператора DEF. В качестве примера последней ниже проведен текст, в котором все исходные данные вводятся в диалоге.

```

3000 REM *****
3002 REM * ТЕСТ ДЛЯ ПРОВЕРКИ ПП 7210 *
3004 REM *****
3010 DIM M[3,100]
3020 PRINT "ПП 7210 ПОДГОТОВЛЕНА К РАБОТЕ (1-ДА,0-НЕТ) ?"
3030 INPUT W
3040 IF W=1 THEN 3090
3050 PRINT "ВСТАВЬТЕ В ПП 7210 ПОД НОМЕРАМИ 7215,"
3060 PRINT "7223 И 7259 ОПЕРАТОР '<N> LET M9=M(M7,M8)' "
3070 PRINT "И СНОВА ПЕРЕЗАПУСТИТЕ ПРОГРАММУ."
3080 STOP
3090 PRINT "ПОДГОТОВКА СИГУСИДАЛЬНОГО СИГНАЛА,"

```



```

3100 PRINT "ВВЕДИТЕ ЕГО ПАРАМЕТРЫ: АМПЛИТУДУ, ПОСТОЯННУЮ"
3110 PRINT "СОСТАВЛЯЮЩУЮ, ФАЗУ (В ГРАД.), ПЕРИОД, ШАГ"
3120 PRINT "КВАНТОВАНИЯ ПО ВРЕМЕНИ И ЧИСЛО ТОЧЕК."
3130 PRINT "А,В,Ф,Т,Н,Н=";
3140 INPUT А,В,Ф,Т,Н,Н
3150 PRINT "В КАКОЙ СТОЛБЕЦ МАССИВА М ЗАНЕСТИ ЗНАЧЕНИЯ"
3160 PRINT "СИНУСА ?"
3170 INPUT М0
3180 PRINT "УКАЖИТЕ ПЕРВУЮ И ПОСЛЕДНЮЮ ТОЧКИ ГРАФИКА ."
3190 INPUT М1,М2
3200 PRINT "НУЛЕВУЮ ЛИНИЮ (ОСЬ АБСЦИСС) ПЕЧАТАТЬ "
3210 PRINT "(1-ДА,0-НЕТ) ?"
3220 INPUT М3
3230 LET М3=2-М3
3240 PRINT "ЗАДАЙТЕ ШИРИНУ ЗОНЫ НОМЕРОВ ТОЧЕК, ЗОНЫ"
3250 PRINT "ЗНАЧЕНИЙ ФУНКЦИИ И ЗОНЫ ГРАФИКА."
3260 INPUT М4,М5,М6
3270 LET Р=3.14159
3280 LET F0=F*Р/180
3290 LET Н0=2*Р*Н/Т
3300 FOR I=1 TO N
3310 LET М[М0, I]=А*SIN(F0)+В
3320 LET F0=F0+Н0
3330 NEXT I
3340 GOSUB 7210
3350 STOP

```

Протокол работы этой программы приведен далее.

RUN

ПП 7710 НАСТРОЕНА ДЛЯ РАБОТЫ (1-ДА,0-НЕТ) ?
?1

ПОДГОТОВКА СИНУСОИДАЛЬНОГО СИГНАЛА,
ВВЕДИТЕ ЕГО ПАРАМЕТРЫ: АМПЛИТУДУ, ПОСТОЯННУЮ
СОСТАВЛЯЮЩУЮ, ФАЗУ (В ГРАД.), ПЕРИОД, ШАГ
КВАНТОВАНИЯ ПО ВРЕМЕНИ И ЧИСЛО ТОЧЕК.
А,В,Ф,Т,Н,Н=?1,0,0,20,1,100

В КАКОЙ СТОЛБЕЦ МАССИВА М ЗАНЕСТИ ЗНАЧЕНИЯ
СИНУСА ?
?3

УКАЖИТЕ ПЕРВУЮ И ПОСЛЕДНЮЮ ТОЧКИ ГРАФИКА .
?30,53



Помимо массива X размерностью $K1 \times 3$ (третий столбец нужен для хранения номеров переставляемых величин) в ПП используется вспомогательный массив K размерностью $10 + K2$. Оба массива X и K должны быть описаны в ГП.

```

7500 REM *****
7502 REM * ПЕЧАТЬ ГОДОГРАФА КОМПЛЕКСНОЗНАЧНОЙ *
7504 REM * ФУНКЦИИ . *
7506 REM *****
7510 LET K9=K1+1
7512 LET K[1]=X[1,1]
7514 LET K[2]=K[1]
7516 LET K[4]=X[1,2]
7518 LET K[5]=K[4]
7520 FOR KO=1 TO K1
7522 FOR K=1 TO 2
7524 LET K7=X[KO,K]
7526 IF K[3*K-2] >= K7 THEN 7530
7528 LET K[3*K-2]=K7
7530 IF K[3*K-1] <= K7 THEN 7534
7532 LET K[3*K-1]=K7
7534 NEXT K
7536 IF K4=0 THEN 7540
7538 LET X[KO,3]=K1-KO
7540 NEXT KO
7542 LET K[3]=(K[1]-K[2])/(K2-1)
7544 LET K[6]=(K[4]-K[5])/(K3-1)
7546 LET K9=SGN(K[3])*SGN(K[6])
7548 IF K9>0 THEN 7570
7550 LET K9=K[3]+K[6]
7552 IF K9>0 THEN 7558
7554 PRINT "ГОДОГРАФ - ТОЧКА С КООРДИНАТАМИ : ";K[1];K[2]
7556 RETURN
7558 PRINT "ТОЧКИ ГОДОГРАФА ПРИНАДЛЕЖАТ";
7560 IF K[6]=0 THEN 7566
7562 PRINT "ПРЯМОЙ Y(X)=";K[1]
7564 RETURN
7566 PRINT "ПРЯМОЙ X(Y)=";K[4]
7568 RETURN
7570 LET K9=2
7572 GOSUB 7684

```



```

7574 PRINT "Y(MAX)=";K[4];TAB(25);"Y(MIN)=";K[5]
7576 PRINT "X(MAX)=";K[1];TAB(25);"X(MIN)=";K[2]
7578 PRINT "DY=";K[6];TAB(25);"DX=";K[3]
7580 GOSUB 7672
7582 LET K[7]=INT(-K[2]/K[3]+1.5)
7584 IF SGN(K[7]-K[1])*SGN(K[7]-K[2])>0 THEN 7588
7586 LET K[7]=0
7588 LET K[8]=INT(K[4]/K[6]+1.5)
7590 IF SGN(K[8]-K[4])*SGN(K[8]-K[5])>0 THEN 7594
7592 LET K[8]=0
7594 LET K9=1
7596 FOR K=1 TO K3
7598 LET K8=SGN(ABS(K-K[8]))
7600 FOR KO=1 TO K2
7602 LET K[10+KO]=1
7604 NEXT KO
7606 IF K9>K1 THEN 7618
7608 IF INT((K[4]-X[K9,2])/K[6]+1.5)>K THEN 7618
7610 LET K7=10+INT((X[K9,1]-K[2])/K[3]+1.5)
7612 LET K[K7]=K[K7]+1
7614 LET K9=K9+1
7616 GOTO 7606
7618 IF K[7]=0 THEN 7624
7620 IF K[10+K[7]]>1 THEN 7624
7622 LET K[10+K[7]]=0
7624 PRINT "I";
7626 FOR KO=1 TO K2
7628 LET K7=K[10+KO]
7630 IF K7>0 THEN 7636
7632 PRINT ".";
7634 GOTO 7656
7636 IF K7>1 THEN 7648
7638 IF K8=0 THEN 7644
7640 PRINT " ";
7642 GOTO 7656
7644 PRINT " .";
7646 GOTO 7656
7648 IF K7>2 THEN 7654
7650 PRINT "+";
7652 GOTO 7656
7654 PRINT "***";

```

```

7656 NEXT KO
7658 PRINT "I"
7660 NEXT K
7662 GOSUB 7672
7664 IF K4=0 THEN 7670
7666 LET K9=3
7668 GOSUB 7684
7670 RETURN
7672 PRINT "I";
7674 FOR K=1 TO K2
7676 PRINT "-";
7678 NEXT K
7680 PRINT "I"
7682 RETURN
7684 LET KO=0
7686 GOTO 7706
7688 IF X[K,K9]>X[K+1,K9] THEN 7706
7690 FOR K8=1 TO 2+K4
7692 LET K7=X[K,K8]
7694 LET X[K,K8]=X[K+1,K8]
7696 LET X[K+1,K8]=K7
7698 NEXT K8
7700 IF K=1 THEN 7706
7702 LET K=K-1
7704 GOTO 7688
7706 LET KO=KO+1
7708 LET K=KO
7710 IF K<K1 THEN 7688
7712 RETURN

```

Приложение 1. Пример расчета длинных линий

Приведем пример вычисления вторичных параметров длинной линии по первичным при заданной частоте с помощью подпрограмм, описанных в § 3.5.

Составим головную программу

```

100 DIM Z(10, 3), K(10)
110 LET K(1) = 1
120 LET K(2) = 8
130 LET K(3) = 4
140 PRINT "ВВЕДИТЕ ЧЕРЕЗ ЗАПЯТУЮ УДЕЛЬНЫЕ ПАРАМЕТРЫ"

```

```

150 PRINT "ЛИНИИ: R0 (ОМ/КМ), L0(ГН/КМ), G0(СМ/КМ)"
160 PRINT "И C0 (Ф/КМ)"
170 INPUT Z(K(1), 1), Z(K(1), 2), Z(K(1) + 1, 1), Z(K(1) + 1, 2)
180 PRINT "УКАЖИТЕ ЧАСТОТУ В ГЦ)"
190 INPUT F
200 GO SUB 4600
210 PRINT "B="; Z(K(2), 1); " + J*"; Z(K(2), 2): "ОМ"
220 PRINT "C="; Z(K(2) + 1, 1); " + J*"; Z(K(2) + 1, 2): "1/КМ
230 STOP

```

В ней в строках 110–130 указаны номера строк массива Z, в которых должны храниться исходные данные и результаты. Затем (строки 140–190) в диалоге вводятся данные. После вызова ПП 4600 на экране отображаются результаты расчета.

Протокол одного расчета приведен ниже.

RUN

ВВЕДИТЕ ЧЕРЕЗ ЗАПЯТУЮ УДЕЛЬНЫЕ ПАРАМЕТРЫ
ЛИНИИ R0(ОМ/КМ), L0(ГН/КМ), G0(СМ/КМ)
И C0 (Ф/КМ)

? 6.5, 2.29E-3, .5E-6, 5.22E-9

УКАЖИТЕ ЧАСТОТУ В ГЦ

?1000

B = 6.78159E-2 + J* 1.45318E-2 ОМ

C = 4.9198E-3 + J* 2.18998E-2 1/КМ

Приложение 2. Программа "Комплексный калькулятор"

Приведенными в данной книге программами охватывается лишь небольшая часть вычислительных задач в электротехнике. Как быть в остальных случаях? Можно надеяться, что программа, позволяющая оперировать с комплексными числами (С-числами) так, как это осуществляется в обычных непрограммируемых микрокалькуляторах, может найти свое применение при решении других задач. Кроме того, она может оказаться полезной при обучении методике расчетов.

Одним из возможных вариантов "Комплексного калькулятора" (КК) реализованного на подмножестве языка БЕЙСИК, который описан в гл. 1, может быть программа КК, которая имитирует действия пользователя при работе на обычном непрограммируемом калькуляторе с действительными числами.

Программа КК выполняет следующие процедуры:

а) простые операции с С-числами (сложение, вычитание, умножение, деление и возведение в степень);

б) вычисление модуля и аргумента, а также экспоненты и логарифмы С-чисел;

в) вычисление гиперболических функций;

г) запоминание и считывание информации из регистров памяти.

КК содержит ряд элементарных ПП, комбинации которых обеспечивают выполнение перечисленных выше процедур. В отличие от ПП, описанных в гл. 5, здесь в каждой ПП не только находится результат вычислений, но и проверяется его принадлежность допустимому множеству решений. Например, если при делении одного числа на другое результат по абсолютной величине превышает верхнюю границу представляемых в КК чисел (если делитель много меньше делимого) и возникает ситуация, аналогичная делению на нуль, то вместо чисел на экране выводится сообщение "ERROR", вычисления прерываются и затем пользователю предлагается снова ввести код операции.

В том случае, когда результат вычисления в какой-либо ПП по абсолютной величине много меньше единицы и не равен нулю (т. е. не представим на "индикаторе калькулятора"), то производится округление его до нуля и вычисления продолжают. Таким образом, в отличие от интерпретатора языка БЕЙСИК осуществляется (моделируется) одна из функций калькулятора — проверка результата на переполнение. Кроме того, использование элементарных ПП позволяет избежать ненужного дублирования операций и существенно сократить текст КК.

Каждая элементарная операция кодируется четырехзначным числом. В нем не только содержится информация о номере вызываемой ПП, но и указывается месторасположение операндов и результата вычислений. Списки кодов приведены в строках 210–240 программы (см. ее текст). Строки 300–512 организуют загрузку констант при работе — в массив N вводятся коды, задаются числовые константы (π , e и верхняя граница используемых в КК чисел).

Строки 1000–1092 организуют взаимодействие пользователя с программой. Диалог организован очень кратко и может быть развернут и дополнен выводом звуковой, строковой и графической информации для того, чтобы улучшить комфортабельность программы.

В строках 3000–3142 коды расшифровываются и вызываются нужные ПП, размещенные в строках 3200–3588.

В целом структура программы КК является "открытой", т. е. позволяет легко изменять ее текст, дает возможность и усовершенствовать его: а) сократить длину и время счета; б) сделать диалог более "дружественным"; в) расширить возможности программы, добавив новые процедуры.

Остановимся кратко на практической работе с КК.

Программа запускается на выполнение (т. е. калькулятор "включается") обычной в БЕЙСИКе командой RUN.

Для выполнения операций и вычислений требуется ввести следующую информацию: а) код операции и б) одно С-число (если операция

одноместная, т. е. требующая одного числа) или два С-числа (если операция двухместная).

При выполнении операций используются два комплексных "регистра" – регистр X (PX) и регистр Y (PY). Числа при вводе заносятся в PX, а из него можно записать число в PY. Результат вычислений всегда помещается в регистр X. Кроме того, для сохранения и последующего использования промежуточных результатов предусмотрено еще 99 "регистров памяти", в которые можно заносить С-числа из PX, считывать из них информацию, пересылая ее в PX, и производить обмен информацией между любыми из этих регистров. (Все они предназначены для хранения С-чисел.)

Программа КК работает в режиме диалога. После ее запуска на экране дисплея появляется запрос:

КОД ОПЕРАЦИИ (0 – ВЫЗОВ МЕНЮ) = ?

В ответ пользователь выбирает одну из допустимых операций и вводит ее код с клавиатуры. Если пользователь не помнит код, то вводит в ответ 0 и получает на экране меню – перечень всех операций КК и соответствующие коды в виде следующей таблицы.

Код	Операция	Код	Операция	Код	Операция
-2	КОНЕЦ	-1	PN → PM	0	МЕНЮ
1	C → PX	2	PX → PY	3	PX + PY
4	PY - PX	5	PY * PY	6	PY/PX
7	PX ↑ PX	8	(MOD, APG)	9	(RE, IM)
10	EXP	11	LOG	12	SH
13	CH	14	TH	15	ARC TH

Поясним обозначения в тексте таблицы

"КОНЕЦ" означает "Выключение калькулятора", т. е. завершение работы программ КК.

PN → PM означает пересылку содержимого регистра PN в PM (регистр с номером M), "МЕНЮ" – вывод приведенной выше таблицы на экран дисплея, C → PX означает ввод С-числа в регистр X, а PX → PY – пересылку из PX в PY. PX + PY означает сложение комплексных чисел: к содержимому регистра Y прибавляется содержимое регистра X. Аналогично этому обозначены и следующие четыре операции. Следует обратить внимание на то, что при вычитании, делении и возведении в степень сначала надо ввести вычитаемое, делимое и показатель степени и поместить их в регистр Y. PX ↑ PY означает возведение числа из регистра X в степень, показатель которой находится в PY. Если на данной ПЭВМ степень обозначается иным символом, то целесообразно заменить его в строке 1610 теста программы.

(MOD, ARG) означает, что для С-числа находится модуль и аргумент, а (RE, IM) соответствует обратному преобразованию. Остальные функции обозначены обычной математической нотацией (LOG – натуральный логарифм).

Ввод С-числа в КК осуществляется следующим образом. На приведенный выше запрос вводится код операции, равный 1. На экране появляется запрос-подсказка, в каком порядке вводить компоненты С-числа:

RE(X), IM(X) =

Это означает, что сначала следует ввести действительную часть, а затем и мнимую компоненту С-числа, отделив ее запятой.

Приведем пример выполнения операции сложения двух комплексных чисел. Надо найти $Z = Z_1 + Z_2$, где $Z_1 = 1 + j2$ и $Z_2 = 3 + j4$. Сначала в ответ на запрос о коде операции отвечаем 1 и вводим 1 и 2 (части первого числа). Это число следует переслать из РХ в РУ. Затем вводится в РХ второе число и код операции сложения. На экране появляется ответ. Эти действия изобразим в виде небольшой таблицы

Сообщение или запрос ЭВМ	Ответ пользователя
1 КОД =	1
2 RE(X), IM(X) =	1,2
3 КОД =	2
4 КОД =	1
5 RE(X), IM(X) =	3,4
6 КОД =	3
7 RE(X) = 4, IM(X) = 6	—

Аналогичные действия совершаются и при выполнении других операций.

Как отмечалось ранее, в основу построения диалога в программе КК положена аналогия с действиями пользователя с непрограммируемым микрокалькулятором. Эта идея представляется плодотворной для построения различных диалоговых систем программ прикладного характера. На ее основе могут быть, например, построены такие диалоговые программы, как "Матричный калькулятор", "Алгебраический калькулятор", "Калькулятор-интегратор" и "Электрический калькулятор".

```

1 REM *****
2 REM * КОМПЛЕКСНЫЙ КАЛЬКУЛЯТОР *
3 REM *****
10 DIM N(20,3),L(150),P(10),Z(101,2)
100 REM ---ГРАНИЦЫ---
110 DATA 3,2,4,4,5,11,6,24,8,31,9,40

```

```

112 DATA 10,45,11,51,12,67,13,83,14,102
114 DATA 15,123,7,140,0,0
200 REM ---КОДЫ ЭЛЕМЕНТАРНЫХ ОПЕРАЦИЙ---
210 DATA 1313,2423,1134,2244,5315,6425,5654,6145,1058
212 DATA 5235,2563,5115,6225,7563,2022,5315,6425,5654
214 DATA 6145,1058,5235,2563,1716,2726,5115,6225,5563
216 DATA -2125,1051,3018,-4027,-5021,-6022,2515,1615,-4046
218 DATA -5041,-6042,4535,3635,-5021,-6022,-7018,1765,2755
220 DATA 5115,6225,5563,5051,-2125,-1059,-5021,-6022,-7018
222 DATA 3765,4755,1012,2022,-5021,-6022,-7018,1765,2755
224 DATA 1134,2244,1019,2029,-5021,-6022,-7018,3765,4755
226 DATA 1012,2022,-5021,-6022,-7018,1765,2755,1133,2243
228 DATA 1019,2029,1113,2223,-5021,-6022,-7018,1765,2755
230 DATA 4114,-3048,1313,1019,2029,5115,6225,5563,2526
232 DATA 1516,1134,1018,4114,-3048,7333,1314,5115,6225
234 DATA 5653,1516,2526,1715,2725,1313,1012,5115,6225
236 DATA 5563,5051,-5059,-6125,1059,2069,5115,6225,7563
238 DATA 5071,-2125,-1059,5315,6425,7654,5415,6325,8563
240 DATA -7078,-5081,-6082,2575,1675
300 REM ---КОНСТАНТЫ: D=МАХ,С1='ПИ',С2='Е'---
310 LET D=1E+10
320 LET C1=3.1415926
330 LET C2=2.71828
400 REM ---НАСТРОЙКА ПРОГРАММЫ (ЗАПОЛНЕНИЕ МАССИВОВ)---
410 FOR I=1 TO 100
412 READ N(I,1),N(I,3)
414 IF N(I,3)=0 THEN 420
416 LET N(I+1,2)=N(I,3)+1
418 NEXT I
420 LET N(1,2)=1
422 LET M=I-1
424 FOR I=1 TO N(M,3)
426 READ L(I)
428 NEXT I
500 FOR I=1 TO 101
502 LET Z(I,1)=0
504 LET Z(I,2)=0
506 NEXT I
510 GOSUB 1000
512 STOP
1000 REM ---ВВОД ДАННЫХ В РЕЖИМЕ ДИАЛОГА---

```

```

1010 PRINT "МОД (0-ВНЗОВ МЕНД) = ";
1012 INPUT K
1013 LET G3=K
1014 IF K<=0 THEN 1020
1016 GOSUB 1500
1018 GOTO 1010
1020 IF K=-2 THEN 1092
1022 IF K>0 THEN 1036
1024 PRINT "PN->PM: N,M= ";
1026 INPUT K1,K2
1028 FOR I=1 TO 2
1030 LET Z(K2+1,I)=Z(K1+1,I)
1032 NEXT I
1034 GOTO 1088
1036 IF K>1 THEN 1044
1038 PRINT "RE(X),IM(X)= ";
1040 INPUT Z(1,1),Z(1,2)
1042 GOTO 1010
1044 IF K>2 THEN 1052
1046 LET K1=0
1048 LET K2=1
1050 GOTO 1028
1052 LET Y1=Z(2,1)
1054 LET Y2=Z(2,2)
1056 FOR I0=1 TO M
1058 IF K=ABS(N(I0,1)) THEN 1064
1060 NEXT I0
1062 GOTO 1016
1064 LET G1=N(I0,2)
1066 LET G2=N(I0,3)
1067 LET G3=K
1068 FOR G=G1 TO G2
1070 LET K=L(G)
1072 GOSUB 3000
1074 IF E<1 THEN 1080
1076 PRINT " ERROR "
1078 GOTO 1082
1080 NEXT G
1081 IF (G3-8)*(G3-9)=0 THEN 1086
1082 LET Z(2,1)=Y1
1084 LET Z(2,2)=Y2

```

```

1086 IF B=1 THEN 1010
1088 PRINT "R(X)= ";Z(1,1)," IM(X)= ";Z(1,2)
1089 IF ABS(G3-8)*ABS(G3-9)<=0 THEN 1010
1090 PRINT "RE(Y)= ";Z(2,1)," IM(Y)= ";Z(2,2)
1091 GOTO 1010
1092 RETURN
1500 REM ---МЕНЮ (СПИСОК ОПЕРАЦИЙ И ИХ КОДОВ)---
1510 FOR K=1 TO 3
1520 PRINT TAB(1+(K-1)*20);" КОД   ОПЕРАЦИЯ ";
1530 NEXT K
1540 PRINT
1550 PRINT " -2   КОВЕР ";TAB(21);" -1   ПМ<->ПМ ";
1560 PRINT TAB(41);" 0   МЕНЮ "
1570 PRINT "  1   С->PX ";TAB(21);"  2   FX->PY ";
1580 PRINT TAB(41);"  3   PY+PX "
1590 PRINT " 44   PY-PX ";TAB(21);"  5   PY*PX ";
1600 PRINT TAB(41);"  6   PY/PX "
1610 PRINT "  7   PX-PY ";TAB(21);"  8   (MOD,ARG) ";
1620 PRINT TAB(41);"  9   (RE,IM) "
1630 PRINT " 10   EXP ";TAB(21);" 11   LOG ";TAB(41);
1640 PRINT " 12   SH "
1650 PRINT " 13   CH ";TAB(21);" 14   TH ";TAB(41);
1660 PRINT " 15   ARC TH "
1670 RETURN
3000 REM ---ВЫПОЛНЕНИЕ ЭЛЕМЕНТАРНОЙ ОПЕРАЦИИ---
3010 LET A=SGN(K)
3012 LET K=ABS(K)
3014 FOR I=1 TO 4
3016 LET P(I)=K-INT(K/10)*10
3018 LET K=INT(K/10)
3020 NEXT I
3022 LET P(1)=P(1)*A
3024 FOR I=2 TO 3
3025 IF P(I)=0 THEN 3038
3026 IF P(I)>4 THEN 3036
3028 LET J=INT((P(I)+1)/2)
3030 LET K=P(I)-(J-1)*2
3032 LET P(I)=Z(J,K)
3034 GOTO 3038
3036 LET P(I)=P(P(I)+1-SGN(P(I)))
3038 NEXT I

```

```
3040 IF P(1)>-9 THEN 3046
3042 GOSUB 3580
3044 GOTO 3124
3046 IF P(1)>-8 THEN 3052
3048 GOSUB 3560
3050 GOTO 3124
3052 IF P(1)>-6 THEN 3058
3054 GOSUB 3530
3056 GOTO 3124
3058 IF P(1)>-5 THEN 3064
3060 GOSUB 3400
3062 GOTO 3124
3064 IF P(1)>-4 THEN 3070
3066 GOSUB 3510
3068 GOTO 3124
3070 IF P(1)>0 THEN 3076
3072 GOSUB 3470
3074 GOTO 3124
3076 IF P(1)>1 THEN 3082
3078 LET P(3)=-.5
3080 GOTO 3108
3082 IF P(1)>2 THEN 3088
3084 LET P(3)=-1
3086 GOTO 3096
3088 IF P(1)>4 THEN 3096
3090 GOSUB 3210
3092 GOTO 3124
3094 IF P(1)>5 THEN 3100
3096 GOSUB 3240
3098 GOTO 3124
3100 IF P(1)>6 THEN 3106
3102 GOSUB 3270
3104 GOTO 3124
3106 IF P(1)>7 THEN 3112
3108 GOSUB 3330
3110 GOTO 3124
3112 IF P(1)>8 THEN 3118
3114 LET R=P(2)
3116 GOTO 3122
3118 LET P(3)=2
3120 GOTO 3102
```

```

3122 LET E=0
3124 IF E=1 THEN 3142
3126 IF E=0 THEN 3130
3128 LET R=0
3130 IF P(4)>4 THEN 3140
3132 LET J=INT((P(4)+1)/2)
3134 LET K=P(4)-(J-1)*2
3136 LET Z(J,K)=R
3138 RETURN
3140 LET P(P(4))=R
3142 RETURN
3200 REM---СЛОЖЕНИЕ И ВЫЧИТАНИЕ---
3210 LET E=1
3212 IF SQN(P(3))*SQN(P(2))*SQN(3.5-P(1))<=0 THEN 3216
3214 IF ABS(P(2))>D-ABS(P(3)) THEN 3220
3216 LET R=P(2)+P(3)*SQN(3.5-P(1))
3218 LET E=0
3220 RETURN
3230 REM---УМНОЖЕНИЕ---
3240 LET E=0
3242 IF SQN(P(2))*SQN(P(3))=0 THEN 3248
3244 GOSUB 3300
3246 IF E=1 THEN 3250
3248 LET R=(SQN(E+1)*P(2))*P(3)
3250 RETURN
3260 REM---ДЕЛЕНИЕ---
3270 LET E=1
3272 IF P(3)=0 THEN 3284
3274 LET E=0
3276 IF P(2)=0 THEN 3282
3278 GOSUB 3300
3280 IF E=1 THEN 3284
3282 LET R=(SQN(E+1)*P(2))/P(3)
3284 RETURN
3290 REM---ВСПОМОГАТЕЛЬНАЯ ПП---
3300 LET C=LOG(ABS(P(2)))+LOG(ABS(P(3)))*SQN(5.5-P(1))
3302 LET E=0
3304 IF C<=LOG(D) THEN 3310
3306 LET E=1
3308 RETURN

```

```

3310 IF C>=-LOG(D) THEN 3314
3312 LET E=-1
3314 RETURN
3320 REM---ВОЗВЕДЕНИЕ В СТЕПЕНЬ---
3330 LET E=1
3332 IF P(2)≤0 THEN 3342
3334 IF P(3)≤0 THEN 3340
3336 LET E=0
3338 LET R=0
3340 RETURN
3342 LET Z=1
3344 IF P(2)>0 THEN 3348
3346 IF P(3)≤INT(P(3)) THEN 3340
3347 LET Z=SGN(.5-P(3)+INT(P(3)/2)*2)
3348 LET A=D
3350 LET D=LOG(D)
3352LET B=P(2)
3354 LET P(2)=LOG(ABS(B))
3356 LET P(1)=5
3358 GOSUB 3240
3360 LET D=A
3362 LET P(2)=B
3364 IF P(3)>0 THEN 3368
3366 LET E=-E
3368 IF E=1 THEN 3376
3370 IF E=0 THEN 3378
3372 LET E=0
3374 LET R=1
3376 RETURN
3378 LET R=Z*ABS(P(2))-P(3)
3380 RETURN
3390 REM---ВЫЧИСЛЕНИЕ ARC TG---
3400 LET P(1)=6
3402 GOSUB 3270
3404 IF E=0 THEN 3416
3406 IF E=1 THEN 3412
3408 LET R=0
3410 GOTO 3422
3412 LET R=C1/2
3414 GOTO 3418
3416 LET R=ATN(R)

```

```

3418 IF P(2)>0 THEN 3422
3420 LET R=-R
3422 IF P(2)>=0 THEN 3426
3424 LET R=(C1-ABS(R))*SGN(SGN(R)+.5)
3426 LET E=0
3428 RETURN
3440 REM---ПРЕОБРАЗОВАНИЕ УГЛОВ---
3450 LET P(1)=6
3452 LET P(3)=2*C1
3454 GOSUB 3270
3456 LET R=P(2)-INT(R)*P(3)
3458 RETURN
3460 REM---ВЫЧИСЛЕНИЕ SIN, COS, TG---
3470 LET P1=P(1)
3471 GOSUB 3450
3472 IF P1=-2 THEN 3480
3474 IF P1=-3 THEN 3484
3476 LET R=SIN(R)
3478 RETURN
3480 LET R=COS(R)
3482 RETURN
3484 LET P(1)=6
3486 LET P(2)=SIN(R)
3488 LET P(3)=COS(R)
3490 GOSUB 3270
3492 RETURN
3500 REMВЫЧИСЛЕНИЕ      ATN---
3510 LET E=0
3512 LET R=ATN(P(2))
3514 RETURN
3520 REM---ГРАД.<---->РАД.---
3530 LET A=P(1)
3532 LET P(1)=6
3534 LET P(3)=180*SGN(7+A)+C1*SGN(A+6)
3536 GOSUB 3270
3538 LET P(1)=5
3540 LET P(2)=R
3542 LET P(3)=C1*SGN(7+A)+180*SGN(A+6)
3544 GOSUB 3240
3546 RETURN
3550 REM---ВЫЧИСЛЕНИЕ EXP---

```

```

3560 LET P(1)=7
3562 LET P(3)=P(2)
3564 LET P(2)=C2
3566 GOSUB 3330
3568 RETURN
3570 REM---ВЫЧИСЛЕНИЕ LOG---
3580 LET E=1
3582 IF P(2)<=0 THEN 3588
3584 LET E=0
3586 LET R=LOG(P(2))
3588 RETURN
9999 END

```

СПИСОК ЛИТЕРАТУРЫ

1. Кетков Ю. Л. Программирование на БЕЙСИКе. М.: Статистика, 1978.
2. Уорт Т. Программирование на языке БЕЙСИК. Пер. с англ. М.: Машиностроение, 1981.
3. Программирование на БЕЙСИК-ПЛЮС для СМ-4/ В. П. Семик, Б. Р. Монцибович, Д. П. Непочатых и др. М.: Финансы и статистика, 1982.
4. Пул Л. Работа на персональном компьютере: Пер. с англ. М.: Мир, 1986.
5. Трейстер Р. Персональная ЭВМ IBM PC: Пер. с англ. М.: 1986.
6. Kemeny J. G., Kurtz T. E. BASIC programming. N. Y., Wiley, 1971.
7. Фаулджер М. Программирование встроенных микропроцессоров: Пер. с англ. М.: Мир, 1985.
8. Peterson S.// Comput. Des. 1982. Vol. 21 N 10. P. 103-114.
9. Dearnley P. A.// Comput. J. 1982. Vol. 25, N 2. P. 253-256.
10. Denning W., Essig H., Maos S. The adaptation of virtual man-computer interface to users requirements in dialogs. Berlin: Springer, 1981.
11. Шаповалов Ю. И., Шийка Я. В.// Изв. вузов. Радиоэлектроника. 1985. Т. 28, № 6 С. 74-76.
12. Teskey F. N. Principles of text processing. Chichester: Horwood, 1982.
13. Первин Ю. А., Шевякова Т. К. Динамические информационные системы на предприятии. М.: Статистика, 1975.
14. Ван Тассел Ф. Стил, разработка, эффективность, отладка и испытание программ: Пер. с англ. М.: Мир, 1985.
15. Николаев Ф. А., Фомин В. Н., Хохлов А. М. Проблемы повышения достоверности в информационных системах. Л.: Энергоатомиздат, 1982.
16. Ефременко Ю. А.// Электронная техника. Сер. Экономика и системы управления. 1985. Вып. 4. С. 60-61.
17. Абрамович С. Н., Бойко В. В., Бутрин Б. П. и др.// Микропроцессорные средства и системы. 1985, № 2. С. 29-36.
18. Кожарский В. Г., Орехов В. И. Методы автоматизированного проектирования источников вторичного электропитания. М.: Сов. радио. 1985.
19. Великов В. А. Переходные электромагнитные процессы в электрических схемах. М.: Высшая школа, 1985.
20. Чуа Л. О., Лин П. М. Машинный анализ электрических схем: Пер. с англ. М.: Энергия, 1980.
21. Дмитришин Р. В. Оптимизация электронных схем на ЭВМ. Киев: Техника, 1980.

22. Сигорский В. Г., Петренко А. И. Алгоритмы анализа электронных схем. М.: Сов. радио, 1976.
23. Калахан Д. Методы машинного расчета электронных схем: Пер. с англ. М.: Мир, 1970.
24. Мельников Н. А. Матричный метод анализа электрических цепей. М.: Энергия, 1972.
25. Vaske P. Elektrotechnik und BASIC-Rechnern (SHARP). Stuttgart: Teubner, 1984. Т. 1. Grundlagen, Wechselstrom.
26. Бессонов Л. А. Теоретические основы электротехники. Электрические цепи. М.: Высшая школа, 1978.
27. Атабеков Г. И. Теоретические основы электротехники. Ч 1. М.: Энергия, 1978.
28. Нейман Л. Р., Демирчян К. С. Теоретические основы электротехники. Т. 1. – 3-е изд. Л.: Энергоиздат, 1981.
29. Матханов П. Н. Основы анализа электрических цепей. Линейные цепи: Учебник для вузов. – 2-е изд. М.: Высшая школа, 1981.
30. Бармаков Ю. Н., Бахов В. А., Ильин В. Н. и др.// Изв. вузов. Радиоэлектроника. 1981. Т. 24, вып. 6. С. 27- 37.
31. Брамеллер А., Аллан Р., Хэмэм Я. Слабозаполненные матрицы: Анализ электроэнергетических схем: Пер. с англ. М.: Энергия, 1979.
32. Павлюченко А. А., Стерлин А. М. Программа решения линейных алгебраических уравнений со слабозаполненными матрицами коэффициентов. Инф. бюл. "Алгоритмы и программы". ВПТИЦ. М.: 1982, № 7, (51), с. 46.
33. Седнев А.М.// Электричество. 1980, № 8 С. 61–64.
34. Щедрин Н. Н. Упрощение электрических систем при моделировании. М.–Л.: Энергия, 1966.
35. Дмитришин Р. В. Оптимизация электронных схем на ЭВМ Киев: Техніка, 1980.
36. Непомнящий В. А.// Кибернетика 1984, № 2. С. 21- 28.
37. Головкин Б.А.// Автоматика и телемеханика, 1968, № 7. С. 5–20.
38. Гамм А. З. Статистические методы оценивания состояния электроэнергетических систем. М.: Наука, 1976.
39. Ушаков И. А., Литвак Е. И.// Изв. АН СССР. Техническая кибернетика, 1977, № 1. С. 72–80.
40. Ellison D., Herschdorf I., Wilson J. T.// Simulation, 1982. Vol. 38, N 5. p. 161–175.
41. Miller R. R.// Simulation, 1982. Vol. 38, N 6. p. 215–220.
42. Геворкян Г. Х., Геворкян Э. Т.// Тр. ВНИИЭМ. М.: 1966, т. 23, с. 97–101.
43. Фокин Ю. А., Арсамаков И. И.// Электричество, 1972, № 10, с. 23–28.
44. Ильин В. И. Машинное проектирование электронных схем. М.: Энергия, 1972.
45. Геворкян Г. Х., Геворкян В. Х., Иванов А. И.// Электротехническая промышленность Сер. Электротехнические материалы. 1974, вып. 1 (54), с. 24–26.
46. Фомин А. В., Борисов В. Ф., Чермошенский В. В. Допуски в радиоэлектронной аппаратуре. М.: Сов. радио, 1973
47. Гнеденко Б. В. Курс теории вероятностей. М.: Физматгиз, 1961.
48. Ермаков С. Н., Михайлов Г. А. Курс статистического моделирования. М.: Наука, 1976
50. Шторм Р. Теория вероятностей. Математическая статистика, статистический контроль качества: Пер. с нем. М.: Мир, 1970.
51. Бусленко Н. П. Моделирование сложных систем. М.: Наука, 1978.
52. Сигорский В. П. Анализ электронных схем. Киев: Гостехиздат УССР, 1964.
53. Заболотни Р. А.// Изв. вузов. Сер. Радиоэлектроника. 1986, т. 29, № 6. С. 95–97.

54. Сигорский В. П.// Изв. вузов. Сер. Радиоэлектроника. 1986, т. 29, № 6, с. 3–15.
55. Геворкян Г. Х., Геворкян В. Х.// Тр. ВНИИКЭ, т. 8, Ереван: 1976, с. 255–265.
56. Башкис А. И.// Изв. вузов. Сер. Радиоэлектроника. 1986, т. 29, № 6, с. 98–99.
57. Соболев И. М., Статников Р. Б. Выбор оптимальных параметров в задачах со многими критериями. М.: Наука, 1981.
58. Исаев И. П., Иньков Ю. И., Маричев М. А. Вероятностные методы расчета полупроводниковых преобразователей. М.: Энергоатомиздат, 1983.
59. Бахвалов Н. С. Численные методы. М.: Наука, 1975.
60. Калиткин Н. Н. Численные методы. М.: Наука, 1978.
61. Крылов В. П., Бобков В. В., Монастырский П. И. Вычислительные методы. Т. 1. М.: Наука, 1976.
62. Гильберг А. Как работать с матрицами: Пер. с нем. М.: Статистика, 1981.
63. Жуков Л. А., Стратан И. П. Установившиеся режимы сложных электрических сетей и систем. М.: Энергия, 1979.
64. Корн Г., Корн Т. Справочник по математике для научных работников и инженеров. М.: Наука, 1977.
65. Бронштейн И. Н., Семендяев К. А. Справочник по математике для инженеров и учащихся вузов. М.: Наука, 1980.
66. Muench U., Bunzel E., Kunkel K.-H.// Elektrik, 1985. Bd 39, N 2. S. 42–474.
67. Дьяконов В. П. Справочник по алгоритмам и программам на языке БЕЙСИК для персональных ЭВМ. М.: Наука, 1987.
68. W. Nerretter. Berechnung elektischer Schaltungen mit dem Personal Computer. Carl Hanser Verlag: Munchen, 1987.
69. Баласаян В. Э., Богдюкевич С. В., Шахвердов В. А. Программирование на микроЭВМ "Искра-226". М.: Финансы и статистика, 1987.
70. Моррилл Г. БЕЙСИК для персонального компьютера фирмы ИБМ: Пер. с англ. М.: Финансы и статистика, 1987.
71. Уолш Б. Программирование на БЕЙСИКе: Пер. с англ. М.: Радио и связь. 1987.
72. Кетков Ю. Л. Диалог с мини- и микроЭВМ. М.: Наука, 1988.
73. Reinking E. F. Mathematische Routinen VC20// Elektrotechnik/ Elektronik. Braunschweig; Weisbaden: F. Vieweg and Sohn, 1983, VIII, s. 78.
74. Jobert P. Circuits electriques and systems. Methodes de calcul. Paris: Technique et Documentation. 1986. 206 p.
75. Fusco V. F. Microwave circuits. Analysis and computer-aided design. Englewood cliffs: Prentice-Hall, 1987.
76. Doberenz W.// Radio, Fernsehen, Elektronik, 1987. Bd. 36. N 7. S. 435–438.

Список печаток

Страница	Строка	Напечатано	Следует читать
17	16 сверху	N	N1
17	17 сверху	2×10^D	$N2/10^D$
20	19 сверху	математический	математический
56	8 снизу	5.58677	5.58677
174	18 снизу (4-я колонка таблицы)	APG	ARG
174	19 снизу (4-я колонка таблицы)	PY * PY	PX * PY

60 к.

ЭНЕРГОАТОМИЗДАТ